

BASIC COMPUTER

David H. Ahl

Spil

Bind I

BORGEN



**BASIC Computerspil
bind 1**

BASIC Computerspil, 1

Oversat fra amerikansk af Christian Borup.

Originalens titel:

BASIC Computer Games

Copyright © 1978 by Creative Computing Press

Danske udgave © Borgens Forlag 1986

Omslag: Jesper Bach Jensen

Tegninger: George Beker

Published by Borgens Forlag, Valbygaardsvej 33,

DK-2500 Copenhagen Valby

Trykt hos Narayana Press, Gylling

ISBN 87-418-7419-6

1. udgave. 1. oplag. 1986

David Ahl

BASIC

computerspil

Bind 1

Redigeret af David H. Ahl

Programændringer: Steve North

**Dansk oversættelse, bearbejdning og programmering:
Christian Borup**

Borgen

Tak

I en bog som denne er det svært at vide, hvem man skal begynde med at takke.

Forfatterne til spillene. I det hele taget takkes enhver, der på et eller andet tidspunkt har programmeret spil.

Steve North. Fordi han, mere end pligten forlangte, har skrevet om, forandret og afluset alle spillene skrevet i MITS BASIC. Et virkeligt Herkulesarbejde!

John Lee. Fordi han hjalp Steve North.

Bob Albrecht. For hans (fortløbende) bidrag til computerspillenes verden, og fordi han fortalte mig, hvem der skrev hvad.

Dartmouth College. Fordi computerspil her blev anerkendt som et virkeligt pædagogisk redskab, og fordi vi fik tilladelse til at skrive og prøvekøre programmerne på Dartmouth Timesharing System.

Om forfatteren:

David H. Ahl har været interesseret i datamater i mange år. I 1956 lærte han at programmere i ALGOL og CAL (Cornell Assembly Language) på en Burroughs B-200. Siden da har han prøvet kræfter med så forskellige uhyrer som: Bendix G-15 og G-20, IBM 650, 704, 7090, 1401, 1130, 360/40 og 360/50, CDC 3200 og 3600, GE 115, 235 og 635, HP 2000, NCR Century 50 og 100, DEC PDP-8, 10, 11, 12 og 15 samt så at sige alle eksisterende mikrodatamater.

Han skrev sit første computerspil allerede ugen efter, at han havde lært at programmere. Siden da har han arbejdet med både alvorlige spil (som f.eks. Carnegie-Mellon Management Game og energisimulering i USA) og underholdningsspil.

Hans liv har altid været knyttet til datamater inden for områder som marketing, produktprognose, pædagogisk forskning, salg af datamater til skoler og universiteter, og han har fungeret som lærer og forelæser. Som udgiver af det amerikanske tidsskrift »Creative Computing« har han fortsat med at udvikle nye områder, hvor datamater kan anvendes.

Indhold · bind 1

Indledning · 6	
Om denne bog · 6	
BASIC · 7	
Oversættelse til andre former for BASIC · 9	
Til slut · 10	
Forord til den danske udgave · 11	
Spillene:	
AKTIER · 12	
AMERIKANSK FODBOLD · 15	
AMERIKANSK FODBOLD, VERSION 2 · 19	
AWARI · 22	
BAGELS · 24	
BAGLÆNS · 26	
BANNER · 28	
BASKETBALL · 30	
BILLE · 33	
BOGSTAV · 36	
BOKSNING · 37	
BOMBARDEMENT · 39	
BONDESPIL · 41	
BORGERKRIG · 44	
BOWLING · 49	
BUNKE · 52	
BYTTEPENGE · 53	
CIFFER · 54	
CRAPS · 56	
DAM · 57	
DAM FOR EN · 59	
DRONNINGESKAK · 61	
DYRERIGET · 63	
ENOGTYVE · 65	
FLIP-FLOP · 68	
FRASEGENERATOR · 69	
GODDAV · 70	
GOLF · 72	
Appendix · 75	
Fortegnelse over spillene opdelt i kategorier · 77	

Indhold · bind 2

Om denne bog · 6
GOMOKO · 7
GÆT · 8
HAMMURABI · 9
HAPS · 11
HASARD · 13
HESTEVÆDDELØB · 14
HOP · 16
HURKLE · 17
HÆNGT MAND · 18
HØJT-LAVT · 21
HØVDING · 22
ISHOCKEY · 23
KALENDER · 27
KAMP · 29
KANIN · 31
KAST · 32
KAST BOMBERNE · 33
KEMIKER · 35
KONGE · 36

KRIG · 39
KRYDS OG BOLLE · 40
KRYDS OG BOLLE, VERSION 2 · 41
KRYDS OG BOLLE I 3-D · 42
KUBUS · 45
LABYRINT · 47
LIGE VINDER · 49
LIGE VINDER, VERSION 2 · 50
LITTERATUR QUIZ · 52
LIV · 53
LIV FOR TO · 56
LOVE · 58
MASTER MIND · 59
MATEMATISK TERNING · 62
MÅL · 63
MÅNELANDING · 65
MÅNELANDING, L.E.M. · 67
MÅNELANDING, RAKET · 70
NAVN · 72
Appendix · 73
Fortegnelse over spillene opdelt i kategorier · 75

Indhold · bind 3

Om denne bog · 6
NICOMACHUS · 7
NIM · 8
OMRING · 10
ORBIT · 11
ORD · 13
PELSHANDLER · 14
PILESPIL · 16
PIZZA · 18
PLOT I 3-D · 20
POESI · 21
POKER · 22
ROULETTE · 26
RUDERE · 29
RUSSISK ROULETTE · 30
SAKS-STEN-PAPIR · 31
SINUSKURVE · 32
SKYTTE · 33
SLAGSKIB · 34
SLALOM · 37
SMUGORM · 39
SOLITAIRE · 41
SPLAT · 44
STJERNER · 46
SUPER STAR TREK INSTRUKTIONER · 47
SUPER STAR TREK X · 50
SYNONYM · 57
SÆNKE SLAGSKIBE · 58
TAL · 61
TERNING · 62
TOG · 63
TYREFÆGTNING · 64
TYVEKNÆGT · 66
TÆNDSTIKKER · 68
TÅRNET I HANOI · 69
UBÅDSJAGT · 71
UGEDAG · 72
Appendix · 74
Fortegnelse over spillene opdelt i kategorier · 76

Indledning

Du sidder i en blød og velpolstret lænestol, der vibrerer en smule. Pludselig fornemmer du en rystelse, og din dataskærm, som for lidt siden var helt sort, viser nu et rumskib, der forsvinder i det fjerne. Rumskibet vil fortsætte med at kredse rundt om månen, mens du skal forsøge at lande dit månelandingsmodul, så måneoverfladen kan udforskes. Dataskærmen viser nu, at du er ved at kæntre. Du trykker derfor på din højre styreknop, der tænder for stabiliseringsraketterne på styrbord side. Endnu et par tryk mod højre, lidt til venstre, og så til højre igen. Dataskærmen viser nu, at det er lykkedes dig at stabilisere rumfartøjet. Uden forstyrrelser – men med alt for høj hastighed – nærmer du dig måneoverfladen. Pludselig rammes du af panik, fordi du mærker, at du har brugt alt for megen tid på stabiliseringsraketterne. Du sætter bremsaraketterne på fuld tryk. Din dataskærm er nu helt fuld af flammer og røg, så du stiller over på computerstyret dataskærm, der viser et grafisk diagram over månelandingsmodulets position på dets vej mod måneoverfladen. Det er lykkedes dig at nedsætte hastigheden tilstrækkeligt, men du opdager, at du er på vej ned mod siden på et månekrater. Du har ikke tid til yderligere navigation. Du slår panoramadataskærmen til og håber på det bedste. Uheldigvis er kratervæggene så stejle, at et af modulets landingsstel rammer kratervæggen skævt, så fartøjet vælter. Rumkontrollen i Houston sender dig følgende meddelelse: »Målinger viser alvorlige skader på månelandingsmodulet. Du har ilt til 13.2 dage. Yderligere oplysninger om redningsforsøg følger. Der er ingen grund til panik«.

Nu trykker du på en anden knap på tastaturet, og dataskærmen viser teksten: »Månelandingen er afsluttet. Vil du spille igen?«. Med fugtige håndflader læner du dig tilbage i lænestolen og skriver: »Nej. Hent Fodbold«. Et øjeblik efter kommer en dommer og en fodboldspiller til syne på skærmen. Dommeren vender sig imod dig og spørger: »Plat eller krone?«

Synes ovenstående langt ude? Det er det slet ikke. Teknisk set er det muligt allerede i dag. Hvis vi gør os nogle tanker om omkostningerne, så vil der sandsynligvis gå mange år, før et system med sådanne muligheder ligger inden for den almindelige brugers rækkevidde, men allerede i dag findes mange helt utrolige spil for hjemme- og mikrodatamater. Det er disse spil, nærværende bog handler om. Spil til hjemmedatamater, mikrodatamater, terminaler med forbindelse til main-frame datamater, osv. – kort sagt spil til alle datamater, der kan tale BASIC.

Computerspil er ikke noget nyt fænomen. Allerede i 1952 – kun kort tid efter den første kommercielle datamat var blevet introduceret – skrev A.L. Samuel fra IBM et damspil på en IBM 701 computer. Dette spil blev skrevet med den bagtanke, at man kan lære en hel del om de menneskelige tankeprocesser, hvis man kan simulere dem på en datamat. Af samme grund skrev også Nevell, Shaw og Simon fra Rand Corporation det allerførste skakprogram. Men selv de brugere, der ikke kendte kunstig intelligens, fik stor glæde af programmet (også selv om de ikke var perfekte til at spille dam eller skak).

Selv om disse programmer blev skrevet som dele af forskningsprojekter, så var endnu flere ansatte i al hemmelighed i færd med at skrive og køre spil på deres arbejdsgiveres datamater, enten før eller efter arbejdstid eller måske i nogle tilfælde i frokostpausen. Der fandtes altid mindst to eller tre spillefanatikere ved ethvert større dataanlæg. Da minidatamaterne og timesharingssystemerne begyndte at komme frem i 1960'erne øgedes antallet

af edb-entusiaster, og i 1966 begyndte de at mødes for at planlægge turneringer i computerskak.

Inderkredsen af disse »hackere« – den virkelig hårde kerne – var især dem, der deltog i rumkrigen (spacewar). Det første rumkrigsspil blev forfattet allerede i 1961-62 af nogle »hackere« fra elektronikafdelingen på MIT. Dette program blev skrevet på en DEC PDP-1, den første minidatamat. Rumkrigsspillet lagde grunden til en fantastisk stor skare entusiaster, der spillede, modificerede, forbedrede og eksperimenterede.

»Åh, rumkrig. Hver eneste nat er der hundreder af driftsoperatører, der sidder fanget af dataskærmenes mystik, mens de i timevis fascineres af spillet på liv og død. De sidder med øjne, der løber i vand, og med følelseløse fingre der endeløst bearbejder tastaturet, mens de med glæde slagter deres venner og misbruger deres arbejdsgiveres kostbare kørselstid. Der er ved at ske noget.« (Steward Brand i bogen: »II Cybernetic Frontiers«, Random House, New York, 1974).

Eksempel: Oktober 1972. PDP-10 computeren i Stanfords laboratorier for kunstig intelligens er reserveret fra klokken 20.00, for afholdelse af »Den intergalaktiske rumkrigssolympiade«.

Eksempel: Oktober 1976. Computerfirmaet Cromemco præsenterer programmet »Spacewar« for 8080-processoren og TV-Dazzler. Prisen er 15 dollars.

For en lægmand betyder disse to eksempler, at spillet »Spacewar« på kun fire år udviklede sig fra et spil der kun kunne køres på et dataanlæg til flere millioner kroner, til et spil beregnet for en hjemmedatamat, der kostede under 8000,- kroner.

Hvad sker der med en kult, når man åbner templets porte og lader alle tage skatterne med sig hjem? Ærlig talt, så ved vi det ikke endnu. Tempelportene har ikke været åbne længe nok, men det er sandsynligt, at børnene i en generation, der ikke længere kan regne i hovedet eller bruge en regnestok – fordi de bruger lommeregnerne – kommer til at lære, at en TV-skærm kan være et nyttigt værktøj til aktivitet, og ikke nødvendigvis en indretning hvor man passivt henter information.

Om denne bog

BASIC computerspil: Nærværende mikrodatamat-udgave er en kraftig bearbejdet version af min første bog: »101 BASIC Computer Games«, der i 1973 blev udgivet af Digital Equipment Corporation. Programmerne i den første bog var skrevet i seks forskellige versioner af BASIC. Programmerne i nærværende bog er alle skrevet i Microsoft BASIC. I den første bog fandtes flere programmer, der enten ikke kunne omskrives til Microsoft BASIC, eller som ikke var besværet værd. Disse programmer er fjernet og erstattet af nye. Jeg siger tak til alle, som med tålmodighed og forstørrelsesglas tydede de næsten ulæselige programmer i den første bog og fik dem til at virke.

David H. Ahl
September 1978.

BASIC

Næsten alle spil i bogen er skrevet i Microsoft BASIC. Nedenfor følger en kort beskrivelse af nogle af instruktionerne i Microsoft BASIC. Nedenstående liste af BASIC kodeord vil lette en eventuel oversættelse til andre mere specielle versioner af BASIC. Afvigelser fra »normal« BASIC er angivet for sig. Programmet »BASIC-TEST«, der findes i denne bogs appendiks, indeholder alle de instruktioner, der er anvendt i bogens programmer. Indtast og kød dette program, inden du går i gang med resten af programmerne.

Instruktioner

DATA

Strengene i DATA-sætninger i denne bog står alle i anførelsestegn. Det er ikke alle BASIC-varianter, der kræver dette.

DEF FNX(X)

Hvis din version af BASIC ikke har DEF-funktionen, bør du indsatte hele funktionen på de steder, hvor FN-instruktionen forekommer. Hvis der ikke er plads, kan der eventuelt laves en subroutine.

DIM

DIM-instruktionen anvendes her til dimensionering af både numeriske variabler og strengvariabler. Bemærk, at index starter med 0, og ikke med 1. De fleste af spillene anvender ikke-indicerede variabler med index 0 (Se også kommentar om oversættelse til HP BASIC).

END

Visse BASIC-varianter kræver, at programmet afsluttes med END-instruktionen.

FOR . . . TO . . . STEP

Anvendes som i standard BASIC, dog med den undtagelse, at løkkens test først udføres, når løkken er gennemløbet en gang. Dette betyder, at hvis man kører programmet:

```
10 FOR X = 2 TO 1
20 PRINT "DAVS"
30 NEXT X
40 END
```

Så udskrives »DAVS« på skærmen. Denne lille afvigelse skulle dog ikke give problemer med BASIC-varianter, der tester løkken, inden denne udføres. Når en løkke som:

```
10 FOR I = 1 TO 2
20 PRINT
30 NEXT I
40 PRINT I
50 END
```

er gennemløbet, har løkkens kontrolvariabel I en værdi, der er lig slutværdien + 1. I ovenstående eksempel udskrives således 3 på skærmen.

GOTO, GOSUB

Begge disse instruktioner skal efterfølges af det programlinienummer, der skal hoppes til.

IF . . . THEN

Denne instruktion kan efterfølges af både et udtryk og/eller et li-

nienummer. Bemærk, at dersom IF-testen er falsk, så kører programmet videre fra næste linienummer. I visse BASIC-varianter kører programmet videre fra den næste instruktion. Hvis nedenstående program udskriver »FEJL I TEST« på skærmen, så tilhører din BASIC den sidstnævnte gruppe. Hvis programmet skriver »TEST I ORDEN«, vil du ikke få problemer med denne instruktion.

```
10 A = 1
20 IF A = 0 THEN PRINT "TEST": PRINT "FEJL I TEST": GOTO
40
30 PRINT "TEST I ORDEN"
40 END
```

INPUT

I bogens programmer er INPUT-instruktionen altid anvendt på en linie for sig.

LET

Visse varianter af BASIC kræver, at LET-instruktionen altid skal anvendes, når variabler skal tilskrives værdier. Hvis din BASIC ikke kan acceptere en linie som: 10 A = 1, så omskriv denne til: 10 LET A = 1, så er dette problem løst. Instruktionen LET anvendes ikke i bogens programmer.

NEXT

Denne instruktion efterfølges af løkkens kontrolvariabel. I bogens programmer angives denne altid, selv om visse BASIC-versioner ikke kræver denne nævnt.

ON . . . GOTO, ON . . . GOSUB

Beregnet GOTO og GOSUB. Begge instruktioner efterfølges af en liste af linienumre, der er adskilt af kommaer. Ved en variabelværdi på 1 hoppes til første linienummer, ved en værdi på 2 hoppes til andet linienummer, osv.

PRINT

Denne instruktion udskriver den variabel eller den tekststreng (i anførelsestegn), der står efter instruktionen. Når to af PRINT-instruktionens udtryk er adskilt af semikolon (;), udskrives disse helt tæt sammen. Hvis to udtryk er adskilt af komma (,), tabuleres 16 tegn til højre. Måske anvendes anden tabuleringsværdi på din datamat.

READ

Denne instruktion indlæser den næste numeriske variabel eller strengvariabel i en DATA-sætning et andet sted i programmet.

REM

Alle informationer, der indtastes på linien efter REM-instruktionen, opfattes af programmet som notater, der ikke vil blive udført som instruktion.

RESTORE

Når denne instruktion er udført, starter næste READ-instruktion med at læse det første element i programmets første DATA-sætning.

RETURN

Slut subrutinen og hop tilbage til hovedprogrammet og udfør den instruktion, der står umiddelbart efter den GOSUB-instruktion, der hoppede til denne subroutine.

Funktioner

ABS(X)

Returnerer den absolutte værdi af X.

ASC(X\$)

Returnerer ASCII-værdien for det første tegn i argumentet (her indholdet af strengvariablen X\$). ASC("A") er 65, ASC("B") er 66, osv.

ATN(X)

Returnerer arctangens til vinklen X (hvor X måles i radianer).

CHR\$(X)

Forvandler ASCII-værdien til en tegnstring. CHR\$(65) er "A", CHR\$(66) er "B". CHR\$(7) er klokke tegnet (BELL), der får de fleste computere til at sige pip når den udskrives i en PRINT-sætning: PRINT CHR\$(7).

COS(X)

Returnerer cosinus for en vinkel X (hvor X måles i radianer).

EXP(X)

Returnerer exponentialfunktionen, dvs. værdien for e (2.71828) opløftet til X'ende potens.

INT(X)

Returnerer største heltal der er mindre end X.

LEFT\$(X\$,Y)

Returnerer de Y første tegn i strengvariablen X\$.

LEN(X\$)

Returnerer det antal tegn, som strengvariablen indeholder, dvs. længden af X\$.

LOG(X)

Returnerer den naturlige logaritme (med grundtallet e, dvs. 2.71828) af X.

MID\$(X\$,Y,Z)

Returnerer Z tegn fra X\$, startende fra tegn nummer Y i X\$.

RND(1)

Danner tilfældige tal. Skal give en værdi mellem 0 og 1. I Microsoft BASIC bør tilfældighedsgeneratoren initialiseres ved at indsætte en RANDOMIZE-instruktion i programmets begyndelse, før RND(1) anvendes. Dette sikrer, at der dannes nye tilfældige tal, hvergang programmet køres. Hvis du får problemer med denne instruktion, så læs om denne instruktion i den BASIC-manual, der fulgte med din datamat.

RIGHT\$(X\$,Y)

Returnerer alle tegn i strengen X\$, til højre for og inklusive tegn nummer Y.

SGN(X)

Tegnfunktion. Returnerer $\div 1$ hvis X er negativ, 0 hvis X er nul, og hvis X er positiv.

SIN(X)

Returnerer sinus for vinklen X (hvor X måles i radianer).

SQR(X)

Uddrager kvadratroden af X. X må ikke indeholde en negativ værdi.

STR\$(X)

Forvandler den numeriske værdi X til en numerisk streng. Hvis X har værdien 8.45, vil strengen blive lig med »8.45«.

TAB(X)

Flytter markøren til position X tegn fra skærmens venstre side. Fungerer også på printeren.

TAN(X)

Returnerer tangens for vinklen X (hvor X måles i radianer).

VAL(X\$)

Returnerer den numeriske værdi af strengen X\$. Er det modsatte af STR\$(X)-funktionen.

I Microsoft BASIC kan ethvert udtryk vurderes som værende sandt eller falsk. Hvis et udtryk er sandt, returnerer Microsoft BASIC $\div 1$, og hvis det er falsk, fås værdien 0. Hvis vi således skriver:

```
10 Q=(X=Y)
```

bliver Q lig $\div 1$ hvis X er lig Y, mens Q bliver 0, hvis X er forskellig fra Y. I visse versioner af BASIC returneres 1 (og ikke $\div 1$), hvis udtrykket er sandt. For at undgå problemer med dette anvender alle bogens programmer den absolutte værdi (ABS) af udtrykket, idet begge BASIC versioner så vil give samme resultat.

Logiske udtryk som de nævnte anvendes især i »Bondespillet«, hvor de optræder i den brugerdefinerede funktion. Hvis din BASIC slet ikke kan evaluere logiske udtryk, kan man erstatte de logiske udtryk med en tabelsøgning. Andre af spillene anvender de logiske operatører OR og AND på sædvanlig vis.

Når der anvendes flere instruktioner på en programlinie, anvendes et kolon (:) som skilletegn. Visse former for BASIC kræver andre skilletegn.

Oversættelse til andre former for BASIC

Bogens programmer vil i langt de fleste tilfælde kunne køre på din datamat uden at skulle oversættes, idet er skrevet i en forenklet form for Microsoft BASIC, hvor der er anvendt så få BASIC-ord som muligt. Nedenfor er dog, for de mest brugte former for BASIC, gennemgået de områder, hvor ændring eller oversættelse kan blive nødvendig. Bogens første program: »BASIC-TEST« er til stor hjælp ved lokalisering af de instruktioner, der giver problemer på netop din datamat.

ABC80/800 BASIC

Alle programmer kan køres direkte på ABC-datamaterne. ABC-BASIC behøver intet argument ved RND, og du kan således nøjes med at skrive RND. TAB(0) accepteres ikke af denne BASIC-variant, TAB skal mindst have værdien 1 (TAB(1)).

AMSTRAD 464/664/6128

Programmerne kører uden væsentlige ændringer.

APPLE II BASIC (Applesoft)

Alle programmer fungerer direkte eller med kun få ændringer. Bemærk at Applesoft ikke som Microsoft BASIC medtager blanktegn foran og efter udskrift af numeriske variabler. Når numeriske variabler skal udskrives, skal disse derfor udskrives med et blanktegn før og et blanktegn efter. F.eks. skal linien

```
10 PRINT A
```

udskiftes med:

```
10 PRINT " ";A;" "
```

ATARI

Brugerdefinerede funktioner ikke tilladt (Se afsnittet »Til slut«). Undersøg om strenge skal dimensioneres. Instruktionerne LEFT\$, MID\$, RIGHT\$ erstattes med strengfunktionen. ON . . . GOTO erstattes med serie af IF . . . THEN-sætninger.

BASIC E (under CP/M)

Alle sæt skal dimensioneres, idet der ikke findes standardværdier. Flere sætninger på samme programlinie er ikke tilladt.

BBC/ELECTRON BASIC

Programmerne kører uden væsentlige ændringer.

COMMODORE 64/VIC 20

Programmerne kører uden væsentlige ændringer.

CROMEMCO 16 Extended BASIC

Tillader brug af flere udtryk på samme programlinie, men visse udtryk skal stå sidst på programlinien, så derfor skal der foretages visse ændringer alligevel. Sæt med to dimensioner skal altid dimensioneres. Strenge i Cromemco BASIC skal håndteres som i HP BASIC, og ikke som i Altair BASIC.

DEC BASIC PLUS

Programmer, der er skrevet i Altair BASIC, bør være direkte kompatible med DEC BASIC PLUS.

DRAGON BASIC

Programmerne kører uden væsentlige ændringer.

ENTERPRISE

ASC-funktionen erstattes med ORD. Instruktionerne LEFT\$,

MID\$, RIGHT\$ erstattes af strengfunktionen. Hver ON . . . GOTO-sætning erstattes af en række IF . . . THEN-sætninger.

HP BASIC

Der kræves en del ændringer, før visse af programmerne kan køre i HP BASIC. For det første tillader denne BASIC ikke indexerede sæt med et index på 0. Hvis et sådant index anvendes, vil HP BASIC udskrive en fejlmeddelelse og oplyse, at index er uden for de tilladte grænser. Løsningen på dette problem er at lægge 1 til alle index. For det andet har HP BASIC ikke mulighed for at anvende strengsæt. Hvis et program derfor bruger strengsæt, så må dette skrives om. Man kan f.eks. overføre sættet til en enkelt streng, en numerisk liste, en serie DATA-sætninger eller eventuelt en datafil. HP BASIC har heller ikke funktionerne LEFT\$, MID\$ og RIGHT\$. Udskift dem i stedet med de normale index-metoder, der anvendes i HP-BASIC. F.eks. bliver RIGHT\$(X\$,2) oversat til: X\$(LEN(X\$)-1,LEN(X\$)).

IBM PC-BASIC A

Programmerne kører uden væsentlige ændringer.

IMSAI 8K BASIC

Programmerne kører uden væsentlige ændringer.

NORTH STAR DISK BASIC

North Star BASIC håndterer strenge ligesom HP BASIC, og derfor må visse strengrutiner skrives om. SQR-funktionen skal udskiftes med SQRT-funktionen.

MEMOTECH

LOG-funktionen erstattes med LN-funktionen.

MSX-BASIC

Programmerne kører uden væsentlige ændringer.

OHIO SCIENTIFIC BASIC

De fleste programmer kan køre direkte på Challenger 2P og de større systemer. Da Challenger 1P og Superboard kun viser 32 tegn pr. linie på skærmen, skal visse udskriftsrutiner skrives om.

ORIC-1/ATMOS

Programmerne kører uden væsentlige ændringer.

PET BASIC

Commodores PET BASIC er så at sige identisk med den BASIC, der anvendes i denne bog. Dog udskriver PET BASIC ikke blanktegn på hver side af variablerne (Se Apple II BASIC).

POLYMORPHIC 11K BASIC

I denne BASIC-variant adskilles flere instruktioner på en programlinie af en skråstreg, og ikke af et kolon. SQRT-funktionen anvendes i stedet for SQR. Det er ikke tilladt at forlade en FOR . . . NEXT-løkke, før den er kørt til ende, medmindre EXIT-instruktionen bruges. Så på alle steder hvor en IF . . . THEN-sætning befinder sig inde i en FOR . . . NEXT-løkke, skal du bruge EXIT. Bemærk, at der anvendes RND(0) og ikke RND(1) til at bestemme tilfældige tal.

QL/SINCLAIR BASIC

Erstat ASC-funktion med CODE-funktion, ATN- med ATAN-funktion og LOG- med LN-funktion. Indsæt en END-DEF-instruktion efter definitionerne af de brugerdefinerede funktioner. LEFT\$, MID\$, RIGHT\$- og STR\$-funktioner udskiftes med

strengfunktionen. Erstat RND(1) med RND(0 TO 1). END-instruktionerne erstattes af REM-instruktioner. Bemærk, at VAL-funktionen ikke findes i QL-BASIC.

RESEARCH MACHINES 480Z

Programmerne kører uden væsentlige ændringer.

SHARP MZ-80K

(Tape BASIC). LOG-instruktionen erstattes af LN-instruktionen.

SOL BASIC

Bogens programmer kan køres direkte i SOL Extended BASIC. Visse af programmerne kan endda køre i 5K BASIC, men de fleste kræver større ændringer.

SPECTRUM

Erstat ASC-funktion med CODE-funktion og LOG- med LN-funktion. Ændringer i brugerdefinerede funktioner. LEFT\$, MID\$- og RIGHT\$-funktionerne udskiftes med strengfunktionen. ON . . . GOTO-sætninger erstattes med en række IF . . . THEN-sætninger. Erstat RND(1) med RND.

SWTPC 6800 BASIC

Version 2.0 og senere versioner svarer faktisk til den i bogen anvendte BASIC. Ligesom HP BASIC tillader SWTPC BASIC ikke, at indicerede variabler har et index på 0. Strengene må ikke indeholde over 32 tegn. Alle RND(1)-instruktioner skal udskiftes med RND(0). Dette gælder også for MSI DISK BASIC.

TANDY BASIC

Cirka 25 af programmerne, der ikke anvender strenge, kan køres med Level 1 BASIC. Yderligere kan cirka 20 køre, dersom input-sætninger med streng-indtastning omskrives til numerisk input (f.eks. 1 i stedet for JA, 0 i stedet for NEJ, etc.). Hvis man har Level 2 BASIC skal RND(1) udskiftes med RND(0). Visse af de brugerdefinerede funktioner skal skrives om.

TDL ZAPPLE 8K og 12K BASIC

Alle programmerne bør kunne køre med begge varianter af TDL BASIC.

Til slut

Vi får ofte spørgsmål fra læsere som ikke kan få programmerne til at virke rigtigt. Husk derfor, at alle programmerne i bogen er ægte og virkelige, og at de alle sammen er blevet kørt, afprøvet og fungerer. Hvis du får nogle katastrofale resultater, som helt bremser programmet, eller hvis du opdager nogle småfejl, så er det sandsynligvis din egen version af programmet (og ikke bogens originale program), du skal give skylden. Oversættelse af bogens programmer til en BASIC-variant, der radikalt adskiller sig fra denne, (som f.eks. HP-BASIC eller Apple Integer BASIC), er meget vanskelige at foretage, så her må du regne med at du får problemer. Hvis du har et TRS Level II-system, skal du huske at ændre RND(1) til RND(0) og udvide de brugerdefinerede funktioner. Når brugerdefinerede funktioner er brugt, er det ofte for at forkorte indtastningen af et udtryk, og det eneste, du behøver at gøre, hvis du ønsker at fjerne DEF FN-instruktionen fra et program, er at erstatte kaldet til funktionen med selve funktionsudtrykket. Hvis det oprindelige program f.eks. lyder:

```
10 DEF FNR(X)=INT(X * RND(1)+1)
900 Q=FNR(10)+R
```

så ændrer du dette til:

```
10 REM FUNKTIONEN SLETTET
900 Q=INT(10 * RND(0)+1)+R
```

Læg mærke til, at vi erstattede parameteren X i funktionsdefinitionen med den aktuelle værdi fra funktionskaldet (10) i linie 900, og at vi ændrede RND(1) til RND(0).

Den mest sikre måde at finde fejl i programmer på er at lægge dem i hænderne på mange tusind brugere. Fordi vi har gjort dette, er vi overbevist om, at (næsten) alle fejl er rettede.

Forord til den danske udgave

I den danske udgave af Ahl's »BASIC Computer Games« er der anvendt to forskellige programformater. Det ene er beregnet for en skærm med en linielængde på 40 tegn, mens det andet format er beregnet til en skærm med en linielængde på 80. Visse af bogens programmer er beregnet til kørsel på linieskriver, og derfor kan kørslerne af disse programmer virke forvirrende ved første øjekast.

Der er overalt anvendt så »skrabet« en BASIC som muligt, og visse af programmerne er grundet forskelle mellem engelsk og dansk grammatik revideret kraftigt. Det samme gælder for visse af programmernes udskriftsrutiner.

I den danske oversættelse er der, i bogens appendiks, tilføjet programmet BASIC-TEST, der giver mulighed for at se en fuld prøvekørsel af et program, der indeholder alle de BASIC-instruktioner som resten af bogens programmer indeholder. Det anbefales derfor, at man starter med at indtaste og prøvekøre BASIC-TEST, idet man derved hurtigt får et overblik over, hvor eventuelle afvigelser fra bogens BASIC kan optræde.

Christian Borup
Oktober 1985

Aktier

Dette program »spiller på børsen«. Du vil få tildelt 10.000 kroner, og du kan sælge eller købe aktier. Markedstendenserne dannes tilfældigt, og derfor repræsenterer dette program ikke det, der virkelig foregår på børsen, (afhængigt af, hvilken personlig mening du har om dette, føler du måske, at programmet netop derfor giver et ganske godt billede af virkeligheden).

Hver dag børsen har åben, udskrives en tabel med dine aktier, deres kurser og det antal aktier, du allerede besidder. Derefter udskrives forkortelserne for hver af aktierne med et spørgsmålstegn efter. Du kan nu indtaste, om du ønsker at købe eller sælge. Du indtaster en positiv værdi, hvis du vil købe, og en negativ værdi hvis du vil sælge, eller et nul (0) hvis du ikke ønsker at foretage dig noget. Ved hver aktiehandel beregnes et mæglerløn på 1% (det er meget billigt sluppet!). Bemærk: selv om en akties værdi falder til nul (konkurs), kan den nogle gange stige igen.

Dette program blev skabt af D. Pessel, L. Braun og C. Losik fra Huntington Computer projektet ved SUNY, Stony Brook, New York.

```

MORRISTOWN    CREATIVE COMPUTING    NEW JERSEY
-----
                                AKTIER
-----
ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DETTE PROGRAM SIMULERER EN AKTIEBØRS. DU VIL FÅ 10.000 KR.
UDLEVERET, OG DET ER DIN OPGAVE AT KØBE OG SÆLGE AKTIER.

```

AKTIEPRISERNE VIL BLIVE SKABT TILFÆLDIGT, OG DERFOR HAR DETTE PROGRAM INGEN FORBINDELSE MED HVAD DER SKER PÅ DEN RIGTIGE BØRS. EN OVERSIGT OVER DE AKTIER DER KAN HANDLES, DERES KURS, OG DET ANTAL AKTIER DU HAR I FORVEJEN UDSKRIVES. EFTER DENNE OVERSIGT VIL HVER AKTIES FORKORTELT BLIVE UDSKREVET, EFTERFULGT AF ET SPØRGSMÅLSTEGN. NU KAN DU FORETAGE EN TRANSAKTION. HVIS DU ØNSKER AT KØBE AKTIE IND TASTER DU '+NNN', OG HVIS DU ØNSKER AT SÆLGE IND TASTER DU '-NNN', HVOR 'NNN' ER ANTAL AF AKTIER. DER VIL BLIVE BEREGNET ET MÆGLERSLØN PÅ 1% AF ALLE TRANSAKTIONER. BEMÆRK, AT HVIS EN AKTIES VÆRDI FALDER TIL 0, KAN DEN GODT STIGE IGEN. DU HAR 10.000 KRONER SOM DU KAN INVESTERE. BRUG KUN HELTAL NÅR DU IND TASTER VÆRDIER. (FOR AT FÅ EN 'FORNEMMELSE' AF AKTIEMARKED, SÅ GØR DU KLOGT I AT OVERVÅGE DET I 10 DAGE, NÅR DU STARTER SPILLET).

```

----- HELD OG LYKKE -----
AKTIE                FORKORTELT                PRIS/AKTIE
-----
INTERNATIONAL BUSINESS MACHINES    IBM                79
DE DANSKE STATSBANER              DSB                79.75
BAYRISCHE MOTORWERKE              BMW                152
F.L.SMITH                          FLS                130.5
ZOLUNA DATA/TEXT                  ZDT                99.5

```

BØRSINDEX: 108.15

```

DU HAR IALT AKTIER FOR:      0    KR.
DU HAR IALT I KONTANTER:    10000  KR.
-----
DU ER IALT GOD FOR:        10000  KR.
=====

```

HVILKE TRANSAKTIONER VIL DU FORETAGE MED:

```

INTERNATIONAL BUSINESS MACHINES    IBM ?0
DE DANSKE STATSBANER              DSB ?0
BAYRISCHE MOTORWERKE              BMW ?0
F.L.SMITH                          FLS ?0
ZOLUNA DATA/TEXT                  ZDT ?0

```

***** SLUT PÅ DAGENS AKTIEOPKØB *****

AKTIE	PRIS/AKTIE	ANTAL	VÆRDI	KURSÆNDRING
IBM	76.25	0	0	-2.75
DSB	76.5	0	0	-3.25
BMW	139	0	0	-13
FLS	108.75	0	0	-21.75
ZDT	92.25	0	0	-7.25

-- BØRSINDEX: 98.55
NETTOÆNDRING: -9.6

```

DU HAR IALT AKTIER FOR:      0    KR.
DU HAR IALT I KONTANTER:    10000  KR.
-----
DU ER IALT GOD FOR:        10000  KR.
=====

```

ØNSKER DU AT FORTSÆTTE ('J'=JA, 'N'=NEJ) ?J

HVILKE TRANSAKTIONER VIL DU FORETAGE MED:

```

INTERNATIONAL BUSINESS MACHINES    IBM ?12
DE DANSKE STATSBANER              DSB ?1
BAYRISCHE MOTORWERKE              BMW ?4
F.L.SMITH                          FLS ?20
ZOLUNA DATA/TEXT                  ZDT ?0

```

***** SLUT PÅ DAGENS AKTIEOPKØB *****

AKTIE	PRIS/AKTIE	ANTAL	VÆRDI	KURSÆNDRING
IBM	78	12	936	1.75
DSB	74.25	1	74.25	-2.25
BMW	139	4	556	0
FLS	111.5	20	2230	2.75
ZDT	89.5	0	0	-2.75

-- BØRSINDEX: 98.45
NETTOÆNDRING: -.1

```

DU HAR IALT AKTIER FOR:      3796.25  KR.
DU HAR IALT I KONTANTER:    6240.27  KR.
-----
DU ER IALT GOD FOR:        10036.52  KR.
=====

```

ØNSKER DU AT FORTSÆTTE ('J'=JA, 'N'=NEJ) ?J

HVILKE TRANSAKTIONER VIL DU FORETAGE MED:

```

INTERNATIONAL BUSINESS MACHINES    IBM ?0
DE DANSKE STATSBANER              DSB ?10
BAYRISCHE MOTORWERKE              BMW ?0
F.L.SMITH                          FLS ?0
ZOLUNA DATA/TEXT                  ZDT ?20

```

***** SLUT PÅ DAGENS AKTIEOPKØB *****

AKTIE	PRIS/AKTIE	ANTAL	VÆRDI	KURSÆNDRING
IBM	73.5	12	882	-4.5
DSB	70.25	11	772.75	-4
BMW	126	4	504	-13
FLS	106.25	20	2125	-5.25
ZDT	79.25	20	1585	-10.25

-- BØRSINDEX: 91.05
NETTOÆNDRING: -7.4

```

DU HAR IALT AKTIER FOR:      5868.75  KR.
DU HAR IALT I KONTANTER:    3682.44  KR.
-----
DU ER IALT GOD FOR:        9551.19  KR.
=====

```

ØNSKER DU AT FORTSÆTTE ('J'=JA, 'N'=NEJ) ?J

HVILKE TRANSAKTIONER VIL DU FORETAGE MED:

INTERNATIONAL BUSINESS MACHINES IBM ?-12
DE DANSKE STATSANER DSB ?0
BAYRISCHE MOTORWERKE BMW ?0
F.L.SMITH FLS ?0
ZOLUNA DATA/TEXT ZDT ?5

***** SLUT PÅ DAGENS AKTIEOPKØB *****

AKTIE	PRIS/AKTIE	ANTAL	VÆRDI	KURSÆNDRING
IBM	69.5	0	0	-4
DSB	64.5	11	709.5	-5.75
BMW	116	4	464	-10
FLS	98.25	20	1965	-8
ZDT	70.5	25	1762.5	-8.75

-- BØRSINDEX: 83.75
NETTOÆNDRING: -7.3

DU HAR IALT AKTIER FOR: 4901 KR.
DU HAR IALT I KONTANTER: 4155.41 KR.

DU ER IALT GOD FOR: 9056.41 KR.
=====

ØNSKER DU AT FORTSÆTTE ('J'=JA, 'N'=NEJ) ?J

HVILKE TRANSAKTIONER VIL DU FORETAGE MED:

INTERNATIONAL BUSINESS MACHINES IBM ?0
DE DANSKE STATSANER DSB ?0
BAYRISCHE MOTORWERKE BMW ?0
F.L.SMITH FLS ?0
ZOLUNA DATA/TEXT ZDT ?0

***** SLUT PÅ DAGENS AKTIEOPKØB *****

AKTIE	PRIS/AKTIE	ANTAL	VÆRDI	KURSÆNDRING
IBM	74.25	0	0	4.75
DSB	57.75	11	635.25	-6.75
BMW	108.25	4	433	-7.75
FLS	82.25	20	1645	-16
ZDT	65.25	25	1631.25	-5.25

-- BØRSINDEX: 77.55
NETTOÆNDRING: -6.2

DU HAR IALT AKTIER FOR: 4344.5 KR.
DU HAR IALT I KONTANTER: 4155.41 KR.

DU ER IALT GOD FOR: 8499.91 KR.
=====

ØNSKER DU AT FORTSÆTTE ('J'=JA, 'N'=NEJ) ?N

JEG HÅBER DU HAR MORET DIG!!

```
100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
```

```
110 PRINT "-----"
```

```
120 PRINT " AKTIER"
```

```
130 PRINT "-----"
```

```
140 PRINT
```

```
150 REM Af D. Pessek, L. Braun, C. Losik
```

```
160 REM Aktier - program der simulerer en aktiebørs
```

```
170 REM Revideret den 18/8 1970
```

```
180 DIM S(5),P(5),Z(5),C(5),N$(5),F$(5)
```

```
190 FOR I = 1 TO 5
```

```
200 READ N$(I)
```

```
210 READ F$(I)
```

```
220 NEXT I
```

```
230 X = 1
```

```
240 A = INT (( RND (1) / 10) * 100 + .5) / 100
```

```
250 T5 = 0
```

```
260 X9 = 0
```

```
270 N1 = 0
```

```
280 N2 = 0
```

```
290 E1 = 0
```

```
300 E2 = 0
```

```
310 REM Introduktion
```

```
320 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
```

```
330 INPUT Z$
```

```
340 PRINT
```

```
350 IF LEFT$(Z$,1) = "N" THEN 570
```

```
360 PRINT "DETTE PROGRAM SIMULERER EN AKTIEBØRS. DU VIL FR 10.000 KR."
```

```
370 PRINT "UDLEVERET, OG DET ER DIN OPGAVE AT KØBE OG SÆLGE AKTIER."
```

```
380 PRINT
```

```
390 PRINT "AKTIEPRISERNE VIL BLIVE SKABT TILFÆLDIGT, OG DERFOR HAR DET--"
```

```
400 PRINT "TE PROGRAM INGEN FORBINDELSE MED HVAD DER SKER PÅ DEN"
```

```
410 PRINT "RIGTIGE BØRS. EN OVERSIGT OVER DE AKTIER DER KAN HANDLES,"
```

```
420 PRINT "DERES KURS, OG DET ANTAL AKTIER DU HAR I FORVEJEN UDSKRIVES."
```

```
430 PRINT "EFTER DENNE OVERSIGT VIL HVER AKTIES FORKORTELSE BLIVE"
```

```
440 PRINT "UDSKREVET, EFTERFULGT AF ET SPØRGSMALESTEGN. NU KAN DU"
```

```
450 PRINT "FORETAGE EN TRANSAKTION. HVIS DU ØNSKER AT KØBE AKTIEN"
```

```
460 PRINT "INDTASTER DU '+NNN', OG HVIS DU ØNSKER AT SÆLGE INDTASTER DU"
```

```
470 PRINT "'-NNN', HVOR 'NNN' ER ANTAL AF AKTIER. DER VIL BLIVE"
```

```
480 PRINT "BEREGNET ET MÆGLERSALÆR PÅ 1% AF ALLE TRANSAKTIONER."
```

```
490 PRINT "BEMÆRK, AT HVIS EN AKTIES VÆRDI FALDER TIL 0, KAN DEN GODT"
```

```
500 PRINT "STIGE IGEN. DU HAR 10.000 KRONER SOM DU KAN INVESTERE."
```

```
510 PRINT "BRUG KUN HELTAL NÅR DU INDTASTER VÆRDIER."
```

```
520 PRINT "(FOR AT FR EN 'FORNEMMELSE' AF AKTIEMARKED, SÅ GØR DU KLOGT"
```

```
530 PRINT "I AT OVERVÅGE DET I 10 DAGE, NÅR DU STARTER SPILLET)."
```

```
540 PRINT
```

```
550 PRINT "----- HELD OG LYKKE -----"
```

```
560 PRINT
```

```
570 REM Fremstilling af oversigt, input fra spiller
```

```
580 S(1) = 100
```

```
590 S(2) = 85
```

```
600 S(3) = 150
```

```
610 S(4) = 140
```

```
620 S(5) = 110
```

```
630 REM Første T8 - antal dage til første tendens (A)
```

```
640 T8 = INT (4.99 * RND (1) + 1)
```

```
650 REM Tilfældig første tendens (A)
```

```
660 IF RND (1) > .5 THEN 680
```

```
670 A = - A
```

```
680 REM Tilfældige start-værdier
```

```
690 GOSUB 1870
```

```
700 REM Hvad haves til starttidspunktet
```

```
710 FOR I = 1 TO 5
```

```
720 P(I) = 0
```

```
730 Z(I) = 0
```

```
740 NEXT I
```

```
750 PRINT
```

```
760 REM Initialisering af startbeløb (C)
```

```
770 C = 10000
```

```
780 PRINT "AKTIE FORKORTEELSE PRIS/AKTIE"
```

```
790 PRINT "-----"
```

```
800 FOR J = 1 TO 5
```

```
810 PRINT N$(J); " "; F$(J); " ";
```

```
820 V = S(J)
```

```
830 GOSUB 2500
```

```
840 PRINT
```

```
850 NEXT J
```

```
860 PRINT
```

```
870 Z4 = Z5
```

```
880 Z5 = 0
```

```
890 T = 0
```

```
900 FOR I = 1 TO 5
```

```
910 Z5 = Z5 + S(I)
```

```
920 T = T + S(I) * P(I)
```

```
930 NEXT I
```

```
940 Z5 = INT (100 * (Z5 / 5) + .5) / 100
```

```
950 Z6 = INT ((Z5 - Z4) * 100 + .5) / 100
```

```
960 REM Totalt aktiver (D)
```

```
970 D = T + C
```

```

980 IF X9 > 0 THEN 1010
990 PRINT "BØRSINDEX:";Z5
1000 GOTO 1030
1010 PRINT "-- BØRSINDEX:";Z5
1020 PRINT "NETTOÆNDRING:";Z6
1030 PRINT
1040 T = INT (100 * T + .5) / 100
1050 PRINT " DU HAR IALT AKTIER FOR:";
1060 V = T
1070 GOSUB 2500
1080 PRINT "KR."
1090 C = INT (100 * C + .5) / 100
1100 PRINT "DU HAR IALT I KONTANTER:";
1110 V = C
1120 GOSUB 2500
1130 PRINT "KR."
1140 PRINT "-----"
1150 D = INT (100 * D + .5) / 100
1160 PRINT " DU ER IALT GOD FOR:";
1170 V = D
1180 GOSUB 2500
1190 PRINT "KR."
1200 PRINT " ====="
1210 PRINT
1220 IF X9 = 0 THEN 1270
1230 PRINT "ØNSKER DU AT FORTSÆTTE ('J'=JA, 'N'=NEJ) ";
1240 INPUT Z$
1250 PRINT
1260 IF LEFT$(Z$,1) = "N" THEN 2480
1270 REM Indtast transaktion
1280 PRINT "HVILKE TRANSAKTIONER VIL DU FORETAGE MED:"
1290 PRINT
1300 FOR J = 1 TO 5
1310 PRINT N$(J);" ";F$(J);" ";
1320 INPUT Z(J)
1330 NEXT J
1340 PRINT
1350 REM Dagens totale køb i kroner (P5)
1360 P5 = 0
1370 REM Dagens totale salg i kroner (S5)
1380 S5 = 0
1390 FOR I = 1 TO 5
1400 Z(I) = INT (Z(I) + .5)
1410 IF Z(I) < = 0 THEN 1440
1420 P5 = P5 + Z(I) * S(I)
1430 GOTO 1490
1440 S5 = S5 - Z(I) * S(I)
1450 IF - Z(I) < = P(I) THEN 1490
1460 PRINT "DU HAR SOLGT FOR MANGE AKTIER - PRØV IGEN."
1470 PRINT
1480 GOTO 1280
1490 NEXT I
1500 REM Transaktionernes totale værdi (I5)
1510 I5 = P5 + S5
1520 REM Mæglersalær
1530 B5 = INT (.01 * I5 * 100 + .5) / 100
1540 C5 = C - P5 - B5 + S5
1550 IF C5 > = 0 THEN 1580
1560 PRINT "DU HAR BRUGT";(- 1) * C5;"KR MERE END DU HAR!"
1570 GOTO 1280
1580 C = C5
1590 REM Beregn ny total
1600 FOR I = 1 TO 5
1610 P(I) = P(I) + Z(I)
1620 NEXT I
1630 GOSUB 1870
1640 PRINT
1650 PRINT "***** SLUT PÅ DAGENS AKTIEOPKØB *****"
1660 PRINT
1670 PRINT
1680 IF X9 < 1 THEN 1690
1690 PRINT "AKTIE PRIS/AKTIE ANTAL VÆRDI KURSÆNDRING"
1700 PRINT "-----"
1710 FOR J = 1 TO 5
1720 PRINT F$(J);
1730 V = S(J)
1740 GOSUB 2500
1750 V = P(J)
1760 GOSUB 2500
1770 V = S(J) * P(J)
1780 GOSUB 2500
1790 V = C(J)
1800 GOSUB 2500
1810 PRINT
1820 NEXT J
1830 PRINT
1840 PRINT
1850 X9 = 1
1860 GOTO 860
1870 IF N1 > 0 THEN 1910
1880 I1 = INT (4.99 * RND (1) + 1)
1890 N1 = INT (4.99 * RND (1) + 1)
1900 E1 = 1
1910 IF N2 > 0 THEN 1950
1920 I2 = INT (4.99 * RND (1) + 1)
1930 N2 = INT (4.99 * RND (1) + 1)
1940 E2 = 1
1950 REM Træk 1 dag fra N1 og N2
1960 N1 = N1 - 1
1970 N2 = N2 - 1
1980 REM Gå igennem alle aktier
1990 FOR I = 1 TO 5
2000 X1 = RND (1)
2010 IF X1 > .25 THEN 2040
2020 X1 = .25
2030 GOTO 2130
2040 IF X1 > .5 THEN 2070
2050 X1 = .5
2060 GOTO 2130
2070 IF X1 > .75 THEN 2100
2080 X1 = .75
2090 GOTO 2130
2100 X1 = 0
2110 REM Stor ændringskonstant (W3)
2120 REM (Oprindeligt sat til 0)
2130 W3 = 0
2140 IF E1 < 1 THEN 2200
2150 IF INT (I1 + .5) < > INT (I + .5) THEN 2200
2160 REM Læg 10 point til denne aktie
2170 REM Nulstil E1
2180 W3 = 10
2190 E1 = 0
2200 IF E2 < 1 THEN 2270
2210 IF INT (I2 + .5) < > INT (I + .5) THEN 2270
2220 REM Træk 10 point fra denne aktie
2230 REM Nulstil E2
2240 W3 = W3 - 10
2250 E2 = 0
2260 REM C(I) indeholder forandring af aktieværdien
2270 C(I) = INT (A * S(I)) + X1 + INT (3 - 6 * RND (1) + .5) + W3
2280 C(I) = INT (100 * C(I) + .5) / 100
2290 S(I) = S(I) + C(I)
2300 IF S(I) > 0 THEN 2340
2310 C(I) = 0
2320 S(I) = 0
2330 GOTO 2350
2340 S(I) = INT (100 * S(I) + .5) / 100
2350 NEXT I
2360 REM Efter T8 dage skal tendens og hældning forandres
2370 T8 = T8 - 1
2380 IF T8 < 1 THEN 2400
2390 RETURN
2400 REM Tilfældig forandring af tendens og hældning,
2410 REM og tendensens varighed (T8)
2420 T8 = INT (4.99 * RND (1) + 1)
2430 A = INT ((RND (1) / 10) * 100 + .5) / 100
2440 S4 = RND (1)
2450 IF S4 < = .5 THEN 2470
2460 A = - A
2470 RETURN
2480 PRINT "JEG HÅBER DU HAR MORET DIG!!"
2490 GOTO 2670
2500 IF ABS (V) > = 1000000 THEN PRINT " ";GOTO 2570
2510 IF ABS (V) > = 100000 THEN PRINT " ";GOTO 2570
2520 IF ABS (V) > = 10000 THEN PRINT " ";GOTO 2570
2530 IF ABS (V) > = 1000 THEN PRINT " ";GOTO 2570
2540 IF ABS (V) > = 100 THEN PRINT " ";GOTO 2570
2550 IF ABS (V) > = 10 THEN PRINT " ";GOTO 2570
2560 PRINT " ";
2570 PRINT V;
2580 V = (ABS (V) - INT (ABS (V))) * 100
2590 IF V = 0 THEN PRINT " ";GOTO 2610
2600 IF V / 10 = INT (V / 10) THEN PRINT " ";GOTO 2610
2610 RETURN
2620 DATA "INTERNATIONAL BUSINESS MACHINES ","IBM"
2630 DATA "DE DANSKE STATSANER ","DSB"
2640 DATA "BAYRISCHE MOTORWERKE ","BMW"
2650 DATA "F.L.SMITH ","FLS"
2660 DATA "ZOLUNA DATA/TEXT ","ZDT"
2670 END

```

Amerikansk fodbold

Amerikansk fodbold (football) er antagelig det mest populære simulerede sportsspil i USA. Der findes endda dokumenterede eksempler på, at man har foretrukket at spille fodbold med datamaten fremfor at se en rigtig kamp i fjernsynet.

Du vil blive præsenteret for to versioner af dette populære spil. Den første version er mere »traditionel«, fordi du spiller mod datamaten. Du kan vælge mellem 7 offensive spilletaktikker. Når datamaten spiller forsvarsspil, synes den at spille en form for zonenforsvar, men her kan du ikke vælge spillemetode. Programmet præsenterer dig for de nødvendige regler, efterhånden som spillet skrider frem. Datamaten er også dommer og afgør de forskellige former for straf, når reglerne overtrædes.

Programmet er skrevet af John Kemeny ved Dartmouth-universitetet.

I den anden version fungerer datamaten som dommer, mens to mennesker spiller. Hver spiller får en liste med 20 spilletaktikker, hvor hver har en kodeværdi. Denne liste skal holdes hemmelig for modstanderen. Koderne kan ændres i DATA-sætningerne i programlinierne 160 (hold 1) og 170 (hold 2). Alle tyve spilleformer er offensive, en defensiv spillestil vælges ved at forsvare sig mod en bestemt offensiv spillestil. Et forsvar mod en lige aflevering kan meget vel anvendes mod en skæv aflevering, men fungerer ikke godt mod afleveringer bagud, etc.

Spillet blev oprindeligt skrevet af Raymond Mises fra Butler, Pennsylvania.

MORRISTOWN CREATIVE COMPUTING NEW JERSEY

AMERIKANSK FODBOLD

DETTE ER FODBOLDTURNERINGEN I DARTMOUTH.

DU ER HOLDLEDER FOR DARTMOUTH.

VÆLG MELLEMLØB NEDENSTRENGE FORMER FOR SPILLESTIL:

1=LØB FREMAD 2=DRIBLING FREMAD 3=KORT AFLEVERING
4=LANG AFLEVERING 5=STRENGE SKUD 6=HOPSKUD
7=LANGT SKUD

VÆLG DIN MODSTANDER ?CORNELL

CORNELL VANDT PLAT-OG-KRONE-SLAGNINGEN.

CORNELL VÆLGER AT MODTAGE.

52 YARDS KICKOFF.
32 YARDS RUNBACK.
BOLDEN HOS CORNELL PÅ 40 - YARDLINIEN
FØRSTE DOWN TIL CORNELL ***

KORT AFLEVERING. ER I ORDEN. 14 YARDS VUNDET.
BOLDEN HOS DARTMOUTH PÅ 46 - YARDLINIEN
FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. 2 YARDS VUNDET.
BOLDEN HOS DARTMOUTH PÅ 44 - YARDLINIEN
NED MED 2 YARDS TILBAGE: 8

LANG AFLEVERING. SLÆT NED. INGEN YARD GEVINST.
BOLDEN HOS DARTMOUTH PÅ 44 - YARDLINIEN
NED MED 3 YARDS TILBAGE: 8

LANG AFLEVERING. UFULDSTÆNDIG. INGEN YARD GEVINST.
BOLDEN HOS DARTMOUTH PÅ 44 - YARDLINIEN
NED MED 4 YARDS TILBAGE: 8

STRENGE SKUD. 34 YARDS SPARK.
9 YARDS TILBAGELØB.
BOLDEN HOS DARTMOUTH PÅ 19 - YARDLINIEN
FØRSTE DOWN TIL DARTMOUTH ***

VÆLG NY SPILLESTIL ?1

LØB FREMAD. 5 YARDS VUNDET.
BOLDEN HOS DARTMOUTH PÅ 24 - YARDLINIEN
NED MED 2 YARDS TILBAGE: 5

VÆLG NY SPILLESTIL ?3

KORT AFLEVERING. UFULDSTÆNDIG. INGEN YARD GEVINST.
BOLDEN HOS DARTMOUTH PÅ 24 - YARDLINIEN
NED MED 3 YARDS TILBAGE: 5

VÆLG NY SPILLESTIL ?5

STRENGE SKUD. 45 YARDS SPARK.
BOLDEN HOS CORNELL PÅ 31 - YARDLINIEN
FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. INGEN YARD GEVINST.
BOLDEN HOS CORNELL PÅ 31 - YARDLINIEN
NED MED 2 YARDS TILBAGE: 10

DRIBLING FREMAD. 12 YARDS VUNDET.
BOLDEN HOS CORNELL PÅ 43 - YARDLINIEN
FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. 3 YARDS VUNDET.
BOLDEN HOS CORNELL PÅ 46 - YARDLINIEN
NED MED 2 YARDS TILBAGE: 7

LANG AFLEVERING. ANGRIBER TACKLES. 7 YARDS TABT.
BOLDEN HOS CORNELL PÅ 39 - YARDLINIEN
NED MED 3 YARDS TILBAGE: 14

LANG AFLEVERING. UFULDSTÆNDIG. INGEN YARD GEVINST.
BOLDEN HOS CORNELL PÅ 39 - YARDLINIEN
NED MED 4 YARDS TILBAGE: 14

STRENGE SKUD. 44 YARDS SPARK.
0 YARDS TILBAGELØB.
BOLDEN HOS DARTMOUTH PÅ 17 - YARDLINIEN
FØRSTE DOWN TIL DARTMOUTH ***

VÆLG NY SPILLESTIL ?6

HOPSKUD. 45 YARDS SPARK.
BOLDEN HOS CORNELL PÅ 38 - YARDLINIEN
FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. 3 YARDS VUNDET.
BOLDEN HOS CORNELL PÅ 41 - YARDLINIEN
NED MED 2 YARDS TILBAGE: 7

DRIBLING FREMAD. 3 YARDS TABT.
BOLDEN HOS CORNELL PÅ 38 - YARDLINIEN
NED MED 3 YARDS TILBAGE: 10

LANG AFLEVERING. UFULDSTÆNDIG. INGEN YARD GEVINST.
BOLDEN HOS CORNELL PÅ 38 - YARDLINIEN
NED MED 4 YARDS TILBAGE: 10

STRENGE SKUD. 36 YARDS SPARK.
0 YARDS TILBAGELØB.
BOLDEN HOS DARTMOUTH PÅ 26 - YARDLINIEN
FØRSTE DOWN TIL DARTMOUTH ***

VÆLG NY SPILLESTIL ?4

LANG AFLEVERING. SLÆT NED. INGEN YARD GEVINST.
BOLDEN HOS DARTMOUTH PÅ 26 - YARDLINIEN
NED MED 2 YARDS TILBAGE: 10

VÆLG NY SPILLESTIL ?1

LØB FREMAD. 3 YARDS VUNDET.
BOLDEN HOS DARTMOUTH PÅ 29 - YARDLINIEN
NED MED 3 YARDS TILBAGE: 7

VÆLG NY SPILLESTIL ?7

SKUD FRA MARKEN. SKUDET ER FOR KORT.
19 YARDS RUNBACK.
BOLDEN HOS DARTMOUTH PÅ 47 - YARDLINIEN
FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. 4 YARDS VUNDET.
BOLDEN HOS DARTMOUTH PÅ 43 - YARDLINIEN
NED MED 2 YARDS TILBAGE: 6

DRIBLING FREMAD. 1 YARDS VUNDET.
 BOLDEN HOS DARTMOUTH PÅ 42 - YARDLINIEN
 NED MED 3 YARDS TILBAGE: 5

DRIBLING FREMAD. 9 YARDS VUNDET.
 BOLDEN HOS DARTMOUTH PÅ 33 - YARDLINIEN
 FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. 2 YARDS VUNDET.
 BOLDEN HOS DARTMOUTH PÅ 31 - YARDLINIEN
 NED MED 2 YARDS TILBAGE: 8

LANG AFLEVERING. ANGRIBER TACKLES. 8 YARDS TABT.
 BOLDEN HOS DARTMOUTH PÅ 39 - YARDLINIEN
 NED MED 3 YARDS TILBAGE: 16

DRIBLING FREMAD. 4 YARDS VUNDET.
 BOLDEN HOS DARTMOUTH PÅ 35 - YARDLINIEN
 NED MED 4 YARDS TILBAGE: 12

STÆRENDE SKUD. 38 YARDS SPARK.
 TOUCHBACK.
 BOLDEN HOS DARTMOUTH PÅ 20 - YARDLINIEN
 FØRSTE DOWN TIL DARTMOUTH ***

SPILLET AFBRUDT.....DER ER EN HUND PÅ BANEN.

VÆLG NY SPILLESTIL ?3

KORT AFLEVERING. SLÆT NED. INGEN YARD GEVINST.
 BOLDEN HOS DARTMOUTH PÅ 20 - YARDLINIEN
 NED MED 2 YARDS TILBAGE: 10

VÆLG NY SPILLESTIL ?7

SKUD FRA MARKEN. SKUDET ER FOR KORT.
 2 YARDS RUNBACK.
 BOLDEN HOS DARTMOUTH PÅ 46 - YARDLINIEN
 FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. 2 YARDS VUNDET.
 BOLDEN HOS DARTMOUTH PÅ 44 - YARDLINIEN
 NED MED 2 YARDS TILBAGE: 8

LANG AFLEVERING. UFULDSTÆNDIG. INGEN YARD GEVINST.
 BOLDEN HOS DARTMOUTH PÅ 44 - YARDLINIEN
 NED MED 3 YARDS TILBAGE: 8

DRIBLING FREMAD. 6 YARDS VUNDET.
 BOLDEN HOS DARTMOUTH PÅ 38 - YARDLINIEN
 NED MED 4 YARDS TILBAGE: 2

STÆRENDE SKUD. 40 YARDS SPARK.
 TOUCHBACK.
 BOLDEN HOS DARTMOUTH PÅ 20 - YARDLINIEN
 FØRSTE DOWN TIL DARTMOUTH ***

VÆLG NY SPILLESTIL ?4

LANG AFLEVERING. ER I ORDEN. 30 YARDS VUNDET.
 BOLDEN HOS DARTMOUTH PÅ 50 - YARDLINIEN
 FØRSTE DOWN TIL DARTMOUTH ***

VÆLG NY SPILLESTIL ?4

LANG AFLEVERING. SLÆT NED. INGEN YARD GEVINST.
 BOLDEN HOS DARTMOUTH PÅ 50 - YARDLINIEN
 NED MED 2 YARDS TILBAGE: 10

VÆLG NY SPILLESTIL ?6

HOPSKUD. 44 YARDS SPARK.
 BOLDEN HOS CORNELL PÅ 6 - YARDLINIEN
 FØRSTE DOWN TIL CORNELL ***

LØB FREMAD. 2 YARDS VUNDET.
 BOLDEN HOS CORNELL PÅ 8 - YARDLINIEN
 NED MED 2 YARDS TILBAGE: 8

LANG AFLEVERING. ANGRIBER TACKLES. SIKKERHED ***

RESULTAT: 2 MOD 0

CORNELL SKYDER FRA 20 - YARDLINIEN.
 52 YARDS KICKOFF.
 12 YARDS RUNBACK.
 BOLDEN HOS DARTMOUTH PÅ 40 - YARDLINIEN
 FØRSTE DOWN TIL DARTMOUTH ***

VÆLG NY SPILLESTIL ?

```

100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
110 PRINT "-----"
120 PRINT " AMERIKANSK FODBOLD"
130 PRINT "-----"
140 PRINT
150 PRINT "DETTE ER FODBOLDTURNERINGEN I DARTMOUTH."
160 PRINT
170 PRINT "DU ER HOLDLEDER FOR DARTMOUTH."
180 PRINT
190 PRINT "VÆLG MELLEML NEDENSTÆRENDE FORMER FOR SPILLESTIL:"
200 PRINT
210 PRINT "1=LØB FREMAD 2=DRIBLING FREMAD 3=KORT AFLEVERING"
220 PRINT "4=LANG AFLEVERING 5=STÆRENDE SKUD 6=HOPSKUD"
230 PRINT "7=LANGT SKUD"
240 PRINT
250 PRINT "VÆLG DIN MODSTANDER ";
260 INPUT O$(1)
270 O$(0) = "DARTMOUTH"
280 PRINT
290 S(0) = 0
300 S(1) = 0
310 DIM L$(20)
320 FOR I = 1 TO 20: READ L$(I): NEXT I
330 DATA "SPARK","MODTAG","YARDS","LØB TILBAGE EFTER "
340 DATA "BOLDEN HOS "
350 DATA "- YARDLINIEN","LØB FREMAD","DRIBLING FREMAD"
360 DATA "KORT AFLEVERING","LANG AFLEVERING","STÆRENDE SKUD"
370 DATA "HOPSKUD","SKUD FRA MARKEN","TABT"
380 DATA "INGEN YARD GEVINST","VUNDET","TOUCHDOWN","TOUCHBACK"
390 DATA "SIKKERHED ***","DUMT"
400 P = INT ( RND (1) * 2)
410 PRINT O$(P);" VANDT PLAT-OG-KRONE-SLAGNINGEN."
420 PRINT
430 DEF FN F(X) = 1 - 2 * P
440 DEF FN G(Z) = P * (X1 - X) + (1 - P) * (X - X1)
450 IF P = 0 THEN 490
460 PRINT O$(1);" VÆLGER AT MODTAGE."
470 PRINT
480 GOTO 590
490 PRINT "ØNSKER DU 'SPARK' ELLER 'MODTAG' ";
500 INPUT A$
510 PRINT
520 FOR E = 1 TO 2
530 IF A$ = L$(E) THEN 570
540 NEXT E
550 PRINT "FORKERT SVAR. INDTAST 'SPARK' ELLER 'MODTAG' ";
560 GOTO 500
570 IF E = 2 THEN 590
580 P = 1
590 X = 40 + (1 - P) * 20
600 Y = INT (200 * ( RND (1) - .5) ^ 3 + 55)
610 PRINT Y;L$(3);" KICKOFF."
620 X = X - FN F(1) * Y
630 IF ABS (X - 50) > = 50 THEN 710
640 Y = INT (50 * RND (1) ^ 2) + (1 - P) * INT (50 * RND (1) ^ 4)
650 X = X + FN F(1) * Y
660 IF ABS (X - 50) > = 50 THEN 690
670 PRINT Y;L$(3);" RUNBACK."
680 GOTO 730
690 PRINT L$(4);
700 GOTO 2620
710 PRINT "TOUCHBACK TIL ";O$(P)
720 X = 20 + P * 60
730 REM Første down
740 GOSUB 800
750 X1 = X
760 D = 1
770 PRINT "FØRSTE DOWN TIL ";O$(P);" ***"
780 PRINT
790 GOTO 860
800 REM Udskriv positionen
810 IF X > 50 THEN 840
820 PRINT L$(5);O$(0);" PÅ";X;L$(6)
830 GOTO 850
840 PRINT L$(5);O$(1);" PÅ";100 - X;L$(6)
850 RETURN
860 REM Nyt spil
870 T = T + 1
880 IF T = 30 THEN 1110
890 IF T < 50 THEN 980
900 IF RND (1) > .2 THEN 980
910 PRINT "SPILLET ER SLUT ***"
920 PRINT
930 PRINT "SLUTRESULTATET"
940 PRINT "-----"
950 PRINT
960 PRINT O$(0);":":S(0);" - ":O$(1);":":S(1)
970 GOTO 3220
980 IF P = 1 THEN 1920

```

```

990 PRINT "VÆLG NY SPILLESTIL ";
1000 INPUT Z
1010 PRINT
1020 IF Z < > INT (Z) THEN 1040
1030 IF ABS (Z - 4) < = 3 THEN 1060
1040 PRINT "ULOV LIG SPILLESTIL....PRØV IGEN ";
1050 GOTO 1000
1060 F = 0
1070 PRINT L$(Z + 6);". ";
1080 R = RND (1) * (.98 + FN F(1) * .02)
1090 R1 = RND (1)
1100 ON Z GOTO 1160,1200,1310,1530,1620,1620,1730
1110 REM Jean's special
1120 IF RND (1) > 1 / 3 THEN 980
1130 PRINT "SPILLET AFBRUDT.....DER ER EN HUND PÅ BANEN."
1140 PRINT
1150 GOTO 980
1160 REM Løb fremad
1170 Y = INT (24 * (R - .5) ^ 3 + 3)
1180 IF RND (1) < .05 THEN 1230
1190 GOTO 2240
1200 REM Drible fremad
1210 Y = INT (20 * R - 5)
1220 IF RND (1) > .1 THEN 2240
1230 F = - 1
1240 X3 = X
1250 X = X + FN F(1) * Y
1260 IF ABS (X - 50) > = 50 THEN 1290
1270 PRINT "*** BOLDEN TABT EFTER ";
1280 GOTO 2280
1290 PRINT "*** BOLDEN TABES."
1300 GOTO 2470
1310 REM Kort aflevering
1320 Y = INT (60 * (R1 - .5) ^ 3 + 10)
1330 IF R < .05 THEN 1380
1340 IF R < .15 THEN 1440
1350 IF R < .55 THEN 1470
1360 PRINT "ER I ORDEN. ";
1370 GOTO 2240
1380 IF D = 4 THEN 1470
1390 PRINT "STANDES."
1400 F = - 1
1410 X = X + FN F(1) * Y
1420 IF ABS (X - 50) > = 50 THEN 2470
1430 GOTO 2330
1440 PRINT "ANGRIBER TACKLES. ";
1450 Y = - INT (10 * R1)
1460 GOTO 2240
1470 Y = 0
1480 IF RND (1) < .3 THEN 1510
1490 PRINT "UFULDSTÆNDIG. ";
1500 GOTO 2240
1510 PRINT "SLÆT NED. ";
1520 GOTO 2240
1530 REM Lang aflevering
1540 Y = INT (160 * (R1 - .5) ^ 3 + 30)
1550 IF R < .1 THEN 1380
1560 IF R < .3 THEN 1590
1570 IF R < .75 THEN 1470
1580 GOTO 1360
1590 PRINT "ANGRIBER TACKLES. ";
1600 Y = - INT (15 * R1 + 3)
1610 GOTO 2240
1620 REM Stående skud
1630 Y = INT (100 * (R - .5) ^ 3 + 35)
1640 IF D = 4 THEN 1660
1650 Y = INT (Y * 1.3)
1660 PRINT Y;L$(3);" SPARK."
1670 IF ABS (X + Y * FN F(1) - 50) > = 50 THEN 1720
1680 IF D < 4 THEN 1720
1690 Y1 = INT (R1 ^ 2 * 20)
1700 PRINT Y1;L$(3);" TILBAGELØB."
1710 Y = Y - Y1
1720 GOTO 1400
1730 REM Langt skud
1740 Y = INT (100 * (R - .5) ^ 3 + 35)
1750 IF R1 > .15 THEN 1800
1760 PRINT "SKUDET BLOKERES ***"
1770 X = X - 5 * FN F(1)
1780 P = 1 - P
1790 GOTO 730
1800 X = X + FN F(1) * Y
1810 IF ABS (X - 50) > = 60 THEN 1860
1820 PRINT "SKUDET ER FOR KORT."
1830 IF ABS (X - 50) > = 50 THEN 2730
1840 P = 1 - P
1850 GOTO 640
1860 IF R1 > .5 THEN 1890

1870 PRINT "SKUDET GIK UD OVER SIDELINIEN."
1880 GOTO 2730
1890 PRINT "FIELD GOAL ***"
1900 S(P) = S(P) + 3
1910 GOTO 2660
1920 REM Modspillere spil
1930 IF D > 1 THEN 1990
1940 IF RND (1) > 1 / 3 THEN 1970
1950 Z = 3
1960 GOTO 1060
1970 Z = 1
1980 GOTO 1060
1990 IF D = 4 THEN 2140
2000 IF 10 + X - X1 < 5 THEN 1940
2010 IF X < 5 THEN 1940
2020 IF X < = 10 THEN 2210
2030 IF X > X1 THEN 2070
2040 A = INT (2 * RND (1))
2050 Z = 2 + A * 2
2060 GOTO 1060
2070 IF D < 3 THEN 2040
2080 IF X < 45 THEN 2040
2090 IF RND (1) > 1 / 4 THEN 2120
2100 Z = 6
2110 GOTO 1060
2120 Z = 4
2130 GOTO 1060
2140 IF X > 30 THEN 2190
2150 IF 10 + X - X1 < 3 THEN 1940
2160 IF X < 3 THEN 1940
2170 Z = 7
2180 GOTO 1060
2190 Z = 5
2200 GOTO 1060
2210 A = INT (2 * RND (1))
2220 Z = 2 + A
2230 GOTO 1060
2240 REM Vind eller tab
2250 X3 = X
2260 X = X + FN F(1) * Y
2270 IF ABS (X - 50) > = 50 THEN 2470
2280 IF Y = 0 THEN 2300
2290 PRINT ABS (Y);L$(3);" ";
2300 PRINT L$(15 + SGN (Y));". "
2310 IF ABS (X3 - 50) > 40 THEN 2330
2320 IF RND (1) < .1 THEN 2870
2330 GOSUB 800
2340 IF F = 0 THEN 2370
2350 P = 1 - P
2360 GOTO 750
2370 IF FN G(1) > = 10 THEN 750
2380 IF D = 4 THEN 2350
2390 D = D + 1
2400 PRINT "NED MED";D;" ";
2410 IF (X1 - 50) * FN F(1) < 40 THEN 2440
2420 PRINT "MANGLENDE MRL."
2430 GOTO 2450
2440 PRINT "YARDS TILBAGE:";10 - FN G(1)
2450 PRINT
2460 GOTO 860
2470 REM Bolden er i endzone
2480 IF X > = 100 THEN 2510
2490 E = 0
2500 GOTO 2520
2510 E = 1
2520 ON 1 + E - F * 2 + P * 4 GOTO 2530,2610,2780,2730,
2610,2530,2730,2780
2530 REM Sikkerhed
2540 S(1 - P) = S(1 - P) + 2
2550 PRINT L$(19)
2560 GOSUB 2820
2570 PRINT O$(P);" SKYDER FRA 20 - YARDLINIEN."
2580 X = 20 + P * 60
2590 P = 1 - P
2600 GOTO 600
2610 REM Angreb
2620 PRINT L$(17);" ***"
2630 IF RND (1) > .8 THEN 2700
2640 S(P) = S(P) + 7
2650 PRINT "DET ER ET GODT SPARK."
2660 GOSUB 2820
2670 PRINT O$(P);" KICKS OFF."
2680 P = 1 - P
2690 GOTO 590
2700 PRINT "SPARKET GÅR UD OVER SIDELINIEN."
2710 S(P) = S(P) + 6
2720 GOTO 2660
2730 REM Touchback

```

```

2740 PRINT L$(18);"."
2750 P = 1 - P
2760 X = 20 + P * 60
2770 GOTO 730
2780 REM Forsvar
2790 PRINT L$(17);"TIL ";O$(1 - P);" ***"
2800 P = 1 - P
2810 GOTO 2620
2820 REM Resultat
2830 PRINT
2840 PRINT "RESULTAT:";S(0);"MOD";S(1)
2850 PRINT
2860 RETURN
2870 REM Straf
2880 P3 = INT (2 * RND (1))
2890 PRINT O$(P3);" ER OFFSIDE --- STRAF ER 5 YARDS."
2900 PRINT
2910 IF P3 = 0 THEN 3000
2920 PRINT "ACCEPTERES DOMMEN ";
2930 INPUT A$
2940 PRINT
2950 IF LEFT$(A$,1) = "N" THEN 2330
2960 IF LEFT$(A$,1) = "J" THEN 3150
2970 PRINT "TAST 'J' ELLER 'N'"
2980 PRINT
2990 GOTO 2920
3000 REM Modstanders straffe-strategi
3010 IF P = 1 THEN 3060
3020 IF Y < = 0 THEN 3100
3030 IF F < 0 THEN 3100
3040 IF FN G(1) < 3 * D - 2 THEN 3100
3050 GOTO 3130
3060 IF Y < = 5 THEN 3130
3070 IF F < 0 THEN 3130
3080 IF D < 4 THEN 3100
3090 IF FN G(1) < 10 THEN 3130
3100 PRINT "DOMMEN AFVIST."
3110 PRINT
3120 GOTO 2330
3130 PRINT "DOMMEN ACCEPTERES."
3140 PRINT
3150 F = 0
3160 D = D - 1
3170 IF P < > P3 THEN 3200
3180 X = X3 - FN F(1) * 5
3190 GOTO 2330
3200 X = X3 + FN F(1) * 5
3210 GOTO 2330
3220 END

```

MORRISTOWN CREATIVE COMPUTING NEW JERSEY

AMERIKANSK FODBOLD, VERSION 2

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DETTE ER ET SPIL FOR TO HOLD, HVOR HVER AF DE TO SPILLERE SKAL FORBEREDE EN RÆKKE DATA (I LINIENUMMER 160 FOR HOLD 1, OG I LINIENUMMER 170 FOR HOLD 2). I DISSE DATA BLANDER SPILLERNE TAL FRA 1 TIL 20. DISSE TAL TILDELES DEREFTER FORSKELLIGE SPILLEMRÅDER, DER DELES UD TIL DE TO HOLD.

JO MERE ENS DE TO HOLDS SPIL ER, JO MINDRE VINDER HOLDENE MED. STILLINGEN UDSKRIVES NÅR DER SCORES, MEN DU KAN OGSÅ FRÅ UDSKREVET STILLINGEN VED AT INDTASTE '99,99' SOM SPILLENUMMER. HVIS DU ØNSKER AT AFLEVERE ELLER FORSØGE ET LANGT SKUD MOD MÅL, KAN DU INDTASTE '77,77' SOM SPILLENUMMER. DU VIL I SÅ FALD SKULLE SVARE PÅ NOGLE SPØRGSMÅL.

HVIS SVARENE PÅ BEGGE SPØRGSMÅL ER NEJ, GÅS DER UD FRA AT DU ØNSKER AT ANGRIBE.

ALLE SPØGSMÅL SKAL BESVARES MED 'J' (JA), OG 'N' (NEJ).

SPILLET FORTSÆTTER INDTIL DET AFBRYDES MED CONTROL-C.

GØR NU EN TALRÆKKE KLAR TIL PROGRAMLINIERNE 160 OG 170.

INDTAST MAKSIMALE ANTAL POINT FOR SPILLET ?28

HOLD NUMMER 1 SPILLEMETODELISTE

```

-----
NR  SPILLEMETODE
-----
17  PITCHOUT
8   TREDOBBELT BAGUD
4   UAFGJORT
14  QB SNEAK
19  BAGOM OG RUNDT
3   DOBBELT BAGUD
10  ANGREB TIL VENSTRE
1   ANGREB TIL HØJRE
7   TACKLING
11  TILFÆLDIG
15  FLARE PASS
9   SCREEN PASS
5   ROLL OUT
20  HØJRE CURL
13  VENSTRE CURL
18  TILFÆLDIG
16  SIDELINIEAFLEVERING
2   HALFBACK
12  RAZZLE-DAZZLE
6   BOMBE!!

```

RIV HER-----

HOLD NUMMER 2 SPILLEMETODELISTE

```

-----
NR  SPILLEMETODE
-----
20  PITCHOUT
2   TREDOBBELT BAGUD
17  UAFGJORT
5   QB SNEAK
8   BAGOM OG RUNDT
18  DOBBELT BAGUD
12  ANGREB TIL VENSTRE
11  ANGREB TIL HØJRE
1   TACKLING
4   TILFÆLDIG
19  FLARE PASS
14  SCREEN PASS
10  ROLL OUT
7   HØJRE CURL
9   VENSTRE CURL
15  TILFÆLDIG
6   SIDELINIEAFLEVERING
13  HALFBACK
16  RAZZLE-DAZZLE
3   BOMBE!!

```

RIV HER-----

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

HOLD 1 FORSVARER 0-YARD MÅLET.
HOLD 2 FORSVARER 100-YARD MÅLET.

DER KASTES PLAT-ELLER-KRONE.

+++++

HOLD 1 FÅR KICKOFF.
SKUDET VAR PÅ 47 YARDS.
BOLDEN ER NU VED 13 - YARDLINIEN.

--->

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

ØNSKER HOLD 1 TILBAGELØB ?J

RUNBACK FOR HOLD 1 ER 1 YARDS.

=====

HOLD 1 ER 1 DOWN, NU PÅ 14 - YARDLINIEN.

10 YARDS TIL FØRSTE DOWN.

--->

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?13,4

KVARTERBACKEN STOPPET.

NETTOGEVINST VED DOWN 1 ER -4 YARDS.

HOLD 1 ER 2 DOWN, NU PÅ 10 - YARDLINIEN.

14 YARDS TIL FØRSTE DOWN.

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?14,2

BOLDEN HOPPER.

NETTOGEVINST VED DOWN 2 ER -2 YARDS.

HOLD 1 ER 3 DOWN, NU PÅ 8 - YARDLINIEN.

16 YARDS TIL FØRSTE DOWN.

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?20,9

KVARTERBACKEN STOPPET.

NETTOGEVINST VED DOWN 3 ER 3 YARDS.

HOLD 1 ER 4 DOWN, NU PÅ 11 - YARDLINIEN.

13 YARDS TIL FØRSTE DOWN.

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

ØNSKER HOLD 1 PUNT ?N

ØNSKER HOLD 1 AT FORSØGE FIELDGOAL ?N

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?12,3

KVARTERBACKEN STOPPET.

NETTOGEVINST VED DOWN 4 ER 1 YARDS.

FORVANDLING MISLYKKES FOR HOLD 1

HOLD 2 ER 1 DOWN, NU PÅ 12 - YARDLINIEN.

10 YARDS TIL FØRSTE DOWN.

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?3,7

MISLYKKET AFLEVERING FOR HOLD 2 .

NETTOGEVINST VED DOWN 1 ER 0 YARDS.

HOLD 2 ER 2 DOWN, NU PÅ 12 - YARDLINIEN.

10 YARDS TIL FØRSTE DOWN.

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?77,77

ØNSKER HOLD 2 PUNT ?J

HOLD 2 FÅR PUNT.

BOLDEN GIK UDENFOR BANEN.

-- AUTOMATISK TOUCHBACK --

HOLD 1 ER 1 DOWN, NU PÅ 20 - YARDLINIEN.

10 YARDS TIL FØRSTE DOWN.

HOLD 1 (0 10 20 30 40 50 60 70 80 90 100) HOLD 2

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?99,99

HOLD 1 HAR 0 POINT.

HOLD 2 HAR 0 POINT.

INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ?

```

100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
110 PRINT "-----"
120 PRINT " AMERIKANSK FODBOLD, VERSION 2"
130 PRINT "-----"
140 PRINT
150 REM DATA-sætninger med koder (kan ændres)
160 DATA 17,8,4,14,19,3,10,1,7,11,15,9,5,20,13,18,16,2,12,6
170 DATA 20,2,17,5,8,18,12,11,1,4,19,14,10,7,9,15,6,13,16,3
180 REM
190 DIM A(20),B(20),C(40),H(2),T(2),W(2),X(2),Y(2),Z(2)
200 DIM M$(2),D(2),P$(20)
210 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
220 INPUT A$
230 PRINT
240 IF LEFT$(A$,1) = "N" THEN 470
250 IF LEFT$(A$,1) < > "J" THEN 210
260 PRINT "DETTE ER ET SPIL FOR TO HOLD, HVOR HVER AF DE TO SPILLERE"
270 PRINT "SKAL FORBEREDE EN RÆKKE DATA (I LINIENUMMER 160 FOR HOLD"
280 PRINT "1, OG I LINIENUMMER 170 FOR HOLD 2). I DISSE DATA BLANDER"
290 PRINT "SPILLERNE TAL FRA 1 TIL 20. DISSE TAL TILDELES DEREFTER"
300 PRINT "FORSKELLIGE SPILLEMÅDER, DER DELES UD TIL DE TO HOLD."
310 PRINT
320 PRINT "JO MERE ENS DE TO HOLDS SPIL ER, JO MINDRE VINDER HOLDENE"
330 PRINT "MED. STILLINGEN UDSKRIVES NÅR DER SCORES, MEN DU KAN OGSÅ"
340 PRINT "FÅ UDSKREVET STILLINGEN VED AT INDASTE '99,99' SOM"
350 PRINT "SPILLENUMMER. HVIS DU ØNSKER AT AFLEVERE ELLER FORSØGE"
360 PRINT "ET LANGT SKUD MOD MÅL, KAN DU INDASTE '77,77' SOM SPILLE-"
370 PRINT "NUMMER. DU VIL I SÅ FALD SKULLE SVARE PÅ NOGLE SPØRGSMALE."
380 PRINT
390 PRINT "HVIS SVARENE PÅ BEGGE SPØRGSMALE ER NEJ, GÅS DER UD FRA AT"
400 PRINT "DU ØNSKER AT ANGRIBE."
410 PRINT
420 PRINT "ALLE SPØRGSMALE SKAL BESVARES MED 'J' (JA), OG 'N' (NEJ)."

```

```

760 PRINT
770 PRINT "RIV HER-----"
780 FOR X = 1 TO 11
790 PRINT
800 NEXT X
810 FOR Z = 1 TO 3000
820 NEXT Z
830 IF L = 20 THEN 650
840 D(1) = 0
850 D(2) = 3
860 M$(1) = "---->"
870 M$(2) = "<----"
880 H(1) = 0:H(2) = 0:T(1) = 2:T(2) = 1
890 W(1) = - 1:W(2) = 1:X(1) = 100:X(2) = 0
900 Y(1) = 1:Y(2) = - 1:Z(1) = 0:Z(2) = 100
910 GOSUB 2950
920 PRINT "HOLD 1 FORSVARER 0-YARD MÅLET."
930 PRINT "HOLD 2 FORSVARER 100-YARD MÅLET."
940 T = INT (2 * RND (1) + 1)
950 PRINT
960 PRINT "DER KASTES PLAT-ELLER-KRONE."
970 PRINT
980 P = X(T) - Y(T) * 40
990 GOSUB 2850
1000 PRINT
1010 PRINT "HOLD";T;"FØR KICKOFF."
1020 K = INT (26 * RND (1) + 40)
1030 P = P - Y(T) * K
1040 IF W(T) * P < Z(T) + 10 THEN 1090
1050 PRINT
1060 PRINT "BOLDEN GIK UDEFOR BANEN."
1070 PRINT "-- AUTOMATISK TOUCHBACK --"
1080 GOTO 1170
1090 PRINT "SKUDET VAR PÅ";K;"YARDS."
1100 PRINT "BOLDEN ER NU VED";P;"- YARDLINIEN."
1110 GOSUB 2900
1120 PRINT "ØNSKER HOLD";T;"TILBAGELØB ";
1130 INPUT A$
1140 IF LEFT$(A$,1) = "J" THEN 2330
1150 IF LEFT$(A$,1) < > "N" THEN 1120
1160 IF W(T) * P < Z(T) THEN 1180
1170 P = Z(T) - W(T) * 20
1180 D = 1
1190 S = P
1200 PRINT
1210 FOR I = 1 TO 72
1220 PRINT "=";
1230 NEXT I
1240 PRINT
1250 PRINT
1260 PRINT "HOLD";T;"ER";D;"DOWN, NU PÅ";P;" - YARDLINIEN."
1270 IF D < > 1 THEN 1320
1280 IF Y(T) * (P + Y(T) * 10) > = X(T) THEN 1310
1290 C = 4
1300 GOTO 1320
1310 C = 8
1320 IF C = 8 THEN 1370
1330 PRINT
1340 PRINT " ";10 - (Y(T) * P - Y(T) * S);
1350 PRINT "YARDS TIL FØRSTE DOWN."
1360 GOTO 1390
1370 PRINT
1380 PRINT " ";X(T) - Y(T) * P;"YARDS."
1390 GOSUB 2900
1400 IF D = 4 THEN 1950
1410 U = INT (3 * RND (1) - 1)
1420 GOTO 1440
1430 PRINT "ULOVLIGT SPILNUMMER, SE EFTER OG"
1440 PRINT "INDTAST OFFENSIVT SPIL, DEFENSIVT SPIL ";
1450 IF T = 2 THEN 1480
1460 INPUT P1,P2
1470 GOTO 1490
1480 INPUT P2,P1
1490 IF P1 = 77 THEN 1950
1500 IF P1 > 20 THEN 2740
1510 IF P1 < 1 THEN 2740
1520 IF P2 > 20 THEN 2740
1530 IF P2 < 1 THEN 2740
1540 P1 = INT (P1)
1550 P2 = INT (P2)
1560 Y = INT ( ABS (A(P1) - B(P2)) / 19 * ((X(T) - Y(T) * P + 25) * RND
(1) - 15))
1570 PRINT
1580 IF T = 2 THEN 1610
1590 IF A(P1) < 11 THEN 1730
1600 GOTO 1620
1610 IF B(P2) < 11 THEN 1730
1620 IF U < > 0 THEN 1660
1630 PRINT "MISLYKKET AFLEVERING FOR HOLD";T;"."
1640 Y = 0
1650 GOTO 1740
1660 G = RND (1)
1670 IF G > .025 THEN 1690
1680 IF Y > 2 THEN 1710
1690 PRINT "KVARTERBACKEN STOPPET."
1700 GOTO 1740
1710 PRINT "AFLEVERING I ORDEN."
1720 GOTO 1740
1730 PRINT "BOLDEN HOPPER."
1740 P = P - W(T) * Y
1750 PRINT
1760 PRINT "NETTOGEVINST VED DOWN";D;"ER";Y;"YARDS."
1770 G = RND (1)
1780 IF G > .025 THEN 1850
1790 PRINT
1800 PRINT "** HOLD";T;"MISTER BOLDEN TIL HOLD";T(T);"."
1810 GOSUB 2840
1820 PRINT
1830 T = T(T)
1840 GOTO 1120
1850 IF Y(T) * P > = X(T) THEN 2200
1860 IF W(T) * P > = Z(T) THEN 2120
1870 IF Y(T) * P - Y(T) * S > = 10 THEN 1180
1880 D = D + 1
1890 IF D < > 5 THEN 1200
1900 PRINT
1910 PRINT "FORVANDLING MISLYKKES FOR HOLD";T
1920 T = T(T)
1930 GOSUB 2840
1940 GOTO 1180
1950 PRINT "ØNSKER HOLD";T;"PUNT ";
1960 INPUT A$
1970 IF LEFT$(A$,1) = "N" THEN 2070
1980 IF LEFT$(A$,1) < > "J" THEN 1950
1990 PRINT
2000 PRINT "HOLD";T;"FØR PUNT."
2010 G = RND (1)
2020 IF G < .025 THEN 1790
2030 GOSUB 2840
2040 K = INT (25 * RND (1) + 35)
2050 T = T(T)
2060 GOTO 1030
2070 PRINT "ØNSKER HOLD";T;"AT FØRSØGE FIELDGOAL ";
2080 INPUT A$
2090 IF LEFT$(A$,1) = "J" THEN 2430
2100 IF LEFT$(A$,1) < > "N" THEN 2070
2110 GOTO 1410
2120 PRINT
2130 PRINT "SIKKERHED MOD HOLD";T;"*****HOV-HOV"
2140 H(T(T)) = H(T(T)) + 2
2150 GOSUB 2750
2160 PRINT "ØNSKER HOLD";T;"PUNT I STEDET FOR MÅLSPARK ";
2170 INPUT A$
2180 P = Z(T) - W(T) * 20
2190 IF LEFT$(A$,1) = "J" THEN 1990
2200 PRINT
2210 PRINT "TOUCHDOWN TIL HOLD";T;"*****YEAH!"
2220 Q = 7
2230 G = RND (1)
2240 IF G > .1 THEN 2280
2250 Q = 6
2260 PRINT "EKSTRA POINT GÆLDER IKKE."
2270 GOTO 2290
2280 PRINT "EKSTRA POINT GÆLDER."
2290 H(T) = H(T) + Q
2300 GOSUB 2750
2310 T = T(T)
2320 GOTO 980
2330 K = INT (9 * RND (1) + 1)
2340 R = INT (((X(T) - Y(T) * P + 25) * RND (1) - 15) / K)
2350 P = P - W(T) * R
2360 PRINT
2370 PRINT "RUNBACK FOR HOLD";T;"ER";R;"YARDS."
2380 G = RND (1)
2390 IF G < .025 THEN 1790
2400 IF Y(T) * P > = X(T) THEN 2200
2410 IF W(T) * P > = Z(T) THEN 2120
2420 GOTO 1180
2430 PRINT
2440 PRINT "HOLD";T;"FØRSØGER FIELDGOAL."
2450 G = RND (1)
2460 IF G < .025 THEN 1790
2470 F = INT (35 * RND (1) + 20)
2480 PRINT
2490 PRINT "SKUDET ER";F;"YARDS LANGT."
2500 P = P - W(T) * F


```




```

2510 G = RND (1)
2520 IF G < .35 THEN 2570
2530 IF Y(T) * P < X(T) THEN 2580
2540 PRINT "HOLD";T;"'S FIELDGOAL LYKKEDES *****YEAH"
2550 Q = 3
2560 GOTO 2290
2570 PRINT "BOLDEN GIK FOR LANGT."
2580 PRINT "MISLYKKET FIELDGOAL FOR HOLD";T;"-----ØV"
2590 GOSUB 2840
2600 IF Y(T) * P < X(T) + 10 THEN 2630
2610 T = T(T)
2620 GOTO 1040
2630 PRINT
2640 PRINT "BOLDEN ER NU PR";P;"YARDLINIEN."
2650 T = T(T)
2660 GOSUB 2900
2670 GOTO 1120
2680 DATA "PITCHOUT","TREDOBBELT BAGUD","UAFGJORT","QB SNEAK"
2690 DATA "BAGOM OG RUNDT","DOBBELT BAGUD","ANGREB TIL VENSTRE"
2700 DATA "ANGREB TIL HØJRE","TACKLING","TILFÆLDIG"
2710 DATA "FLARE PASS","SCREEN PASS","ROLL OUT","HØJRE CURL"
2720 DATA "VENSTRE CURL","TILFÆLDIG","SIDELINIEAFLEVERING"
2730 DATA "HALFBACK","RAZZLE-DAZZLE","BOMBE!!"
2740 IF P1 < > 99 THEN 1430
2750 PRINT
2760 PRINT "HOLD 1 HAR";H(1);"POINT."
2770 PRINT "HOLD 2 HAR";H(2);"POINT."
2780 PRINT
2790 IF H(T) < E THEN 2820
2800 PRINT "HOLD";T;"VINDER*****"
2810 GOTO 2990
2820 IF P1 = 99 THEN 1440
2830 RETURN
2840 PRINT
2850 FOR X = 1 TO 72
2860 PRINT "+";
2870 NEXT X
2880 PRINT
2890 RETURN
2900 PRINT
2910 FOR J = 1 TO D(T) + 4 + P / 2
2920 PRINT " ";
2930 NEXT J
2940 PRINT M$(T)
2950 PRINT "HOLD 1 (0 10 20 30 40 50 60 70 80";
2960 PRINT " 90 100) HOLD 2"
2970 PRINT
2980 RETURN
2990 END

```



Awari



Awari er et gammelt afrikansk spil, der spilles med 7 pinde og 36 sten eller bønner, der er fordelt som vist ovenfor. Brættet er ind delt i 6 afdelinger eller gruber langs hver side. Yderligere er der to særlige depot-gruber ved brættets ender.

Et træk foretages ved at tage alle bønnerne fra en af (de ikke tomme) gruber på din egen side, og idet du starter fra gruben til højre for denne, 'sår' du en af bønnerne i hver af gruberne, idet du går imod uret.

Hver tur består af et eller to træk. Hvis den sidste bønne i dit træk sås i dit eget depot, så får du et ekstra træk.

Hvis den sidste bønne, der sås i et træk, lander i en tom grube, så 'plyndres' den overfor liggende grube for hele sit indhold af bønner (hvis der er nogen), og disse lægges i spillerens depot.

Når der ikke er flere bønner i en af sidernes gruber, så er spillet slut. Den spiller, der har flest bønner i sit depot, har vundet.

I nærværende datamatudgave af spillet udskrives spillebrættet som 14 numre, der hver repræsenterer de 14 gruber.

0	3	3	3	3	3	3	0
	3	3	3	3	3	3	

Gruberne på din side (den nedre) er nummereret fra venstre mod højre med tallene fra 1 til 6. Gruberne på min (datamatens) side er på samme måde nummereret fra min venstre side (dvs. din højre).

Når du skal foretage et træk, indtaster du nummeret på en grube. Hvis den sidste bønne lander i dit eget depot, skriver datamaten »DU HAR ET TRÆK TIL?«, hvorefter du kan indtaste endnu et træk.

Datamatens træk udskrives automatisk og efterfølges af diagrammet over spillebrættet. Datamaten tilbyder dig altid det første træk. Det anses for at være en fordel at trække først.

Der er en »læremekanisme« i spillet. Derfor vil datamaten spille bedre og bedre, jo flere gange der spilles.

Denne version af Awari er en bearbejdning af et spil, der oprindeligt blev skrevet af Geoff Wyvill fra Bradford, Yorkshire, England.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

AWARI

0	3	3	3	3	3	3	0
	3	3	3	3	3	3	

DIT TRÆK ?5

0	3	3	3	3	3	4	1
	3	3	3	3	0	4	

MIT TRÆK =2

0	3	4	4	4	0	4	1
	3	3	3	3	0	4	

DIT TRÆK ?4

0	3	4	4	4	0	4	2
	3	3	3	0	1	5	

DU HAR ET TRÆK TIL ?1

0	3	4	4	0	0	4	7
	0	4	4	0	1	5	

MIT TRÆK =4

6	0	5	0	0	0	4	7
	0	4	4	0	1	5	

DIT TRÆK ?3

6	0	5	0	0	0	4	8
	0	4	0	1	2	6	

DU HAR ET TRÆK TIL ?5

6	0	5	0	0	0	4	9
	0	4	0	1	0	7	

MIT TRÆK =5

7	1	0	0	0	0	4	9
	1	5	1	1	0	7	

DIT TRÆK ?2

7	1	0	0	0	0	4	10
	1	0	2	2	1	8	

DU HAR ET TRÆK TIL ?3

7	1	0	0	0	0	4	10
	1	0	0	3	2	8	

MIT TRÆK =6,1

8	0	1	1	1	1	0	10
	1	0	0	3	2	8	

DIT TRÆK ?5

8	0	1	1	1	1	0	11
	1	0	0	3	0	9	

DU HAR ET TRÆK TIL ?4

8	0	1	1	1	1	0	12
	1	0	0	0	1	10	

MIT TRÆK =5

10	0	0	1	1	1	0	12
	0	0	0	0	1	10	

DIT TRÆK ?6

11	1	0	2	2	2	1	15
	1	0	0	0	1	0	

```

MIT TRÆK =6,4
    0  1  0  2  2  1
14   0  0  0  0  1  0
DIT TRÆK ?
    0  1  0  2  2  0
14   0  0  0  0  0  0

*****
SPILLET ER SLUT
*****

DU VINDER MED 3 POINT.

    3  3  3  3  3  3
0    0  3  3  3  3  3  0
DIT TRÆK ?

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                AWARI"
130 PRINT "-----"
140 PRINT
150 DATA 0
160 DIM B(13),G(13),F(50)
170 READ N
180 E = 0
190 FOR I = 0 TO 12
200 B(I) = 3
210 NEXT I
220 C = 0
230 F(N) = 0
240 B(13) = 0
250 B(6) = 0
260 GOSUB 740
270 PRINT "DIT TRÆK ";
280 GOSUB 510
290 IF E = 0 THEN 380
300 IF M = H THEN GOSUB 500
310 IF E = 0 THEN 380
320 PRINT "MIT TRÆK =";
330 GOSUB 1080
340 IF E = 0 THEN 380
350 IF M = H THEN PRINT ",";: GOSUB 1080
360 PRINT
370 IF E > 0 THEN 260
380 PRINT
390 PRINT "                *****"
400 PRINT "                SPILLET ER SLUT"
410 PRINT "                *****"
420 PRINT
430 D = B(6) - B(13)
440 IF D < 0 THEN PRINT "JEG VINDER MED";(- 1) *
D;"POINT.": PRINT : GOTO 180
450 N = N + 1
460 IF D = 0 THEN PRINT "SPILLET ENDTE UAFGJORT!":
PRINT : GOTO 180
470 PRINT "DU VINDER MED";D;"POINT."
480 PRINT
490 GOTO 180
500 PRINT "DU HAR ET TRÆK TIL ";
510 INPUT M
520 IF M < 7 THEN IF M > 0 THEN M = M - 1: GOTO 570
530 PRINT
540 PRINT "!!!!!! UGYLDIGT TRÆK !!!!!"
550 PRINT
560 GOTO 500
570 IF B(M) = 0 THEN 530
580 H = 6
590 GOSUB 610

```

```

600 GOTO 740
610 K = M
620 GOSUB 950
630 E = 0
640 IF K > 6 THEN K = K - 7
650 C = C + 1
660 IF C < 9 THEN F(N) = F(N) * 6 + K
670 FOR I = 0 TO 5
680 IF B(I) < > 0 THEN 710
690 NEXT I
700 RETURN
710 FOR I = 7 TO 12
720 IF B(I) < > 0 THEN E = 1: RETURN
730 GOTO 690
740 PRINT
750 PRINT "                ";
760 FOR I = 12 TO 7 STEP - 1
770 GOSUB 920
780 NEXT I
790 PRINT
800 I = 13
810 PRINT "                ";
820 GOSUB 920
830 PRINT "                ";
840 PRINT B(6)
850 PRINT "                ";
860 FOR I = 0 TO 5
870 GOSUB 920
880 NEXT I
890 PRINT
900 PRINT
910 RETURN
920 IF B(I) < 10 THEN PRINT " ";
930 PRINT B(I);
940 RETURN
950 P = B(M)
960 B(M) = 0
970 FOR P = P TO 1 STEP - 1
980 M = M + 1
990 IF M > 13 THEN M = M - 14
1000 B(M) = B(M) + 1
1010 NEXT P
1020 IF B(M) = 1 THEN IF M < > 6 THEN IF M < >
13 THEN IF B(12 - M) < > 0 THEN 1040
1030 RETURN
1040 B(H) = B(H) + B(12 - M) + 1
1050 B(M) = 0
1060 B(12 - M) = 0
1070 RETURN
1080 D = - 99
1090 H = 13
1100 FOR I = 0 TO 13
1110 G(I) = B(I)
1120 NEXT I
1130 FOR J = 7 TO 12
1140 IF B(J) = 0 THEN 1370
1150 G = 0
1160 M = J
1170 GOSUB 950
1180 FOR I = 0 TO 5
1190 IF B(I) = 0 THEN 1250
1200 L = B(I) + I
1210 R = 0
1220 IF L > 13 THEN L = L - 14:R = 1: GOTO 1220
1230 IF B(L) = 0 THEN IF L < > 6 THEN IF L < >
13 THEN R = B(12 - L) + R
1240 IF R > Q THEN Q = R
1250 NEXT I
1260 Q = B(13) - B(6) - Q
1270 IF C > 8 THEN 1330
1280 K = J
1290 IF K > 6 THEN K = K - 7
1300 FOR I = 0 TO N - 1
1310 IF F(N) * 6 + K = INT (F(I) / 6 ^ (7 - C) +
.1) THEN Q = Q - 2
1320 NEXT I
1330 FOR I = 0 TO 13
1340 B(I) = G(I)
1350 NEXT I
1360 IF Q > = D THEN A = J:D = Q
1370 NEXT J
1380 M = A
1390 PRINT CHR$( 42 + M);
1400 GOTO 610
1410 FOR I = 0 TO N - 1
1420 PRINT B(I)
1430 NEXT I
1440 END

```

Bagels

I dette spil vælger datamaten et tilfældigt 3-cifret tal, der kan indeholde cifre fra 0 til 9. Du skal forsøge at gætte, hvilket tal datamaten har valgt. Du får op til 20 gæt. I de tal, som datamaten har valgt, forekommer et ciffer kun én gang. Efter hvert gæt vil datamaten give et fingerpeg om dit gæt på følgende måde:

PICO: Et af cifrene er korrekte, men på forkert plads

FERMI: Et af cifrene er rigtigt og på rigtig plads

BAGELS: Ingen af cifrene er rigtige

Du vil meget hurtigt lære at drage nytte af disse fingerpeg, og med øvelse vil du lære at forbedre dine resultater. Der findes flere gode strategier, man kan anvende til at spille Bagels. Når du har fundet en god strategi, så prøv om du ikke kan forbedre den. Eller prøv f.eks. en helt anden strategi, og se om den fungerer bedre. Ganske vist tillader programmet helt op til 20 gæt, men det bør ikke tage dig mere end 8 gæt at finde datamatens 3-cifrede tal.

De oprindelige forfattere af dette program er D. Resek og P. Rowe fra Lawrence Hall of Science, Berkeley, Californien.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BAGELS

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

JEG TÆNKER PÅ ET TRE-CIFRET TAL, OG DU SKAL PRØVE PÅ AT GÆTTE DETTE TAL. JEG HJÆLPER DIG VED AT SKRIVE ET AF NEDENSTÅENDE NØGLEORD PÅ SKÆRMEN, HVER GANG DU HAR GÆTTET:

PICO - ET CIFFER ER RIGTIGT,
MEN DET ER PÅ FORKERT PLADS.
FERMI - ET CIFFER ER RIGTIGT,
OG PÅ RIGTIG PLADS.
BAGELS - INGEN AF CIFRENE ER RIGTIGE.

DU FÅR 20 FORSØG TIL AT GÆTTE TALLET!

O.K., JEG TÆNKER PÅ ET TAL NU.

GÆT NUMMER 1 ?123

FERMI

GÆT NUMMER 2 ?145

FERMI

GÆT NUMMER 3 ?678

PICO

GÆT NUMMER 4 ?192

PICO

GÆT NUMMER 5 ?152

PICO PICO

GÆT NUMMER 6 ?572

PICO PICO PICO

GÆT NUMMER 7 ?725

FANTASTISK! DU HAR GÆTTET DET!!!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?J

O.K., JEG TÆNKER PÅ ET TAL NU.

GÆT NUMMER 1 ?123

BAGELS

GÆT NUMMER 2 ?456

BAGELS

GÆT NUMMER 3 ?789

PICO FERMI

GÆT NUMMER 4 ?708

PICO FERMI

GÆT NUMMER 5 ?798

PICO PICO

GÆT NUMMER 6 ?908

PICO PICO FERMI

GÆT NUMMER 7 ?809

FANTASTISK! DU HAR GÆTTET DET!!!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

TAK FOR SPILLET. DET VAR HYGGELIGT.

JEG HÅBER DU HAR MORET DIG,
NU ER DU BAGELS-SPECIALIST!
MED IKKE MINDRE END 2 POINT!

FARVEL MED DIG.

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                BAGELS"
130 PRINT "-----"
140 REM
150 REM Bagels spillets oprindelse er usikker, men
160 REM man regner med, at spillet stammer fra
170 REM Lawrence Hall of Science, U.C. Berkely
180 REM
190 DIM A1(6),A(3),B(3)
200 Y = 0
210 T = 255
220 PRINT
230 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
240 INPUT A$
250 IF LEFT$(A$,1) = "N" THEN 400
260 PRINT
270 PRINT "JEG TÆNKER PÅ ET TRE-CIFRET TAL, OG DU"
280 PRINT "SKAL PRØVE PÅ AT GÆTTE DETTE TAL."
290 PRINT "JEG HJÆLPER DIG VED AT SKRIVE ET AF"
300 PRINT "NEDENSTÅENDE NØGLEORD PÅ SKÆRMEN, HVER"
310 PRINT "GANG DU HAR GÆTTET:"
320 PRINT
330 PRINT " PICO - ET CIFFER ER RIGTIGT,"
340 PRINT "      MEN DET ER PÅ FORKERT PLADS."
350 PRINT " FERMI - ET CIFFER ER RIGTIGT,"
360 PRINT "      OG PÅ RIGTIG PLADS."
370 PRINT " BAGELS - INGEN AF CIFRENE ER RIGTIGE."
380 PRINT
390 PRINT "DU FÅR 20 FORSØG TIL AT GÆTTE TALLET!"
400 FOR I = 1 TO 3
410 A(I) = INT (10 * RND (1))
420 IF I - 1 = 0 THEN 460
430 FOR J = 1 TO I - 1
440 IF A(I) = A(J) THEN 410
450 NEXT J
460 NEXT I
470 PRINT
480 PRINT "O.K., JEG TÆNKER PÅ ET TAL NU."
490 PRINT
500 FOR I = 1 TO 20
510 PRINT "GÆT NUMMER";I;
520 INPUT A$
530 PRINT
540 IF LEN (A$) < > 3 THEN 1060
550 FOR Z = 1 TO 3
560 A1(Z) = ASC ( MID$( A$,Z,1))
```

```

570 NEXT Z
580 FOR J = 1 TO 3
590 IF A1(J) < 48 THEN 640
600 IF A1(J) > 57 THEN 640
610 B(J) = A1(J) - 48
620 NEXT J
630 GOTO 670
640 PRINT "HVA' BA' ???"
650 PRINT
660 GOTO 510
670 IF B(1) = B(2) THEN 1090
680 IF B(2) = B(3) THEN 1090
690 IF B(3) = B(1) THEN 1090
700 C = 0
710 D = 0
720 FOR J = 1 TO 2
730 IF A(J) < > B(J + 1) THEN 750
740 C = C + 1
750 IF A(J + 1) < > B(J) THEN 770
760 C = C + 1
770 NEXT J
780 IF A(1) < > B(3) THEN 800
790 C = C + 1
800 IF A(3) < > B(1) THEN 820
810 C = C + 1
820 FOR J = 1 TO 3
830 IF A(J) < > B(J) THEN 850
840 D = D + 1
850 NEXT J
860 IF D = 3 THEN 1130
870 IF C = 0 THEN 910
880 FOR J = 1 TO C
890 PRINT "PICO ";
900 NEXT J
910 IF D = 0 THEN 950
920 FOR J = 1 TO D

```

```

930 PRINT "FERMI ";
940 NEXT J
950 IF C + D < > 0 THEN 970
960 PRINT "BAGELS ";
970 PRINT
980 PRINT
990 NEXT I
1000 PRINT "NR....."
1010 PRINT "DET VAR TYVE GÆT....."
1020 PRINT "DET TAL JEG TÆNKTE PÅ VAR: ";
1030 PRINT 100 * A(1) + 10 * A(2) + A(3)
1040 PRINT
1050 GOTO 1160
1060 PRINT "DET ER ET TRE-CIFRET TAL DU SKAL GÆTTE!"
1070 PRINT
1080 GOTO 510
1090 PRINT "ÅH, JEG GLEMTE AT SIGE, AT DE TAL JEG"
1100 PRINT "TÆNKER PÅ, ALDRIG HAR TO ENS CIFRE."
1110 PRINT
1120 GOTO 510
1130 PRINT "FANTASTISK! DU HAR GÆTTET DET!!!"
1140 PRINT
1150 Y = Y + 1
1160 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1170 INPUT AS
1180 IF LEFT$(AS,1) = "J" THEN 400
1190 PRINT
1200 PRINT "TAK FOR SPILLET. DET VAR HYGGELIGT."
1210 PRINT
1220 IF Y < = 0 THEN GOTO 1270
1230 PRINT "JEG HÅBER DU HAR MORET DIG,"
1240 PRINT "NU ER DU BAGELS-SPECIALIST!"
1250 PRINT "MED IKKE MINDRE END";Y;"POINT!"
1260 PRINT
1270 PRINT "FARVEL MED DIG."
1280 END

```


Baglæns

I spillet Baglæns skal du flytte om på cifrene i en talrække, så disse kommer til at stå i nummerorden, målt fra venstre mod højre. For at gøre dette skal du fortælle datamaten, hvor mange cifre (talt fra venstre) du ønsker flyttet om på. Hvis der f.eks. på rækken står følgende:

2 3 4 5 1 6 7 8 9

og du beder om at flytte fire cifre, bliver resultatet:

5 4 3 2 1 6 7 8 9

Hvis du nu flytter fem cifre, har du vundet!

Der findes mange forskellige måder at løse problemet på, men de er alle enten algoritmiske eller heuristiske (dvs. enten matematiske eller »prøven-sig-frem« metoder). Spillet giver altså mulighed for at anvende disse principper i en praktisk (i stedet for teoretisk) sammenhæng.

En algoritmisk løsning garanterer en løsning inden for et forudsigeligt antal træk, når man kender antallet af cifre i talrækken. Der findes f.eks. en metode, som garanterer en løsning på $(2 * N - 3)$ træk, dersom listen indeholder N cifre. Finessen ved den algoritmiske løsning er, at man i forvejen ved, hvad det næste træk bliver. Man kan altså let programmere en datamat til at gøre dette.

Den heuristiske metode gør brug af »delsortering« af talrækken ved hvert træk. Ved denne metode bliver hvert træk afhængigt af, hvordan talrækken ser ud i øjeblikket. Denne løsningsmodel garanterer ikke en løsning inden for et givet antal træk. Men hvis du er heldig – og gør de rigtige træk – kan du løse problemet med færre træk end den algoritmiske metode. Den heuristiske metode er dog vanskelig at programmere.

I praksis anvender de fleste spillere en blandet strategi, der består af både algoritmiske og heuristiske træk. Er dette bedre end en ren strategi?

Programmet er skrevet af Peter Sessions fra People's Computer Company, og ovenstående beskrivelse kommer fra hans oprindelige spillevejledning.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BAGLÆNS

- ET FÆRDIGHEDSSPIL

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DETTE SPIL HEDDER 'BAGLÆNS'. DET ENESTE DU SKAL GØRE FOR AT VINDE ER AT SORTERE EN TALRÆKKE (MED TAL FRA 1 TIL 9) I NUMERISK RÆKKEFØLGE FRA VENSTRE TIL HØJRE.

FOR AT OMROKERE TALLENE SKAL DU INDTASTE HVOR MANGE CIFRE (TALT FRA TALRÆKENS VENSTRE SIDE) DU ØNSKER AT VENDE. HVIS TALRÆKKEN F. EKS. SER SÅLEDES UD:

2 3 4 5 1 6 7 8 9

OG DU VENDER 4, BLIVER RESULTATET:

5 4 3 2 1 6 7 8 9

HVIS DU NU VENDER 5, HAR DU VUNDET!

1 2 3 4 5 6 7 8 9

DU VIL UDEN TVIVL SYNES OM SPILLET, MEN HVIS DU SKULLE ØNSKE AT STOPPE, SR KAN DU BLOT INDTASTE 'O'.

SR GØR VI IGANG. LISTEN SER SÅDAN UD:

9 4 1 8 7 6 2 5 3

HVOR MANGE SKAL JEG VENDE ?9

3 5 2 6 7 8 1 4 9

HVOR MANGE SKAL JEG VENDE ?6

8 7 6 2 5 3 1 4 9

HVOR MANGE SKAL JEG VENDE ?8

4 1 3 5 2 6 7 8 9

HVOR MANGE SKAL JEG VENDE ?4

5 3 1 4 2 6 7 8 9

HVOR MANGE SKAL JEG VENDE ?5

2 4 1 3 5 6 7 8 9

HVOR MANGE SKAL JEG VENDE ?3

1 4 2 3 5 6 7 8 9

HVOR MANGE SKAL JEG VENDE ?2

4 1 2 3 5 6 7 8 9

HVOR MANGE SKAL JEG VENDE ?4

3 2 1 4 5 6 7 8 9

HVOR MANGE SKAL JEG VENDE ?3

1 2 3 4 5 6 7 8 9

DU VANDT PÅ 9 TRÆK!!!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

FARVEL.....HÅBER DU HAR MORET DIG!!

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          BAGLÆNS"
130 PRINT "-----"
140 PRINT
150 PRINT " - ET FÆRDIGHEDSSPIL"
160 PRINT
170 DIM A(20)
180 REM N=antal cifre
190 N = 9
200 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
210 INPUT A$
220 IF LEFT$(A$,1) = "N" THEN 260
230 GOSUB 750
240 REM Lav en tilfældig liste
250 REM A(1) til A(N)
260 A(1) = INT ((N - 1) * RND (1) + 2)
270 FOR K = 2 TO N
280 A(K) = INT (N * RND (1) + 1)
290 FOR J = 1 TO K - 1
300 IF A(K) = A(J) THEN 280
310 NEXT J
320 NEXT K
330 REM Udskriv liste og start spil
340 PRINT
350 PRINT "SÅ GÅR VI IGANG. LISTEN SER SÅDAN UD:"
360 T = 0
370 GOSUB 670
380 PRINT "HVOR MANGE SKAL JEG VENDE ";
390 INPUT R
400 IF R = 0 THEN 590
410 IF R < = N THEN 450
420 PRINT "OOOOPS! ALT FOR MANGE, JEG KAN HØJST"
430 PRINT "VENDE";N
440 GOTO 380
450 T = T + 1
460 REM Vend R cifre og udskriv ny liste
470 FOR K = 1 TO INT (R / 2)
480 Z = A(K)
490 A(K) = A(R - K + 1)
500 A(R - K + 1) = Z
510 NEXT K
520 GOSUB 670
530 REM Er der gevinst?
540 FOR K = 1 TO N
550 IF A(K) < > K THEN 380
560 NEXT K
570 PRINT "DU VANDT PR";T;"TRÆK!!!"
580 PRINT
590 PRINT
600 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
610 INPUT A$
620 PRINT
630 IF LEFT$(A$,1) = "J" THEN 260
640 PRINT "FARVEL.....HÅBER DU HAR MORET DIG!!"
650 GOTO 1020
660 REM Subrutine der udskriver liste
670 PRINT
680 FOR K = 1 TO N
690 PRINT A(K);
700 NEXT K
710 PRINT
720 PRINT
730 RETURN
740 REM Subrutine der skriver instruktion
750 PRINT
760 PRINT "DETTE SPIL HEDDER 'BAGLÆNS'. DET ENESTE"
770 PRINT "DU SKAL GØRE FOR AT VINDE ER AT SORTERE"
780 PRINT "EN TALRÆKKE (MED TAL FRA 1 TIL";N;"") I"
790 PRINT "NUMERISK RÆKKEFØLGE FRA VENSTRE TIL"
800 PRINT "HØJRE."
810 PRINT
820 PRINT "FOR AT OMROKERE TALLENE SKAL DU INDTAS--"
830 PRINT "TE HVOR MANGE CIFRE (TALT FRA TALRÆK--"
840 PRINT "KENS VENSTRE SIDE) DU ØNSKER AT VENDE."
850 PRINT "HVIS TALRÆKKEN F.EK.S. SER SÅLEDES UD:"
860 PRINT
870 PRINT "2 3 4 5 1 6 7 8 9"
880 PRINT
890 PRINT "OG DU VENDER 4, BLIVER RESULTATET:"
900 PRINT
910 PRINT "5 4 3 2 1 6 7 8 9"
920 PRINT
930 PRINT "HVIS DU NU VENDER 5, HAR DU VUNDET!"
940 PRINT
950 PRINT "1 2 3 4 5 6 7 8 9"
960 PRINT
970 PRINT "DU VIL UDEN TVIVL SYNES OM SPILLET,"
980 PRINT "MEN HVIS DU SKULLE ØNSKE AT STOPPE, SÅ"
990 PRINT "KAN DU BLOT INDTASTE '0'."
1000 PRINT
1010 RETURN
1020 END

```



```

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                      BANNER"
130 PRINT "-----"
140 PRINT
150 PRINT
160 PRINT "  BREDDE ";
170 INPUT X
180 PRINT "    HØJDE ";
190 INPUT Y
200 PRINT "CENTRERET ";
210 INPUT L$
220 G1 = 0
230 IF L$ = "Y" THEN G1 = 1
240 PRINT "TEGN (TAST 'ALLE' HVIS ALLE TEGN ØNSKES) ";
250 INPUT M$
260 PRINT "  BUDSKAB ";
270 INPUT A$
280 PRINT
290 PRINT
300 A = ASC ( LEFT$ (A$,1))
310 FOR T = 1 TO LEN (A$)
320 P$ = MID$ (A$,T,1)
330 FOR O = 1 TO 50
340 READ S$,S(1),S(2),S(3),S(4),S(5),S(6),S(7)
350 IF P$ = " " THEN 750
360 IF P$ = S$ THEN 380
370 NEXT O
380 RESTORE
390 X$ = M$
400 IF M$ = "ALLE" THEN X$ = S$
410 FOR U = 1 TO 7
420 FOR K = 8 TO 0 STEP - 1
430 IF 2 ^ K < S(U) THEN 460
440 J(9 - K) = 0
450 GOTO 490
460 J(9 - K) = 1
470 S(U) = S(U) - 2 ^ K
480 IF S(U) = 1 THEN 790
490 NEXT K
500 FOR T1 = 1 TO X
510 PRINT TAB( 40 - 4.5 * Y) * G1 / ( LEN (X$) + 1);
520 FOR B = 1 TO F(U)
530 IF J(B) = 0 THEN 580
540 FOR I = 1 TO Y
550 PRINT X$;
560 NEXT I
570 GOTO 630
580 FOR I = 1 TO Y
590 FOR I1 = 1 TO LEN (X$)
600 PRINT " ";
610 NEXT I1
620 NEXT I
630 NEXT B
640 PRINT
650 NEXT T1
660 NEXT U
670 FOR H = 1 TO 2 * X
680 PRINT
690 NEXT H
700 NEXT T
710 FOR H = 1 TO 75
720 PRINT
730 NEXT H
740 END
750 FOR H = 1 TO 7 * X
760 PRINT
770 NEXT H
780 GOTO 700
790 F(U) = 9 - K: GOTO 500
800 DATA " ",0,0,0,0,0,0,0
810 DATA "A",505,37,35,34,35,37,505
820 DATA "G",125,131,258,258,290,163,101
830 DATA "E",512,274,274,274,274,258,258
840 DATA "T",2,2,2,512,2,2,2
850 DATA "W",256,257,129,65,129,257,256
860 DATA "L",512,257,257,257,257,257,257
870 DATA "S",69,139,274,274,274,163,69
880 DATA "O",125,131,258,258,258,131,125
890 DATA "N",512,7,9,17,33,193,512
900 DATA "F",512,18,18,18,18,2,2
910 DATA "K",512,17,17,41,69,131,258
920 DATA "B",512,274,274,274,274,274,239
930 DATA "D",512,258,258,258,258,131,125
940 DATA "H",512,17,17,17,17,17,512
950 DATA "M",512,7,13,25,13,7,512
960 DATA "?",5,3,2,354,18,11,5
970 DATA "U",128,129,257,257,257,129,128
980 DATA "R",512,18,18,50,82,146,271
990 DATA "P",512,18,18,18,18,15
1000 DATA "Q",125,131,258,258,322,131,381
1010 DATA "Y",8,9,17,481,17,9,8
1020 DATA "V",64,65,129,257,129,65,64
1030 DATA "X",388,69,41,17,41,69,388
1040 DATA "Z",386,322,290,274,266,262,260
1050 DATA "I",258,258,258,512,258,258,258
1060 DATA "C",125,131,258,258,258,131,69
1070 DATA "J",65,129,257,257,257,129,128
1080 DATA "1",0,0,261,259,512,257,257
1090 DATA "2",261,387,322,290,274,267,261
1100 DATA "*",69,41,17,512,17,41,69
1110 DATA "3",66,130,258,274,266,150,100
1120 DATA "4",33,49,41,37,35,512,33
1130 DATA "5",160,274,274,274,274,274,226
1140 DATA "6",194,291,293,297,305,289,193
1150 DATA "7",258,130,66,34,18,10,8
1160 DATA "8",69,171,274,274,274,171,69
1170 DATA "9",263,138,74,42,26,10,7
1180 DATA "=",41,41,41,41,41,41,41
1190 DATA "!",1,1,1,384,1,1,1
1200 DATA "0",57,69,131,258,131,69,57
1210 DATA ".",1,1,129,449,129,1,1
1220 DATA "R",505,21,19,18,512,274,274
1230 DATA "ø",381,131,354,274,270,131,126
1240 DATA "R",481,81,75,70,75,81,481
1250 END

```

Basketball

Dette program simulerer en basketballkamp mellem Dartmouth College og en modstander, som du selv angiver. Du er holdkaptajn på Dartmouth, og du bestemmer, hvilke typer skud og hvilke former for forsvar som holdet skal anvende i spillets løb.

Der er fire typer skud: 1) Langt hopskud (10 m), 2) Kort hopskud (5 m), 3) Oplæg og 4) Stående skud. Begge hold anvender samme forsvarsteknik, men du bestemmer hvilken (også her er der fire slags): 6) Modtryk, 6.5) Mand-mod-mand, 7) Zoneforsvar eller 7.5) Ingen taktik. Hvis du ønsker at ændre forsvarstaktik, skal du blot taste '0' som dit næste skud.

Bemærk: Spillet favoriserer Dartmouth. Den gennemsnitlige sandsynlighed for, at et Dartmouth-skud vil lykkes, er 62.95%, mens sandsynligheden for, at et af modstanderens skud vil lykkes, er 61.85%.

Charles Bacheller fra Dartmouth College er den oprindelige forfatter til dette spil.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BASKETBALL

DETTE ER DARTMOUTH COLLEGE BASKETBALL.

DU ER HOLDKAPTAJN OG TRÆNER FOR DARTMOUTH-HOLDET. DU KAN VÆLGE MELLEML NEDENSTÅENDE FIRE FORMER FOR SKUD:

- 1 = LANGT HOPSKUD (10 M.)
- 2 = KORT HOPSKUD (5 M.)
- 3 = LAY-UP SKUD
- 4 = STÅENDE SKUD

BEGGE HOLD ANVENDER DEN SAMME FORSVARSMETODE. DISSE ER:

- 6 = MODTRYK
- 6.5 = MAND-MOD-MAND
- 7 = ZONEFORSVAR
- 7.5 = INTET FORSVAR

HVIS DU ØNSKER AT ÆNDRE FORSVARSTAKTIK SKAL DU BLOT INDTASTE '0' (NUL).

HVILKET FORSVAR ØNSKER DU ?6.5

HVILKET HOLD VIL DU HAVE SOM MODSTANDER ?CORNELL

CENTERFORWARDEN HOPPER. CORNELL DRIBLER BOLDEN.

LAY-UP SKUD. DÆRLIGT SKUD. CORNELL FÅR RETURBOLDEN. AFLEVERER LYNHURTIGT BAGLÆNS TIL CORNELL'S FORSVAR.

LAY-UP SKUD. DÆRLIGT SKUD. CORNELL FÅR RETURBOLDEN. LAY-UP SKUD. DÆRLIGT SKUD. DARTMOUTH FÅR RETURBOLDEN.

DIT KAST ?4

STÅENDE KAST. BOLDEN RAMMER NETKANTEN. CORNELL FÅR RETURBOLDEN.

HOPSKUD. GODT SKUD. STILLINGEN: 0 - 2

DIT KAST ?1

HOPSKUD. SKUDDET RAMTE VED SIDEN AF. CORNELL FÅR RETURBOLDEN.

HOPSKUD. GODT SKUD. STILLINGEN: 0 - 4

DIT KAST ?4

STÅENDE KAST. BOLDEN RAMMER NETKANTEN. DARTMOUTH FÅR RETURBOLDEN. STÅENDE KAST. KASTEREN UREGLEENTERET. 2 STRAFFE! SPILLEREN SCORER 2 POINT! STILLINGEN: 2 - 4

HOPSKUD. BOLDEN RAMTE NETKANTEN. DARTMOUTH FÅR RETURBOLDEN.

DIT KAST ?0

DIN NYE FORSVARSSTRATEGI ER ?7

DIT KAST ?3

LAY-UP SKUD. SKUDDET BLOKERES. CORNELL HAR BOLDEN IGEN.

HOPSKUD. UREGLEENTERET SPIL. 2 STRAFFE! SPILLEREN SCORER 2 POINT! STILLINGEN: 2 - 6

DIT KAST ?4

STÅENDE KAST. GODT KAST. 2 POINT! STILLINGEN: 4 - 6

LAY-UP SKUD. DÆRLIGT SKUD. DARTMOUTH FÅR RETURBOLDEN.

DIT KAST ?4

STÅENDE KAST. BOLDEN RAMMER NETKANTEN. CORNELL FÅR RETURBOLDEN.

HOPSKUD. BOLDEN RAMTE NETKANTEN. CORNELL FÅR RETURBOLDEN. LAY-UP SKUD. GODT SKUD. STILLINGEN: 4 - 8

DIT KAST ?1

HOPSKUD. UREGLEENTERET SPIL! DARTMOUTH MÅ AFGIVE BOLDEN.

HOPSKUD. GODT SKUD. STILLINGEN: 4 - 10

DIT KAST ?2

HOPSKUD. GODT SKUD. STILLINGEN: 6 - 10

STÅENDE KAST. GODT SKUD. STILLINGEN: 6 - 12

DIT KAST ?2

HOPSKUD. GODT SKUD. STILLINGEN: 8 - 12

LAY-UP SKUD. GODT SKUD. STILLINGEN: 8 - 14

DIT KAST ?4

STÅENDE KAST. GODT KAST. 2 POINT! STILLINGEN: 10 - 14

HOPSKUD. BOLDEN RAMTE NETKANTEN. CORNELL FÅR RETURBOLDEN. LAY-UP SKUD. DÆRLIGT SKUD. CORNELL FÅR RETURBOLDEN. LAY-UP SKUD. GODT SKUD. STILLINGEN: 10 - 16

DIT KAST ?2

HOPSKUD. KASTEREN SPILLER UREGLEENTERET! TO STRAFFE! SPILLEREN KIKSER ET SKUD, OG SCORER KUN ET POINT. STILLINGEN: 11 - 16

LAY-UP SKUD. DÆRLIGT SKUD. CORNELL FÅR RETURBOLDEN. LAY-UP SKUD. DÆRLIGT SKUD. DARTMOUTH FÅR RETURBOLDEN.

DIT KAST ?4

STÅENDE KAST. GODT KAST. 2 POINT! STILLINGEN: 13 - 16

LAY-UP SKUD. GODT SKUD. STILLINGEN: 13 - 18

DIT KAST ?1

HOPSKUD. SKUDDET BLOKERES. BOLDEN SPILLES AF DARTMOUTH.

DIT KAST ?0

DIN NYE FORSVARSSTRATEGI ER ?6

DIT KAST ?2

HOPSKUD.
SKUDET RAMTE VED SIDEN AF.
DARTMOUTH FÅR RETURBOLDEN.
DU FÅR BOLDEN TILBAGE.

DIT KAST ?4

STÅRENDE KAST.
KASTEREN UREGLEMTERET. 2 STRAFFE!
SPILLEREN SCORER 2 POINT!
STILLINGEN: 15 - 18

HOPSKUD.
UREGLEMTERET OFFENSIV. BOLDEN GÅR
TIL DARTMOUTH.

DIT KAST ?2

HOPSKUD.
GODT SKUD.
STILLINGEN: 17 - 18

LAY-UP SKUD.
DÅRLIGT SKUD.
CORNELL FÅR RETURBOLDEN.
AFLEVERER LYNHURTIGT BAGLÆNS TIL
CORNELL'S FORSVAR.

HOPSKUD.
GODT SKUD.
STILLINGEN: 17 - 20

DIT KAST ?2

HOPSKUD.
SKUDET RAMTE VED SIDEN AF.
DARTMOUTH FÅR RETURBOLDEN.
LAY-UP SKUD.
BOLDEN RAMMER NETKANTEN.
CORNELL FÅR RETURBOLDEN.

HOPSKUD.
BOLDEN RAMTE NETKANTEN.
CORNELL FÅR RETURBOLDEN.
AFLEVERER LYNHURTIGT BAGLÆNS TIL
CORNELL'S FORSVAR.

HOPSKUD.
BOLDEN RAMTE NETKANTEN.
CORNELL FÅR RETURBOLDEN.
AFLEVERER LYNHURTIGT BAGLÆNS TIL
CORNELL'S FORSVAR.

HOPSKUD.
BOLDEN RAMTE NETKANTEN.
DARTMOUTH FÅR RETURBOLDEN.

DIT KAST ?4

STÅRENDE KAST.
BOLDEN RAMMER NETKANTEN.
CORNELL FÅR RETURBOLDEN.

HOPSKUD.
BOLDEN RAMTE NETKANTEN.
CORNELL FÅR RETURBOLDEN.
LAY-UP SKUD.
GODT SKUD.
STILLINGEN: 17 - 22

DIT KAST ?3

LAY-UP SKUD.
GODT KAST. 2 POINT!
STILLINGEN: 19 - 22

HOPSKUD.
UREGLEMTERET OFFENSIV. BOLDEN GÅR
TIL DARTMOUTH.

DIT KAST ?3

LAY-UP SKUD.
SKUDET BLOKERES.
CORNELL HAR BOLDEN IGEN.

LAY-UP SKUD.
DÅRLIGT SKUD.
CORNELL FÅR RETURBOLDEN.
AFLEVERER LYNHURTIGT BAGLÆNS TIL
CORNELL'S FORSVAR.

STÅRENDE KAST.
DÅRLIGT SKUD.
CORNELL FÅR RETURBOLDEN.
AFLEVERER LYNHURTIGT BAGLÆNS TIL
CORNELL'S FORSVAR.

HOPSKUD.
BOLDEN RAMTE NETKANTEN.
CORNELL FÅR RETURBOLDEN.
AFLEVERER LYNHURTIGT BAGLÆNS TIL
CORNELL'S FORSVAR.

***** SLUT PÅ FØRSTE HALVLEG *****

STILLING EFTER 1. HALVLEG:
DARTMOUTH: 19
CORNELL: 22

CENTERFORWARDEN HOPPER.
DARTMOUTH DRIBLER BOLDEN.

DIT KAST ?2

HOPSKUD.
SKUDET RAMTE VED SIDEN AF.
CORNELL FÅR RETURBOLDEN.

STÅRENDE KAST.
GODT SKUD.
STILLINGEN: 19 - 24

DIT KAST ?

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          BASKETBALL"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "DETTE ER DARTMOUTH COLLEGE BASKETBALL."  
160 PRINT  
170 PRINT "DU ER HOLDKAPTJEN OG TRÆNER FOR DART-"  
180 PRINT "MOUTH-HOLDET. DU KAN VÆLGE MELLEM"  
190 PRINT "NEDENSTÅRENDE FIRE FORMER FOR SKUD:"  
200 PRINT "    1 = LANGT HOPSKUD (10 M.)"  
210 PRINT "    2 = KORT HOPSKUD (5 M.)"  
220 PRINT "    3 = LAY-UP SKUD"  
230 PRINT "    4 = STÅRENDE SKUD"  
240 PRINT  
250 PRINT "BEGGE HOLD ANVENDER DEN SAMME FORSVARS-"  
260 PRINT "METODE. DISSE ER:"  
270 PRINT  
280 PRINT "    6 = MODTRYK"  
290 PRINT "    6.5 = MAND-MOD-MAND"  
300 PRINT "    7 = ZONEFORSVAR"  
310 PRINT "    7.5 = INTET FORSVAR"  
320 PRINT  
330 PRINT "HVIS DU ØNSKER AT ÆNDRE FORSVARSTAKTIK"  
340 PRINT "SKAL DU BLOT INDTASTE '0' (NUL)."  
350 PRINT  
360 PRINT "HVILKET FORSVAR ØNSKER DU ";  
370 INPUT D  
380 PRINT  
390 IF D < 6 THEN 1690  
400 PRINT "HVILKET HOLD VIL DU HAVE SOM"  
410 PRINT "MODSTANDER ";  
420 INPUT O$  
430 PRINT  
440 PRINT "CENTERFORWARDEN HOPPER."  
450 IF RND (1) > 3 / 5 THEN 480  
460 PRINT O$;" DRIBLER BOLDEN."  
470 GOTO 1740  
480 PRINT "DARTMOUTH DRIBLER BOLDEN."  
490 PRINT  
500 PRINT "DIT KAST ";  
510 INPUT Z  
520 P = 0  
530 IF Z < > INT (2) THEN 570  
540 IF Z < 0 OR Z > 4 THEN 570  
550 PRINT  
560 GOTO 590  
570 PRINT "FORKERT SVAR. OM IGEN!"  
580 GOTO 490  
590 IF RND (1) < .5 THEN 850  
600 IF T < 100 THEN 850  
610 PRINT  
620 IF S(1) < > S(0) THEN 740  
630 PRINT  
640 PRINT "***** SLUT PÅ ANDEN HALVLEG *****"  
650 PRINT  
660 PRINT "STILLINGEN ER I ØJEBLIKKET:"  
670 PRINT "    ","DARTMOUTH:";S(1)  
680 PRINT "    ","O$;" ;S(0)  
690 PRINT  
700 PRINT "*** NU BEGYNDER TO MINUTTERS OVERTID ***"  
710 PRINT  
720 T = 93  
730 GOTO 440  
740 PRINT  
750 PRINT "***** SPILLET ER SLUT *****"  
760 PRINT  
770 PRINT "SLUTRESULTATET BLEV:"  
780 PRINT "    ","DARTMOUTH:";S(1)  
790 PRINT "    ","O$;" ;S(0)  
800 GOTO 2570  
810 PRINT  
820 PRINT "*** TO MINUTTER TILBAGE AF SPILLET ***"  
830 PRINT  
840 RETURN  
850 IF Z > 2 THEN 870  
860 ON Z GOTO 880,880  
870 GOTO 1340  
880 T = T + 1  
890 IF T = 50 THEN 2470  
900 IF T = 92 THEN 920  
910 GOTO 930  
920 GOSUB 810  
930 PRINT "HOPSKUD."  
940 IF RND (1) > .341 * D / 8 THEN 980  
950 PRINT "GODT SKUD."  
960 GOSUB 2440  
970 GOTO 1740
```

```

980 IF RND (1) > .628 * D / 8 THEN 1180
990 PRINT "SKUDET RAMTE VED SIDEN AF."
1000 IF D / 6 * RND (1) > .45 THEN 1030
1010 PRINT "DARTMOUTH FÅR RETURBOLDEN."
1020 GOTO 1050
1030 PRINT OS;" FÅR RETURBOLDEN."
1040 GOTO 1740
1050 IF RND (1) > .4 THEN 1070
1060 GOTO 1340
1070 IF D = 6 THEN 2350
1080 PRINT "DU FÅR BOLDEN TILBAGE."
1090 GOTO 490
1100 IF RND (1) > .9 THEN 1150
1110 PRINT "SPILLEREN SPILLER UREGLEMENTERET."
1120 PRINT "TO STRAFFE!"
1130 GOSUB 2160
1140 GOTO 1740
1150 PRINT "BOLDEN TABES."
1160 PRINT OS;" HAR BOLDEN."
1170 GOTO 1740
1180 IF RND (1) > .782 * D / 8 THEN 1260
1190 PRINT "SKUDET BLOKERES."
1200 PRINT "BOLDEN SPILLES AF ";
1210 IF RND (1) > .5 THEN 1240
1220 PRINT "DARTMOUTH."
1230 GOTO 490
1240 PRINT OS;"."
1250 GOTO 1740
1260 IF RND (1) > .843 * D / 8 THEN 1310
1270 PRINT "KASTEREN SPILLER UREGLEMENTERET!"
1280 PRINT "TO STRAFFE!"
1290 GOSUB 2160
1300 GOTO 1740
1310 PRINT "UREGLEMENTERET SPIL!"
1320 PRINT "DARTMOUTH MÅ AFGIVE BOLDEN."
1330 GOTO 1740
1340 T = T + 1
1350 IF T = 50 THEN 2470
1360 IF T = 92 THEN 1380
1370 GOTO 1390
1380 GOSUB 810
1390 IF Z = 0 THEN 1690
1400 IF Z > 3 THEN 1670
1410 PRINT "LAY-UP SKUD."
1420 IF 7 / D * RND (1) > .4 THEN 1460
1430 PRINT "GODT KAST. 2 POINT!"
1440 GOSUB 2440
1450 GOTO 1740
1460 IF 7 / D * RND (1) > .7 THEN 1560
1470 PRINT "BOLDEN RAMMER NETKANTEN."
1480 IF RND (1) > 2 / 3 THEN 1510
1490 PRINT OS;" FÅR RETURBOLDEN."
1500 GOTO 1740
1510 PRINT "DARTMOUTH FÅR RETURBOLDEN."
1520 IF RND (1) > .4 THEN 1540
1530 GOTO 1340
1540 PRINT "DU FÅR BOLDEN TILBAGE."
1550 GOTO 490
1560 IF 7 / D * RND (1) > .875 THEN 1600
1570 PRINT "KASTEREN UREGLEMENTERET. 2 STRAFFE!"
1580 GOSUB 2160
1590 GOTO 1740
1600 IF 7 / D * RND (1) > .925 THEN 1640
1610 PRINT "SKUDET BLOKERES."
1620 PRINT OS;" HAR BOLDEN IGEN."
1630 GOTO 1740
1640 PRINT "DOMMEREN DØMMER UREGLEMENTERET SPIL!"
1650 PRINT "DARTMOUTH MISTER BOLDEN."
1660 GOTO 1740
1670 PRINT "STRÆNDE KAST."
1680 GOTO 1420
1690 PRINT "DIN NYE FORSVARSSTRATEGI ER ";
1700 INPUT D
1710 IF D < 6 THEN 1690
1720 GOTO 490
1730 GOSUB 810
1740 P = 1
1750 T = T + 1
1760 IF T = 50 THEN 2470
1770 GOTO 1790

1780 GOSUB 810
1790 PRINT
1800 Z1 = 10 / 4 * RND (1) + 1
1810 IF Z1 > 2 THEN 2060
1820 PRINT "HOPSKUD."
1830 IF 8 / D * RND (1) > .35 THEN 1870
1840 PRINT "GODT SKUD."
1850 GOSUB 2410
1860 GOTO 490
1870 IF 8 / D * RND (1) > .75 THEN 1990
1880 PRINT "BOLDEN RAMTE NETKANTEN."
1890 IF D / 6 * RND (1) > .5 THEN 1920
1900 PRINT "DARTMOUTH FÅR RETURBOLDEN."
1910 GOTO 490
1920 PRINT OS;" FÅR RETURBOLDEN."
1930 IF D = 6 THEN 2290
1940 IF RND (1) > .5 THEN 1980
1950 PRINT "AFLEVERER LYNHURTIGT BAGLÆNS TIL"
1960 PRINT OS;"S FORSVAR."
1970 GOTO 1740
1980 GOTO 2060
1990 IF 8 / D * RND (1) > .9 THEN 2030
2000 PRINT "UREGLEMENTERET SPIL. 2 STRAFFE!"
2010 GOSUB 2160
2020 GOTO 490
2030 PRINT "UREGLEMENTERET OFFENSIV. BOLDEN GRÆR"
2040 PRINT "TIL DARTMOUTH."
2050 GOTO 490
2060 IF Z1 > 3 THEN 2140
2070 PRINT "LAY-UP SKUD."
2080 IF 7 / D * RND (1) > .413 THEN 2120
2090 PRINT "GODT SKUD."
2100 GOSUB 2410
2110 GOTO 490
2120 PRINT "DÆRLIGT SKUD."
2130 GOTO 1890
2140 PRINT "STRÆNDE KAST."
2150 GOTO 2080
2160 REM Straffekast
2170 IF RND (1) > .49 THEN 2220
2180 PRINT "SPILLEREN SCORER 2 POINT!"
2190 S(1 - P) = S(1 - P) + 2
2200 GOSUB 2420
2210 RETURN
2220 IF RND (1) > .75 THEN 2270
2230 PRINT "SPILLEREN KIKSER ET SKUD, OG SCORER"
2240 PRINT "KUN ET POINT."
2250 S(1 - P) = S(1 - P) + 1
2260 GOTO 2200
2270 PRINT "BEGGE SKUD KIKSER!"
2280 GOTO 2200
2290 IF RND (1) > .75 THEN 2310
2300 GOTO 1940
2310 PRINT "BOLDEN TAGET. LET LAY-UP FOR"
2320 PRINT "DARTMOUTH."
2330 GOSUB 2440
2340 GOTO 1740
2350 IF RND (1) > .6 THEN 2370
2360 GOTO 1080
2370 PRINT "BOLDEN TAGES AF ";OS;"."
2380 PRINT "ET LET LAY-UP."
2390 GOSUB 2410
2400 GOTO 490
2410 S(0) = S(0) + 2
2420 PRINT "STILLINGEN: ";S(1);"-";S(0)
2430 RETURN
2440 S(1) = S(1) + 2
2450 GOSUB 2420
2460 RETURN
2470 PRINT
2480 PRINT "***** SLUT PÅ FØRSTE HALVLEG *****"
2490 PRINT
2500 PRINT "STILLING EFTER 1. HALVLEG:"
2510 PRINT " ", "DARTMOUTH: ";S(1)
2520 PRINT " ", OS;" ";S(0)
2530 PRINT
2540 PRINT
2550 GOTO 440
2560 PRINT
2570 END

```


Bille

I dette spil gælder det om at gøre tegningen af din bille færdig, før datamaten får tegnet sin.

Du og datamaten skiftes til at kaste en terning, hvor hver af tegningens talværdier svarer til en bestemt del af billen. Når der er føjet en legemsdel til billen, har du mulighed for at få udskrevet begge tegningerne på skærmen.

Hvis du vælger at se alle udskrifterne og bruger en lineskriver, har dette program den særlige egenskab, at det bruger næsten to meter papir pr. spil! Vi foreslår derfor, at du bruger papiret på begge sider.

Brian Leibowitz skrev dette program, mens han gik i syvende klasse på Harrison Jr-Sr High School i Harrison, New York.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BILLE

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

I DETTE SPIL DREJER DET SIG OM AT FÅ
TEGNET DIN BILLE FÆRDIG FØR MIN.
HVERT TAL STÅR FOR EN DEL AF BILLENS
KROP. JEG KASTER EN TERNING FOR DIG, OG
FORTÆLLER DIG HVILKEN DEL AF BILLENS
KROP ANTALLET AF ØJNE PÅ TERNINGEN
SVARER TIL.

JEG FORTÆLLER DIG OGSÅ OM OM DU KAN
BRUGE KROPSDELEN TIL NOGET.

JEG GØR DET SAMME NÅR DET ER MIN TUR.

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?

HVIS EN AF BILLERNE FÅR EN NY KROPSDEL,
HAR DU MULIGHED FOR AT FÅ BILLEN AT SE.

ANTALLET AT ØJNE PÅ TERNINGEN SVARER
TIL FØLGENDE KROPSDELE:

TERNING- ØJNE	KROPS- DEL	ANTAL DELE
1	KROP	1
2	HALS	1
3	HOVED	1
4	FØLEHORN	2
5	HALE	1
6	BEN	6

DU SLOG EN 4 'ER
4 = FØLEHORN
DU MÅNGLER ET HOVED.

JEG SLOG EN 6 'ER.
6 = BEN
JEG HAR INGEN KROP.

DU SLOG EN 2 'ER
2 = HALS
DU HAR JO INGEN KROP.

JEG SLOG EN 1 'ER.
1 = KROP
NU HAR JEG EN KROP.

ØNSKER DU AT SE BILLERNE ?N

DU SLOG EN 6 'ER
6 = BEN
DU HAR INGEN KROP.

JEG SLOG EN 2 'ER.
2 = HALS
NU HAR JEG EN HALS.

ØNSKER DU AT SE BILLERNE ?N

DU SLOG EN 6 'ER
6 = BEN
DU HAR INGEN KROP.

JEG SLOG EN 4 'ER.
4 = FØLEHORN
JEG HAR IKKE NOGET HOVED.

DU SLOG EN 6 'ER
6 = BEN
DU HAR INGEN KROP.

JEG SLOG EN 5 'ER.
5 = HALE
NU HAR JEG EN HALE.

ØNSKER DU AT SE BILLERNE ?N

DU SLOG EN 5 'ER
5 = HALE
DU HAR INGEN KROP.

JEG SLOG EN 3 'ER.
3 = HOVED
JEG MÅNGLER NETOP ET HOVED.

ØNSKER DU AT SE BILLERNE ?J

***** DIN BILLE *****

***** MIN BILLE *****

```

HHHHH
H   H
H - - H
H O O H
H ! H
H --- H
HHHHH
H H
H H
KKKKKKKKKKK
K       K
K       K
====K   K
KKKKKKKKKKK

```

DU SLOG EN 2 'ER
2 = HALS
DU HAR JO INGEN KROP.

JEG SLOG EN 2 'ER.
2 = HALS
JEG HAR IKKE BRUG FOR EN HALS.

DU SLOG EN 6 'ER
6 = BEN
DU HAR INGEN KROP.

JEG SLOG EN 4 'ER.
4 = FØLEHORN
JEG FÅR ET FØLEHORN.

ØNSKER DU AT SE BILLERNE ?N

DU SLOG EN 2 'ER
2 = HALS
DU HAR JO INGEN KROP.

JEG SLOG EN 4 'ER.
4 = FØLEHORN
JEG FÅR ET FØLEHORN.

ØNSKER DU AT SE BILLERNE ?J

***** DIN BILLE *****

***** MIN BILLE *****

```

O O
) (
!!
!!
HHHHH
H   H
H - - H
H O O H
H ! H
H --- H
HHHHH
H H
H H
KKKKKKKKKKK
K       K
K       K
====K   K
KKKKKKKKKKK

```

DU SLOG EN 1 'ER
1 = KROP
NU HAR DU EN KROP.

JEG SLOG EN 4 'ER.
4 = FØLEHORN
JEG HAR ALLEREDE TO FØLEHORN.

ØNSKER DU AT SE BILLERNE ?N

DU SLOG EN 5 'ER
5 = HALE
NU FÅR DU EN HALE.

JEG SLOG EN 3 'ER.
3 = HOVED
JEG HAR IKKE BRUG FOR ET HOVED.

ØNSKER DU AT SE BILLERNE ?N

DU SLOG EN 2 'ER
2 = HALS
NU HAR DU EN HALS.

JEG SLOG EN 4 'ER.
4 = FØLEHORN
JEG HAR ALLEREDE TO FØLEHORN.

ØNSKER DU AT SE BILLERNE ?J

***** DIN BILLE *****

```
  H H
  H H
 KKKKKKKKKKK
  K     K
  K     K
====K     K
 KKKKKKKKKKK
```

***** MIN BILLE *****

```
  O  O
  ) (
  ! !
  ! !
 HHHHH
  H   H
  H - - H
  H O O H
  H ! H
  H --- H
  HHHHH
  H H
  H H
 KKKKKKKKKKK
  K     K
  K     K
====K     K
 KKKKKKKKKKK
```

DU SLOG EN 5 'ER
5 = HALE
DU HAR ALLEREDE EN HALE.

JEG SLOG EN 3 'ER.
3 = HOVED
JEG HAR IKKE BRUG FOR ET HOVED.

DU SLOG EN 6 'ER
6 = BEN
NU HAR DU 1 BEN.

JEG SLOG EN 3 'ER.
3 = HOVED
JEG HAR IKKE BRUG FOR ET HOVED.

ØNSKER DU AT SE BILLERNE ?

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          BILLE"
130 PRINT "-----"
140 A = 0:B = 0:H = 0:L = 0
150 N = 0:P = 0:Q = 0:R = 0
160 S = 0:T = 0:U = 0:V = 0:Y = 0
170 PRINT
180 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
190 INPUT Z$
200 PRINT
210 IF LEFT$(Z$,1) = "N" THEN 560
220 PRINT "I DETTE SPIL DREJER DET SIG OM AT FÅ"
230 PRINT "TEGNET DIN BILLE FÆRDIG FØR MIN."
240 PRINT "HVERT TAL STÅR FOR EN DEL AF BILLENS"
250 PRINT "KROP. JEG KASTER EN TERNING FOR DIG, OG"
260 PRINT "FORTÆLLER DIG HVILKEN DEL AF BILLENS"
270 PRINT "KROP ANTALLET AF ØJNE PÅ TERNINGEN"
280 PRINT "SVARER TIL."
290 PRINT
300 PRINT "JEG FORTÆLLER DIG OGSÅ OM OM DU KAN"
310 PRINT "BRUGE KROPSDELEN TIL NOGET."
320 PRINT
330 PRINT "JEG GØR DET SAMME NÅR DET ER MIN TUR."
340 PRINT
350 PRINT "<< TAST 'RETURN' FOR AT FORTSETTE >> ";
360 INPUT Z$
370 PRINT
```

```
380 PRINT "HVIS EN AF BILLERNE FÅR EN NY KROPSDEL,"
390 PRINT "HAR DU MULIGHED FOR AT FÅ BILLEN AT SE."
400 PRINT
410 PRINT "ANTALLET AT ØJNE PÅ TERNINGEN SVARER"
420 PRINT "TIL FØLGENDE KROPSDELE:"
430 PRINT
440 PRINT
450 PRINT "TERNING-      KROPS-      ANTAL"
460 PRINT "  ØJNE          DEL          DELE"
470 PRINT "-----      -"
480 PRINT "  1            KROP          1"
490 PRINT "  2            HALS          1"
500 PRINT "  3            HOVED          1"
510 PRINT "  4            FØLEHORN       2"
520 PRINT "  5            HALE           1"
530 PRINT "  6            BEN            6"
540 PRINT
550 PRINT
560 IF Y > 0 THEN 2800
570 Z = INT (6 * RND (1) + 1)
580 C = 1
590 PRINT
600 PRINT "DU SLOG EN";Z;"ER"
610 ON Z GOTO 620,700,810,920,1030,1140
620 PRINT "1 = KROP"
630 IF B = 1 THEN 680
640 PRINT "NU HAR DU EN KROP."
650 B = 1
660 C = 0
670 GOTO 1240
680 PRINT "DU HAR IKKE BRUG FOR EN KROP."
690 GOTO 1240
700 PRINT "2 = HALS"
710 IF N = 1 THEN 770
720 IF B = 0 THEN 790
730 PRINT "NU HAR DU EN HALS."
740 N = 1
750 C = 0
760 GOTO 1240
770 PRINT "DU HAR IKKE BRUG FOR EN HALS."
780 GOTO 1240
790 PRINT "DU HAR JO INGEN KROP."
800 GOTO 1240
810 PRINT "3 = HOVED"
820 IF N = 0 THEN 880
830 IF H = 1 THEN 900
840 PRINT "DU HAR BRUG FOR ET HOVED."
850 H = 1
860 C = 0
870 GOTO 1240
880 PRINT "DU HAR JO INGEN HALS."
890 GOTO 1240
900 PRINT "DU HAR ALLEREDE ET HOVED."
910 GOTO 1240
920 PRINT "4 = FØLEHORN"
930 IF H = 0 THEN 1010
940 IF A = 2 THEN 990
950 PRINT "NU FÅR DU ET FØLEHORN."
960 A = A + 1
970 C = 0
980 GOTO 1240
990 PRINT "DU HAR ALLEREDE TO FØLEHORN."
1000 GOTO 1240
1010 PRINT "DU MÅNGLER ET HOVED."
1020 GOTO 1240
1030 PRINT "5 = HALE"
1040 IF B = 0 THEN 1100
1050 IF T = 1 THEN 1120
1060 PRINT "NU FÅR DU EN HALE."
1070 T = T + 1
1080 C = 0
1090 GOTO 1240
1100 PRINT "DU HAR INGEN KROP."
1110 GOTO 1240
1120 PRINT "DU HAR ALLEREDE EN HALE."
1130 GOTO 1240
1140 PRINT "6 = BEN"
1150 IF L = 6 THEN 1210
1160 IF B = 0 THEN 1230
1170 L = L + 1
1180 C = 0
1190 PRINT "NU HAR DU";L;"BEN."
1200 GOTO 1240
1210 PRINT "DU HAR 6 BEN I FORVEJEN."
1220 GOTO 1240
1230 PRINT "DU HAR INGEN KROP."
1240 X = INT (6 * RND (1) + 1)
1250 PRINT
```

```

1260 PRINT "JEG SLOG EN";X;"ER."
1270 ON X GOTO 1280,1360,1470,1580,1690,1800
1280 PRINT "1 = KROP"
1290 IF P = 1 THEN 1340
1300 PRINT "NU HAR JEG EN KROP."
1310 C = 0
1320 P = 1
1330 GOTO 1640
1340 PRINT "JEG HAR IKKE BRUG FOR EN KROP."
1350 GOTO 1910
1360 PRINT "2 = HALS"
1370 IF Q = 1 THEN 1430
1380 IF P = 0 THEN 1450
1390 PRINT "NU HAR JEG EN HALS."
1400 Q = 1
1410 C = 0
1420 GOTO 1910
1430 PRINT "JEG HAR IKKE BRUG FOR EN HALS."
1440 GOTO 1910
1450 PRINT "JEG HAR INGEN KROP."
1460 GOTO 1910
1470 PRINT "3 = HOVED"
1480 IF Q = 0 THEN 1540
1490 IF R = 1 THEN 1560
1500 PRINT "JEG MANGLEDE NETOP ET HOVED."
1510 R = 1
1520 C = 0
1530 GOTO 1910
1540 PRINT "JEG HAR INGEN HALS."
1550 GOTO 1910
1560 PRINT "JEG HAR IKKE BRUG FOR ET HOVED."
1570 GOTO 1910
1580 PRINT "4 = FØLEHORN"
1590 IF R = 0 THEN 1670
1600 IF S = 2 THEN 1650
1610 PRINT "JEG FÅR ET FØLEHORN."
1620 S = S + 1
1630 C = 0
1640 GOTO 1910
1650 PRINT "JEG HAR ALLEREDE TO FØLEHORN."
1660 GOTO 1910
1670 PRINT "JEG HAR IKKE NOGET HOVED."
1680 GOTO 1910
1690 PRINT "5 = HALE"
1700 IF P = 0 THEN 1760
1710 IF U = 1 THEN 1780
1720 PRINT "NU HAR JEG EN HALE."
1730 U = 1
1740 C = 0
1750 GOTO 1910
1760 PRINT "JEG HAR INGEN KROP."
1770 GOTO 1910
1780 PRINT "JEG HAR IKKE BRUG FOR EN HALE."
1790 GOTO 1910
1800 PRINT "6 = BEN"
1810 IF V = 6 THEN 1870
1820 IF P = 0 THEN 1890
1830 V = V + 1
1840 C = 0
1850 PRINT "NU HAR JEG";V;"BEN."
1860 GOTO 1910
1870 PRINT "JEG HAR 6 BEN."
1880 GOTO 1910
1890 PRINT "JEG HAR INGEN KROP."
1900 GOTO 1910
1910 IF A = 2 AND T = 1 AND L = 6 THEN 1930
1920 GOTO 1950
1930 PRINT "DIN BILLE ER TEGNET FÆRDIG."
1940 Y = Y + 1
1950 IF S = 2 AND P = 1 AND V = 6 THEN 1970
1960 GOTO 1990
1970 PRINT "MIN BILLE ER TEGNET FÆRDIG."
1980 Y = Y + 2
1990 IF C = 1 THEN 560
2000 PRINT
2010 PRINT "ØNSKER DU AT SE BILLERNE ";
2020 INPUT Z$
2030 PRINT
2040 PRINT
2050 IF LEFT$(Z$,1) = "N" THEN 560
2060 PRINT "***** DIN BILLE *****"
2070 PRINT
2080 PRINT
2090 IF A = 0 THEN 2140
2100 PRINT "      0  0"
2110 PRINT "      ) ("
2120 PRINT "      !  !"
2130 PRINT "      !  !"
2140 IF H = 0 THEN 2160
2150 GOSUB 2720
2160 IF N = 0 THEN 2200
2170 FOR Z = 1 TO 2
2180 PRINT "          H H"
2190 NEXT Z
2200 IF B = 0 THEN 2280
2210 PRINT "      KKKKKKKKKKK"
2220 FOR Z = 1 TO 2
2230 PRINT "          K          K"
2240 NEXT Z
2250 IF T < > 1 THEN 2270
2260 PRINT "====K          K"
2270 PRINT "      KKKKKKKKKKK"
2280 IF L = 0 THEN 2360
2290 FOR Z = 1 TO 2
2300 PRINT "          ";
2310 FOR X = 1 TO L
2320 PRINT " L";
2330 NEXT X
2340 PRINT
2350 NEXT Z
2360 FOR Z = 1 TO 4
2370 PRINT
2380 NEXT Z
2390 PRINT "***** MIN BILLE *****"
2400 PRINT
2410 PRINT
2420 IF S = 0 THEN 2470
2430 PRINT "          0  0"
2440 PRINT "          ) ("
2450 PRINT "          !  !"
2460 PRINT "          !  !"
2470 IF R < > 1 THEN 2490
2480 GOSUB 2720
2490 IF Q = 0 THEN 2520
2500 PRINT "          H H"
2510 PRINT "          H H"
2520 IF P = 0 THEN 2600
2530 PRINT "      KKKKKKKKKKK"
2540 FOR Z = 1 TO 2
2550 PRINT "          K          K"
2560 NEXT Z
2570 IF U < > 1 THEN 2590
2580 PRINT "====K          K"
2590 PRINT "      KKKKKKKKKKK"
2600 IF V = 0 THEN 2680
2610 FOR Z = 1 TO 2
2620 PRINT "          ";
2630 FOR X = 1 TO V
2640 PRINT " L";
2650 NEXT X
2660 PRINT
2670 NEXT Z
2680 IF Y < > 0 THEN 2800
2690 PRINT
2700 PRINT
2710 GOTO 560
2720 PRINT "          HHHHH"
2730 PRINT "          H  H"
2740 PRINT "          H - - H"
2750 PRINT "          H O O H"
2760 PRINT "          H !  H"
2770 PRINT "          H --- H"
2780 PRINT "          HHHHH"
2790 RETURN
2800 PRINT
2810 PRINT "JEG HÅBER DU KUNNE LIDE SPILLET....."
2820 PRINT
2830 PRINT "LAD OS SPILLE DET IGEN SNART!"
2840 END

```

Bogstav

Spillet Bogstav minder meget om spillet Gæt, hvor du skal gætte et tal, men her skal du gætte et bogstav. Datamaten vælger et tilfældigt bogstav fra alfabetet, og du skal gætte, hvilket det er ud fra de ledetråde, som du får i spillets forløb. Du bør ikke bruge flere end fem gæt for at finde det hemmelige bogstav.

Programmet ligner et program, som oprindeligt blev skrevet af Bob Albrecht fra People's Computer Company.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
BOGSTAV  
-----
```

I DETTE SPIL SKAL DU GÆTTE BOGSTAVER.

JEG VIL TÆNKE PÅ ET AF ALFABETETS BOG-
STAVER, FRA A-Z.

PRØV AT GÆTTE DET BOGSTAV JEG TÆNKER
PÅ, OG JEG VIL GIVE DIG NOGLE TIPS,
DER FORTÆLLER, HVOR TÆT DU ER PÅ
BOGSTAVET.

OK, JEG TÆNKER PÅ ET BOGSTAV.....
DU KAN GODT BEGYNDE AT GÆTTE.

HVILKET BOGSTAV GÆTTER DU PÅ ?L

FOR LAVT, PRØV ET BOGSTAV DER LIGGER
HØJERE OPPE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?W

FOR HØJT, PRØV ET BOGSTAV DER LIGGER
TIDLIGERE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?M

FOR LAVT, PRØV ET BOGSTAV DER LIGGER
HØJERE OPPE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?P

FOR LAVT, PRØV ET BOGSTAV DER LIGGER
HØJERE OPPE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?R

FOR LAVT, PRØV ET BOGSTAV DER LIGGER
HØJERE OPPE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?T

DU RAMTE RIGTIGT PÅ 6 GÆT!
MEN DET BØR IKKE TAGES MERE END 5 GÆT..

LAD OS SPILLE IGEN.....

OK, JEG TÆNKER PÅ ET BOGSTAV.....
DU KAN GODT BEGYNDE AT GÆTTE.

HVILKET BOGSTAV GÆTTER DU PÅ ?M

FOR HØJT, PRØV ET BOGSTAV DER LIGGER
TIDLIGERE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?F

FOR LAVT, PRØV ET BOGSTAV DER LIGGER
HØJERE OPPE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?I

FOR HØJT, PRØV ET BOGSTAV DER LIGGER
TIDLIGERE I ALFABETET.

HVILKET BOGSTAV GÆTTER DU PÅ ?H

DU RAMTE RIGTIGT PÅ 4 GÆT!
VIRKELIG GODT KLARET!

LAD OS SPILLE IGEN.....

OK, JEG TÆNKER PÅ ET BOGSTAV.....
DU KAN GODT BEGYNDE AT GÆTTE.

HVILKET BOGSTAV GÆTTER DU PÅ ?

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          BOGSTAV"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "I DETTE SPIL SKAL DU GÆTTE BOGSTAVER."  
160 PRINT  
170 PRINT "JEG VIL TÆNKE PÅ ET AF ALFABETETS BOG-"  
180 PRINT "STAVER, FRA A-Z."  
190 PRINT  
200 PRINT "PRØV AT GÆTTE DET BOGSTAV JEG TÆNKER"  
210 PRINT "PÅ, OG JEG VIL GIVE DIG NOGLE TIPS,"  
220 PRINT "DER FORTÆLLER, HVOR TÆT DU ER PÅ"  
230 PRINT "BOGSTAVET."  
240 L = 65 + INT ( RND (1) * 26)  
250 G = 0  
260 PRINT  
270 PRINT "OK, JEG TÆNKER PÅ ET BOGSTAV....."  
280 PRINT "DU KAN GODT BEGYNDE AT GÆTTE."  
290 PRINT  
300 PRINT "HVILKET BOGSTAV GÆTTER DU PÅ ?";  
310 G = G + 1  
320 INPUT A$  
330 A = ASC (A$)  
340 PRINT  
350 IF A = L THEN 430  
360 IF A > L THEN 400  
370 PRINT "FOR LAVT, PRØV ET BOGSTAV DER LIGGER"  
380 PRINT "HØJERE OPPE I ALFABETET."  
390 GOTO 290  
400 PRINT "FOR HØJT, PRØV ET BOGSTAV DER LIGGER"  
410 PRINT "TIDLIGERE I ALFABETET."  
420 GOTO 290  
430 PRINT  
440 PRINT "DU RAMTE RIGTIGT PÅ";G;"GÆT!"  
450 IF G < = 5 THEN 480  
460 PRINT "MEN DET BØR IKKE TAGES MERE END 5 GÆT.."  
470 GOTO 490  
480 PRINT "VIRKELIG GODT KLARET!"  
490 PRINT  
500 PRINT "LAD OS SPILLE IGEN....."  
510 GOTO 240  
520 END
```

Boksning

Dette program simulerer en tre-runders olympisk boksekamp. Datamaten er træner for en af bokserne og bestemmer hans slag og parader, mens du gør det samme for din egen bokser. Ved kampens start har du mulighed for at bestemme, hvilke slag din bokser er god til, og hvilke han er dårlig til. Der bliver omkring syv virkelige slag pr. runde, men dette kan let ændres i programlinje 390. Den, der er bedst i to af tre runder, vinder.

Programmet er forfattet af Jesse Lynch fra St. Paul, Minnesota.

```
MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                        BOKSNING
-----
                        BOKSNING EFTER DE OLYMPISKE REGLER:
-----
3 RUNDER, DEN DER SEJRER SKAL VINDE 2 RUNDER UD AF DE 3

HVAD HEDDER DIN MODSTANDER ?MORTENSEN

HVAD HEDDER DU SELV ?DRÆBERJENSEN

DER ER FIRE FORSKELLIGE FORMER FOR STØD:

1=FULL SWING      2=HOOK      3=UPPERCUT      4=LIGE VENSTRE

HVAD ER DU BEDST TIL ?2

HVAD ER DU DÅRLIG TIL ?1

MORTENSEN ER BEDST TIL 2 .
HANS SVAGHEDER ER EN HEMMELIGHED.

RUNDE 1 BEGYNDER.....
MORTENSEN PRØVER MED ET FULL SWING OG...
POW!!!      RAMMER HAM LIGE I FJÆSET!

DRÆBERJENSEN'S STØD ?2
DRÆBERJENSEN SLÅR ET HOOK... OG RAMMER!

DRÆBERJENSEN'S STØD ?3
DRÆBERJENSEN FØRSØGER EN UPPERCUT OG DEN PARERES VED ET LYKKETRÆF.
MORTENSEN SLÅR EN LIGE VENSTRE, BLODET FLYDER!!!

DRÆBERJENSEN'S STØD ?3
DRÆBERJENSEN FØRSØGER EN UPPERCUT OG DEN PARERES VED ET LYKKETRÆF.

DRÆBERJENSEN'S STØD ?2
DRÆBERJENSEN SLÅR ET HOOK... OG RAMMER!
MORTENSEN PRØVER MED ET FULL SWING OG...
POW!!!      RAMMER HAM LIGE I FJÆSET!
MORTENSEN VINDER RUNDE 1 .

RUNDE 2 BEGYNDER.....

DRÆBERJENSEN'S STØD ?2
DRÆBERJENSEN SLÅR ET HOOK... OG RAMMER!
MORTENSEN PRØVER MED ET FULL SWING OG...
POW!!!      RAMMER HAM LIGE I FJÆSET!
DRÆBERJENSEN ANGRIBES MED EN UPPERCUT (RH, NEJ)..
PARERER OG RAMMER MORTENSEN MED ET HOOK.

DRÆBERJENSEN'S STØD ?4
DRÆBERJENSEN SLÅR EN LIGE VENSTRE MOD MORTENSEN'S HOVED
MORTENSEN SLÅR EN LIGE VENSTRE, BLODET FLYDER!!!
DRÆBERJENSEN ANGRIBES MED EN UPPERCUT (RH, NEJ)..
PARERER OG RAMMER MORTENSEN MED ET HOOK.
MORTENSEN RAMMER DRÆBERJENSEN PÅ KÆBEN. (AVVV!)..
....OG EN GANG TIL!

DRÆBERJENSEN ANGRIBES MED EN UPPERCUT (RH, NEJ)..
PARERER OG RAMMER MORTENSEN MED ET HOOK.
MORTENSEN VINDER RUNDE 2 .

MORTENSEN VINDER ----- GODT KLARET, MORTENSEN
```

OG HERMED SIGER VI FARVEL FRA DET OLYMPISKE STADION

```
100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                        BOKSNING"
130 PRINT "-----"
140 PRINT
150 PRINT "                        BOKSNING EFTER DE OLYMPISKE REGLER:"
160 PRINT "                        -----"
170 PRINT
180 PRINT "      3 RUNDER, DEN DER SEJRER SKAL VINDE 2 RUNDER UD AF DE 3"
190 PRINT
200 PRINT
210 PRINT "HVAD HEDDER DIN MODSTANDER ";
220 INPUT J$
230 PRINT
240 PRINT "HVAD HEDDER DU SELV ";
250 INPUT L$
260 PRINT
270 PRINT "DER ER FIRE FORSKELLIGE FORMER FOR STØD:"
280 PRINT
290 PRINT "1=FULL SWING      2=HOOK      3=UPPERCUT      4=LIGE VENSTRE"
300 PRINT
310 PRINT "HVAD ER DU BEDST TIL ";
320 INPUT B
330 PRINT
340 PRINT "HVAD ER DU DÅRLIG TIL ";
350 INPUT D
360 PRINT
370 J = 0
380 L = 0
390 W = 7
400 B1 = INT (4 * RND (1) + 1)
410 D1 = INT (4 * RND (1) + 1)
420 IF B1 = D1 THEN 400
430 PRINT J$" ER BEDST TIL";B1;". "
440 PRINT "HANS SVAGHEDER ER EN HEMMELIGHED."
450 FOR R = 1 TO 3
460 IF J > = 2 THEN 1650
470 IF L > = 2 THEN 1680
480 X = 0
490 Y = 0
500 PRINT
510 PRINT "RUNDE";R;"BEGYNDER....."
520 FOR R1 = 1 TO W
530 I = INT (10 * RND (1) + 1)
540 IF I > 5 THEN 1040
550 PRINT
560 PRINT L$;"'S STØD ";
570 INPUT P
580 IF P = B THEN 600
590 GOTO 610
600 X = X + 2
610 IF P = 1 THEN 720
620 IF P = 2 THEN 860
630 IF P = 3 THEN 950
640 PRINT L$;" SLÅR EN LIGE VENSTRE MOD ";J$"'S HOVED "
650 IF D1 = 4 THEN 680
660 C = INT (8 * RND (1) + 1)
670 IF C < 4 THEN 700
680 X = X + 3
690 GOTO 1460
700 PRINT "DEN PARERES."
710 GOTO 1460
720 PRINT L$;" SLÅR ET FULL SWING OG ";
730 IF D1 = 4 THEN 820
740 X3 = INT (30 * RND (1) + 1)
750 IF X3 < 10 THEN 820
760 PRINT "RAMMER VED SIDEN AF.."
770 PRINT
780 IF X = 1 THEN 1460
790 PRINT
800 PRINT
810 GOTO 690
820 PRINT "OG RAMMER!"
830 IF X > 35 THEN 1570
840 X = X + 15
850 GOTO 690
860 PRINT L$;" SLÅR ET HOOK... ";
870 IF D1 = 2 THEN 910
```

```


880 H1 = INT (2 * RND (1) + 1)
890 IF H1 = 1 THEN 930
900 PRINT "OG RAMMER!"
910 X = X + 7
920 GOTO 690
930 PRINT "MEN DER PARERES!!!!!!!!!!!!"
940 GOTO 690
950 PRINT L$;" FØRSØGER EN UPPERCUT ";
960 IF D1 = 3 THEN 1010
970 D5 = INT (100 * RND (1) + 1)
980 IF D5 < 51 THEN 1010
990 PRINT "OG DEN PARERES VED ET LYKKETRÆF."
1000 GOTO 690
1010 PRINT "OG HAN RAMMER!"
1020 X = X + 4
1030 GOTO 690
1040 J7 = INT (4 * RND (1) + 1)
1050 IF J7 = 81 THEN 1070
1060 GOTO 1080
1070 Y = Y + 2
1080 IF J7 = 1 THEN 1200
1090 IF J7 = 2 THEN 1300
1100 IF J7 = 3 THEN 1360
1110 PRINT J$;" SLÅR EN LIGE VENSTRE, ";
1120 IF D = 4 THEN 1180
1130 Z4 = INT (7 * RND (1) + 1)
1140 IF Z4 > 4 THEN 1170
1150 PRINT "DEN BLOKERES!"
1160 GOTO 690
1170 PRINT "BLODET FLYDER!!!"
1180 Y = Y + 5
1190 GOTO 690
1200 PRINT J$;" PRØVER MED ET FULL SWING OG..."
1210 IF D = 1 THEN 1260
1220 R6 = INT (60 * RND (1) + 1)
1230 IF R6 < 30 THEN 1260
1240 PRINT "MEN DEN AFVÆRGES!"
1250 GOTO 690
1260 PRINT "POW!!! RAMMER HAM LIGE I FJÆSET!"
1270 IF Y > 35 THEN 1610
1280 Y = Y + 15
1290 GOTO 690
1300 PRINT J$;" RAMMER ";L$;" PÅ KÆBEN. (AVVV!).."
1310 Y = Y + 7
1320 PRINT ".....OG EN GANG TIL!"

```

```

1330 Y = Y + 5
1340 IF Y > 35 THEN 1610
1350 PRINT
1360 PRINT L$;" ANGRIBES MED EN UPPERCUT (ÅH, NEJ!).."
1370 IF D = 3 THEN 1400
1380 Q4 = INT (200 * RND (1) + 1)
1390 IF Q4 > 75 THEN 1430
1400 PRINT " OG ";J$;" RAMMER....."
1410 Y = Y + 8
1420 GOTO 690
1430 PRINT "PARERER OG RAMMER ";J$;" MED ET HOOK.."
1440 X = X + 5
1450 GOTO 690
1460 NEXT R1
1470 IF X > Y THEN 1510
1480 PRINT J$;" VINDER RUNDE";R;".."
1490 J = J + 1
1500 GOTO 1530
1510 PRINT L$;" VINDER RUNDE";R;".."
1520 L = L + 1
1530 PRINT
1540 NEXT R
1550 IF J > = 2 THEN 1650
1560 IF L > = 2 THEN 1680
1570 PRINT
1580 PRINT "DER TÆLLES OVER ";J$;".."
1590 PRINT L$;" ER VINDER OG CHAMPION.>";
1600 GOTO 1710
1610 PRINT
1620 PRINT "DER TÆLLES OVER ";L$;".."
1630 PRINT J$;" ER VINDER OG CHAMPION.>";
1640 GOTO 1600
1650 PRINT
1660 PRINT J$;" VINDER ----- GODT KLARET, ";J$
1670 GOTO 1600
1680 PRINT
1690 PRINT L$;" VINDER OVERRASKENDE.."
1700 GOTO 1600
1710 PRINT
1720 PRINT "....."
1730 PRINT
1740 PRINT " OG HERMED SIGER VI FARVEL FRA DET OLYMPISKE STADION"
1750 PRINT "....."
1760 PRINT
1770 END

```



Bombardement

Bombardement spilles på to spillebrætter, der måler 5 gange 5 felter, ialt 25 felter nummereret fra 1 til 25. Både du og datamaten har fire delinger soldater, der frit kan grupperes på de 25 felter på brættet.

Når spillet begynder, grupperer (eller gemmer du) dine delinger på dit eget spillebræt. Datamaten gør det samme på sit spillebræt. Derefter skiftes du og datamaten til at affyre raketter eller bomber mod modstanderens stillinger i et forsøg på at tilintetgøre alle fire delinger. Den, som tilintetgør modpartens delinger først, har vundet.

Dette program er en noget modificeret udgave af et program, oprindeligt skrevet af Martin Burdash fra Parlin, New Jersey.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BOMBARDEMENT

DU BEFINDER DIG PÅ EN SLAGMARK, OG DU HAR 4 KOMPAGNIER TIL RÅDIGHED. DISSE 4 KOMPAGNIER KAN DU PLACERE PÅ 25 FORSKELLIGE FORPOSTER.

DU KAN DOG KUN PLACERE ET KOMPAGNI PÅ HVER FORPOST.

COMPUTEREN FORDELER OGSÅ SINE 4 KOMPAGNIER.

SPILETS FORMÅL ER, AT DU SKAL AFFYRE MISSILER MOD COMPUTERENS FORPOSTER, I HÅB OM AT RAMME KOMPAGNIERNE. COMPUTEREN AFFYRER OGSÅ MISSILER MOD DINE FORPOSTER.

DEN DER FØRST DESTRUEERER MODSTANDERENS KOMPAGNIER HAR VUNDET.

HELD OG LYKKE. OG GLEM IKKE AT OPLYSE, HVOR KISTERNE SKAL SENDES HEN!

RIV SPILLEPLANEN UD, SÅ DU KAN HOLDE REGNSKAB MED FORPOSTERNE.

1	2	3	4	5
6	7	8	9	10
11	12	13	14	15
16	17	18	19	20
21	22	23	24	25

INDTASTE DE FIRE FORPOSTER, DER ØNSKES BESAT ?10,9,16,24

MISSILET SKAL RETTES MOD FORPOST ?3

HI-HI, FORBIER... SÅ ER DET MIN TUR!

JEG RAMTE VED SIDEN AF. DIN HUND! JEG SKØD PÅ FORPOST 22 . DIN TUR!

MISSILET SKAL RETTES MOD FORPOST ?7

DU RAMTE EN BEMANDET FORPOST!

EN RAMT, TRE TILBAGE

JEG RAMTE VED SIDEN AF. DIN HUND! JEG SKØD PÅ FORPOST 2 . DIN TUR!

MISSILET SKAL RETTES MOD FORPOST ?6

HI-HI, FORBIER... SÅ ER DET MIN TUR!

JEG RAMTE VED SIDEN AF. DIN HUND! JEG SKØD PÅ FORPOST 23 . DIN TUR!

MISSILET SKAL RETTES MOD FORPOST ?8

DU RAMTE EN BEMANDET FORPOST!

TO RAMT, TO TILBAGE.

JEG FIK DIG. NU VARER DET IKKE LÆNGE.. FORPOST 24 BLEV RAMT. HA-HA-HA. DU HAR KUN 3 FORPOSTER TILBAGE!!!!

MISSILET SKAL RETTES MOD FORPOST ?19

DU RAMTE EN BEMANDET FORPOST!

TRE RAMT, KUN EN ENKELT TILBAGE!!

JEG RAMTE VED SIDEN AF. DIN HUND! JEG SKØD PÅ FORPOST 14 . DIN TUR!

MISSILET SKAL RETTES MOD FORPOST ?25

HI-HI, FORBIER... SÅ ER DET MIN TUR!

JEG RAMTE VED SIDEN AF. DIN HUND! JEG SKØD PÅ FORPOST 15 . DIN TUR!

MISSILET SKAL RETTES MOD FORPOST ?14

NU FIK DU MIG. ENDEN ER NÆR. MEN JEG SKAL NOK FÅ REVANCE.....SÅ SNART MINE TRANSISTORER ER LADET OP IGEN!!!!!!

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                BOMBARDEMENT"
130 PRINT "-----"
140 DIM M(100)
150 PRINT
160 PRINT "DU BEFINDER DIG PÅ EN SLAGMARK, OG DU"
170 PRINT "HAR 4 KOMPAGNIER TIL RÅDIGHED. DISSE 4"
180 PRINT "KOMPAGNIER KAN DU PLACERE PÅ 25 FOR-"
190 PRINT "SKELLIGE FORPOSTER."
200 PRINT
210 PRINT "DU KAN DOG KUN PLACERE ET KOMPAGNI PÅ"
220 PRINT "HVER FORPOST."
230 PRINT
240 PRINT "COMPUTEREN FORDELER OGSÅ SINE 4 KOMPAG-"
250 PRINT "NIER."
260 PRINT
270 PRINT "SPILETS FORMÅL ER, AT DU SKAL AFFYRE"
280 PRINT "MISSILER MOD COMPUTERENS FORPOSTER, I"
290 PRINT "HÅB OM AT RAMME KOMPAGNIERNE. COMPUTE-"
300 PRINT "REN AFFYRER OGSÅ MISSILER MOD DINE FOR-"
310 PRINT "POSTER."
320 PRINT
330 PRINT "DEN DER FØRST DESTRUEERER MODSTANDERENS"
340 PRINT "KOMPAGNIER HAR VUNDET."
350 PRINT
360 PRINT "HELD OG LYKKE. OG GLEM IKKE AT OPLYSE,"
370 PRINT "HVOR KISTERNE SKAL SENDES HEN!"
380 PRINT
390 PRINT "RIV SPILLEPLANEN UD, SÅ DU KAN HOLDE"
400 PRINT "REGNSKAB MED FORPOSTERNE."
410 PRINT
420 PRINT
430 PRINT " 1   2   3   4   5"
440 PRINT " 6   7   8   9  10"
450 PRINT "11  12  13  14  15"
460 PRINT "16  17  18  19  20"
470 PRINT "21  22  23  24  25"
480 PRINT
490 PRINT
500 C = INT ( RND (1) * 25) + 1
510 D = INT ( RND (1) * 25) + 1
520 E = INT ( RND (1) * 25) + 1
530 F = INT ( RND (1) * 25) + 1
540 IF C = D THEN 510
550 IF C = E THEN 520
560 IF C = F THEN 530
570 IF D = E THEN 520
580 IF D = F THEN 530
590 IF E = F THEN 530
600 PRINT "INDTASTE DE FIRE FORPOSTER, DER ØNSKES"
610 PRINT "BESAT ";
620 INPUT G,H,K,L
630 PRINT
640 PRINT "MISSILET SKAL RETTES MOD FORPOST ";
650 INPUT Y
660 PRINT
670 IF Y = C THEN 860
680 IF Y = D THEN 860
690 IF Y = E THEN 860
700 IF Y = F THEN 860
710 GOTO 790
720 M = INT ( RND (1) * 25) + 1
730 GOTO 1200
740 IF X = G THEN 1030
750 IF X = H THEN 1030
760 IF X = L THEN 1030
770 IF X = K THEN 1030
780 GOTO 820
790 PRINT "HI-HI, FORBIER... SÅ ER DET MIN TUR!"
800 PRINT
810 GOTO 720
820 PRINT "JEG RAMTE VED SIDEN AF. DIN HUND!"
830 PRINT "JEG SKØD PÅ FORPOST";M;". DIN TUR!"
840 PRINT
850 GOTO 640
860 Q = Q + 1
870 IF Q = 4 THEN 990
880 PRINT "DU RAMTE EN BEMANDET FORPOST!"
890 PRINT
900 IF Q = 1 THEN 930
910 IF Q = 2 THEN 950
920 IF Q = 3 THEN 970
930 PRINT "EN RAMT, TRE TILBAGE"
940 PRINT : GOTO 720
950 PRINT "TO RAMT, TO TILBAGE."
960 PRINT : GOTO 720
970 PRINT "TRE RAMT, KUN EN ENKELT TILBAGE!!"

```



```

980 PRINT : GOTO 720
990 PRINT "NU FIK DU MIG. ENDEN ER NÆR. MEN JEG"
1000 PRINT "SKAL NOK FR REVANCE.....SR SNART MINE"
1010 PRINT "TRANSISTORER ER LADET OP IGEN!!!!!!"
1020 GOTO 1280
1030 Z = Z + 1
1040 IF Z = 4 THEN 1160
1050 PRINT "JEG FIK DIG. NU VARER DET IKKE LÆNGE.."
1060 PRINT "FORPOST";X;"BLEV RAMT. HA-HA-HA."
1070 IF Z = 1 THEN 1100
1080 IF Z = 2 THEN 1120
1090 IF Z = 3 THEN 1140
1100 PRINT "DU HAR KUN 3 FORPOSTER TILBAGE!!!!!"
1110 PRINT : GOTO 640
1120 PRINT "DU HAR NU KUN 2 FORPOSTER TILBAGE!!!"
1130 PRINT : GOTO 640
1140 PRINT "DU HAR KUN EN ENESTE FORPOST TILBAGE!!!"
1150 PRINT : GOTO 640
1160 PRINT "DU ER DØD!! DIN SIDSTE FORPOST BEFANDT"
1170 PRINT "SIG PÅ POSITION";X;" HAR-HAR-HAR-HAR-HAR."
1180 PRINT "BEDRE HELD NÆSTE GANG."
1190 GOTO 1280
1200 P = P + 1
1210 N = P - 1
1220 FOR T = 1 TO N
1230 IF M = M(T) THEN 720
1240 NEXT T
1250 X = M
1260 M(P) = M
1270 GOTO 740
1280 END

```

Bondespil

Bondespillet og en metode til indlæring af spillets strategi blev første gang beskrevet af Martin Gardner i artiklen »Mathematical Games« i marts-nummeret af »Scientific America«, 1962. Metoden, som denne artikel beskrev, drejede sig især om en hypotetisk indlæringsmaskine, som bestod af tændstikæsker og farvede glaskugler. Dette primitive spil er senere blevet generaliseret til det, vi i dag kender som Bondespillet.

Programmet lærer efterhånden at spille spillet, idet det fjerner dårlige træk. Alle de positioner og træk, som programmet godkender, lagres i variabel P\$(I). Når programmet møder en ukendt position, så lægges denne, og alle de træk der kan foretages fra den, ind i listen. Hvis programmet taber et spil, så fjernes alle de træk, som afstedkom denne situation fra listen. Hvis datamaten støder på en position, hvor alle træk er slettet (fordi de førte til, at spillet blev tabt), så sletter den også de træk, som førte brikken derhen og opgiver spillet. Efterhånden lærer programmet at spille bedre og bedre, og til slut kan man ikke længere vinde over datamaten. Denne indlæringsstrategi kan tillempes på alle andre spil, der kun indeholder et begrænset antal træk (f.eks. kryds-og-bolle, damspil på et lille bræt, og andre mindre spil der er baseret på skak).

Den oprindelige version er skrevet af R.A. Kaapke. Det blev senere modificeret af Jeff Dalton og til slut af Steve North fra Creative Computing.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BONDESPIL

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DETTE PROGRAM SPILLER BONDESPILLET, DER SPILLES MED BØNDERNE FRA SKAKSPILLET PÅ ET SPILLEBRÆT DER MÅLER 3 * 3 FELTER.

BRIKKERNE FLYTTES SOM I SKAK - ET FELT FREMAD NÅR FELTET ER FRIT, ELLER ET FELT FREMAD DIAGONALT, HVIS DER STÅR EN AF MODSTANDERENS BØNDER.

PÅ BRÆTTET ER DINE BØNDER VIST MED 'O', DATAMATENS ER VIST MED 'X'. TOMME FELT- VISES MED '.'. NÅR DU SKAL FORETAGE ET TRÆK SKAL DU INDTASTE NUMMERET PÅ DET FELT DU FLYTTER FRA, EFTERFULGT AF DET FELT-NUMMER DU FLYTTER TIL. NUMRENE SKAL VÆRE ADSKILDT AF ET KOMMA.

DATAMATEN STARTER HVER SPILLESERIE UDEN AT HAVE NOGEN STRATEGI, DEN TRÆKKER DERFOR TILFÆLDIGT, MEN LÆRER NOGET NYT AF HVERT SPIL DEN SPILLER. DERFOR BLIVER DET MERE OG MERE VANSKELIGT FOR DIG AT VINDE. FOR AT OPVEJSE DET HANDICAP DATAMATEN HAR, FÅR DU DERFOR IKKE AT VIDE HVORDAN MAN VINDER SPILLET.....DET MÅ DU SELV LÆRE VED AT SPILLE DET!

BRÆTTETS FELTER NUMMERERES SÅLEDES:

123
456
789

HVIS DU VIL FLYTTE DIN HØJREBONDE FREMAD, SKAL DU DERFOR INDTASTE '6,9', SOM SVAR PÅ SPØRGSMÅLET 'DIT TRÆK ?' DA JEG ER EN ÆGTE GENTLEMAN, ER DET ALTID DIG DER BEGYNDER.

XXX

000

DIT TRÆK ?7,4

XXX
0..
.00

JEG FLYTTER FRA 2 TIL 4

X.X
X..
.00

DIT TRÆK ?8,4

X.X
0..
..0

JEG FLYTTER FRA 3 TIL 6

X..
0.X
..0

DU KAN IKKE IKKE FLYTTE,
SÅ JEG VINDER.

JEG HAR VUNDET 1 SPIL, MENS
DU HAR VUNDET 0 SPIL

XXX

000

DIT TRÆK ?9,6

XXX
..0
00.

JEG FLYTTER FRA 2 TIL 5

X.X
.X0
00.

DIT TRÆK ?7,5

X.X
.00
.0.

JEG FLYTTER FRA 1 TIL 4

..X
X00
.0.

DIT TRÆK ?8,4

..X
000

JEG FLYTTER FRA 3 TIL 5

0X0

DIT TRÆK ?6,3

..0
0X.

DU VINDER.

JEG HAR VUNDET 1 SPIL, MENS
DU HAR VUNDET 1 SPIL

XXX

000

DIT TRÆK ?8,5

XXX
.0.
0.0

JEG FLYTTER FRA 1 TIL 5

.XX
.X.
0.0

DIT TRÆK ?9,5

.XX
.0.
0..

JEG FLYTTER FRA 3 TIL 6

.X.
.0X
0..

DIT TRÆK ?7,4

.X.
00X

JEG FLYTTER FRA 2 TIL 4

X0X

DIT TRÆK ?5,2

.0.
X.X

DU VINDER.

JEG HAR VUNDET 1 SPIL, MENS
DU HAR VUNDET 2 SPIL

XXX
...
000

DIT TRÆK ?9,6

XXX
..0
00.

JEG FLYTTER FRA 2 TIL 6

X.X
..X
00.

DIT TRÆK ?8,6

X.X
..0
0..

JEG FLYTTER FRA 1 TIL 4

..X
X.0
0..

DU KAN IKKE IKKE FLYTTE,
SR JEG VINDER.

JEG HAR VUNDET 2 SPIL, MENS
DU HAR VUNDET 2 SPIL

XXX
...
000

DIT TRÆK ?

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          BONDESPIL"
130 PRINT "-----"
140 PRINT
150 REM Bondespillet er en variant af spillet hexa-
160 REM pawn, der blev beskrevet i Martin Gardners
170 REM atikel "The unexpected hanging and other
180 REM mathematical diversions", Kapitel 8, en
190 REM indlæringsmaskine for et tændstiksæksespil
200 REM Den originale version blev skrevet på et
210 REM H-P time-share system af R.A.Kaapke den
220 REM 5/5/1976. Instruktionerne er skrevet af
230 REM Jeff Dalton. Oversættelsen til MITS-BASIC
240 REM er foretaget af Steve North.
250 DIM B(19,9),M(19,4),S(9),P$(3)
260 W = 0
270 L = 0
280 DEF FN S(X) = X * ABS (X = 2 OR X = 5 OR X =
8)
290 DEF FN R(X) = 3 * ABS (X = 1) + ABS (X = 3) +
4 * ABS (X = 6) + 6 * ABS (X = 4) + 7 * ABS (X = 9)
+ 9 * ABS (X = 7) + FN S(X)
300 DEF FN M(Y) = Y - INT (Y / 10) * 10
310 P$ = "X.0"
320 FOR I = 1 TO 19
330 FOR J = 1 TO 9
340 READ B(I,J)
350 NEXT J
360 NEXT I
370 FOR I = 1 TO 19
380 FOR J = 1 TO 4
390 READ M(I,J)
400 NEXT J
410 NEXT I
420 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
430 INPUT A$
440 A$ = LEFT$(A$,1)
450 IF A$ = "J" THEN 1870
460 IF A$ < > "N" THEN 420
470 X = 0
480 Y = 0
490 S(4) = 0
500 S(5) = 0
510 S(6) = 0
520 S(1) = - 1
530 S(2) = - 1
540 S(3) = - 1
550 S(7) = 1
560 S(8) = 1
570 S(9) = 1
580 GOSUB 1770
590 PRINT "DIT TRÆK ";
600 INPUT M1,M2
610 IF M1 = INT (M1) AND M2 = INT (M2) AND M1 > 0
AND M1 < 10 AND M2 > 0 AND M2 < 10 THEN 640
620 PRINT "ULOVVLIGE KOORDINATER."
630 GOTO 590
640 IF S(M1) = 1 THEN 670
650 PRINT "ULOVVLIGT TRÆK."
660 GOTO 590
670 IF S(M2) = 1 THEN 650
680 IF M2 - M1 < > - 3 AND S(M2) < > - 1 THEN
650
690 IF M2 > M1 THEN 650
700 IF M2 - M1 = - 3 AND (S(M2) < > 0) THEN 650
710 IF M2 - M1 < - 4 THEN 650
720 IF M1 = 7 AND M2 = 3 THEN 650
730 S(M1) = 0
740 S(M2) = 1
750 GOSUB 1770
760 IF S(1) = 1 OR S(2) = 1 OR S(3) = 1 THEN 1510
770 FOR I = 1 TO 9
780 IF S(I) = - 1 THEN 810
790 NEXT I
800 GOTO 1510
810 FOR I = 1 TO 9
820 IF S(I) < > - 1 THEN 910
830 IF S(I + 3) = 0 THEN 930
840 IF FN R(I) = I THEN 900
850 IF I > 3 THEN 880
860 IF S(5) = 1 THEN 930
870 GOTO 910
880 IF S(8) = 1 THEN 930
890 GOTO 910
900 IF S(I + 2) = 1 OR S(I + 4) = 1 THEN 930
910 NEXT I
920 GOTO 1510
```

```

930 FOR I = 1 TO 19
940 FOR J = 1 TO 3
950 FOR K = 3 TO 1 STEP - 1
960 T((J - 1) * 3 + K) = B(I,(J - 1) * 3 + 4 - K)
970 NEXT K
980 NEXT J
990 FOR J = 1 TO 9
1000 IF S(J) < > B(I,J) THEN 1040
1010 NEXT J
1020 R = 0
1030 GOTO 1130
1040 FOR J = 1 TO 9
1050 IF S(J) < > T(J) THEN 1090
1060 NEXT J
1070 R = 1
1080 GOTO 1130
1090 NEXT I
1100 REM Husk at løkken er uendelig
1110 PRINT "ULOVLIGT BRÆT-MØNSTER"
1120 GOTO 2290
1130 X = I
1140 FOR I = 1 TO 4
1150 IF M(X,I) < > 0 THEN 1190
1160 NEXT I
1170 PRINT "JEG OPGIVER.."
1180 GOTO 1510
1190 Y = INT ( RND (1) * 4 + 1)
1200 IF M(X,Y) = 0 THEN 1190
1210 IF R < > 0 THEN 1270
1220 PRINT "JEG FLYTTER FRA"; STR$ ( INT (M(X,Y) /
10));
1230 PRINT "TIL"; STR$ ( FN M(M(X,Y)))
1240 S( INT (M(X,Y) / 10)) = 0
1250 S( FN M(M(X,Y))) = - 1
1260 GOTO 1310
1270 PRINT "JEG FLYTTER FRA"; STR$ ( FN R( INT
(M(X,Y) / 10));
1280 PRINT "TIL"; STR$ ( FN R( FN M(M(X,Y))))
1290 S( FN R( INT (M(X,Y) / 10))) = 0
1300 S( FN R( FN M(M(X,Y)))) = - 1
1310 GOSUB 1770
1320 IF S(7) = - 1 OR S(8) = - 1 OR S(9) = - 1
THEN 1590
1330 FOR I = 1 TO 9
1340 IF S(I) = 1 THEN 1370
1350 NEXT I
1360 GOTO 1590
1370 FOR I = 1 TO 9
1380 IF S(I) < > 1 THEN 1470
1390 IF S(I - 3) = 0 THEN 590
1400 IF FN R(I) = I THEN 1460
1410 IF I < 7 THEN 1440
1420 IF S(5) = - 1 THEN 590
1430 GOTO 1470
1440 IF S(2) = - 1 THEN 590
1450 GOTO 1470
1460 IF S(I - 2) = - 1 OR S(I - 4) = - 1 THEN 590
1470 NEXT I
1480 PRINT "DU KAN IKKE IKKE FLYTTE,"
1490 PRINT "SR ";
1500 GOTO 1590
1510 PRINT "DU VINDER."
1520 M(X,Y) = 0
1530 L = L + 1
1540 PRINT
1550 PRINT "JEG HAR VUNDET";W;"SPIL, MENS"
1560 PRINT " DU HAR VUNDET";L;"SPIL"
1570 PRINT
1580 GOTO 470
1590 PRINT "JEG VINDER."
1600 W = W + 1
1610 GOTO 1540
1620 DATA -1,-1,-1,1,0,0,0,1,1,-1,-1,-1,0,1,0,1
1630 DATA -1,0,-1,-1,1,0,0,0,1,0,-1,-1,1,-1,0,0,0,1
1640 DATA -1,0,-1,1,1,0,0,1,0,-1,-1,0,1,0,1,0,0,1
1650 DATA 0,-1,-1,0,-1,1,1,0,0,0,-1,-1,-1,1,1,0,0,0
1660 DATA -1,0,-1,-1,0,1,0,1,0,0,0,-1,-1,0,1,0,0,0,1
1670 DATA 0,-1,-1,0,1,0,1,0,0,-1,0,-1,1,0,0,0,0,1
1680 DATA 0,0,-1,-1,-1,1,0,0,0,-1,0,0,1,1,1,0,0,0,0
1690 DATA 0,-1,0,-1,1,1,0,0,0,-1,0,0,-1,-1,1,0,0,0,0
1700 DATA 0,0,-1,-1,1,0,0,0,0,-1,0,1,-1,0,0,0,0,0
1710 DATA -1,0,0,-1,1,0,0,0,0
1720 DATA 24,25,36,0,14,15,36,0,15,35,36,47,36,58
1730 DATA 59,0,15,35,36,0,24,25,26,0,26,57,58,0
1740 DATA 26,35,0,0,47,48,0,0,35,36,0,0,35,36,0,0
1750 DATA 36,0,0,0,47,58,0,0,15,0,0,0
1760 DATA 26,47,0,0,47,58,0,0,35,36,47,0,28,58,0,
0,15,47,0,0
1770 PRINT
1780 FOR I = 1 TO 3
1790 PRINT " ";
1800 FOR J = 1 TO 3
1810 PRINT MID$(P$,S((I - 1) * 3 + J) + 2,1);
1820 NEXT J
1830 PRINT
1840 NEXT I
1850 PRINT
1860 RETURN
1870 PRINT
1880 PRINT "DETTE PROGRAM SPILLER BONDSPILLET, DER"
1890 PRINT "SPILLES MED BØNDERNE FRA SKAKSPILLET PÅ"
1900 PRINT "ET SPILLEBRÆT DER MÅLER 3 * 3 FELTER."
1910 PRINT
1920 PRINT "BRIKKERNE FLYTTES SOM I SKAK - ET FELT"
1930 PRINT "FREMAD NÅR FELTET ER FRIT, ELLER ET"
1940 PRINT "FELT FREMAD DIAGONALT, HVIS DER STÅR EN"
1950 PRINT "AF MODSTANDERENS BØNDER."
1960 PRINT
1970 PRINT "PÅ BRÆTTET ER DINE BØNDER VIST MED 'O',"
1980 PRINT "DATAMATENS ER VIST MED 'X'. TOMME FEL-"
1990 PRINT "VISES MED '.'. NÅR DU SKAL FORETAGE ET"
2000 PRINT "TRÆK SKAL DU INDTASTE NUMMERET PÅ DET"
2010 PRINT "FELT DU FLYTTER FRA, EFTERFULGT AF DET"
2020 PRINT "FELT-NUMMER DU FLYTTER TIL. NUMRENE"
2030 PRINT "SKAL VÆRE ADSKILDT AF ET KOMMA."
2040 PRINT
2050 PRINT "DATAMATEN STARTER HVER SPILLESERIE UDEN"
2060 PRINT "AT HAVE NOGEN STRATEGI, DEN TRÆKKER"
2070 PRINT "DERFOR TILFÆLDIGT, MEN LÆRER NOGET NYT"
2080 PRINT "AF HVERT SPIL DEN SPILLER. DERFOR"
2090 PRINT "BLIVER DET MERE OG MERE VANSKELIGT FOR"
2100 PRINT "DIG AT VINDE."
2110 PRINT "FOR AT OPVEJE DET HANDICAP DATAMATEN"
2120 PRINT "HAR, FÅR DU DERFOR IKKE AT VIDE HVORDAN"
2130 PRINT "MAN VINDER SPILLET.....DET MÅ DU SELV"
2140 PRINT "LÆRE VED AT SPILLE DET!"
2150 PRINT
2160 PRINT "BRÆTTETS FELTER NUMMERERES SÅLEDES:"
2170 PRINT
2180 PRINT TAB( 10);"123"
2190 PRINT TAB( 10);"456"
2200 PRINT TAB( 10);"789"
2210 PRINT
2220 PRINT "HVIS DU VIL FLYTTE DIN HØJREBONDE"
2230 PRINT "FREMAD, SKAL DU DERFOR INDTASTE '6,9',"
2240 PRINT "SOM SVAR PÅ SPØRGSMÅLET 'DIT TRÆK ?'"
2250 PRINT "DA JEG ER EN ÆGTE GENTLEMAN, ER DET"
2260 PRINT "ALTID DIG DER BEGYNDER."
2270 PRINT
2280 GOTO 470
2290 END

```

Borgerkrig

Dette simulationsspil er baseret på 14 udvalgte slag fra den amerikanske borgerkrig. Fakta og tal er hentet fra de virkelige historiske hændelser. Hvis du vælger den samme strategi, som blev anvendt i det oprindelige slag, vil udfaldet også blive det samme. Disse historiske strategier er ofte ganske gode, idet generalerne i borgerkrigen var gode strateger og taktikere. Du kan imidlertid ofte gøre det bedre end generalerne, især i de tilfælde hvor de ikke sad inde med de nødvendige oplysninger og derfor ikke valgte de bedste metoder. Det er klart, at det er en hjælp at kende noget til den amerikanske borgerkrigs historie, men datamaten beskriver det grundlæggende for dig.

Efter hvert af de 14 slag sammenlignes dine tab med de virkelige historiske tab i slaget, ligesom du får at vide, om du har tabt eller vundet.

Du kan også spille Borgerkrig alene. I så fald er datamaten Nordstaterne. Hvis der er to spillere, fungerer datamaten som dommer.

Borgerkrig blev skrevet i 1968 af tre studerende ved Lexington High School, Massachusetts: L. Cram, L. Goodie og D. Hibbard. Det blev modificeret til at kunne bruges af to spillere af G. Paul og R. Hess fra TIES, St. Paul, Minnesota.

```
MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                        BORGERKRIG
-----
```

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DETTE ER EN SIMULATION AF DEN AMERIKANSKE BORGERKRIG

HVIS DU ØNSKER AT SPILLE SKAL DU INDASTE DINE SVAR, NÅR NÅR DATAMATEN BEDER OM DEM. HUSK, AT ALLE FAKTORERNE I SPILLET ER INDBYRDES AFHÆNGIGE, OG AT DE SVAR DU GIVER KAN ÆNDRE HISTORIENS GANG.

SPILLET ER BASERET PÅ FAKTA OG TAL FRA DE VIRKELIGE HÆNDELSER. DE FLESTE SLAG TENDERER MOD AT ENDE LIGESOM DE GJORDE I DEN VIRKELIGE BORGERKRIG, MEN DET ER FØRST OG FREMMEST AFHÆNGIG AF DIN INDSATS!!!!

DIT MÅL ER AT VINDE SÅ MANGE SLAG SOM MULIGT.

DINE FORSVARSTAKTISKE VALGMULIGHEDER ER:

- (1) ARTILLERIILD
- (2) BEFÆSTNINGER MOD FRONTALE ANGREB
- (3) BEFÆSTNINGER MOD FLANKE ANGREB
- (4) TILBAGETRÆKNING

DINE ANGREBSTAKTISKE VALGMULIGHEDER ER:

- (1) ARTILLERiangreb
- (2) FRONTALangreb
- (3) FLANKEangreb
- (4) OMRINGNING

DU KAN OVERGIVE DIG VED AT VÆLGE TAKTISK MULIGHED '5'.

ER DER 2 GENERALER TIL STEDE (SVAR 'J' ELLER 'N') ?J

VÆLG ET SLAG VED AT INDAST ET TAL MELLEM 1 OG 14, NÅR DATAMATEN BEDER OM DETTE. ET HVILKET SOM HELST ANDET TAL AFSLUTTER SPILLET, MENS 'O' BRINGER DIG TILBAGE TIL DET FØRRIGE SLAG, SÅ DU HAR MULIGHED FOR AT SPILLE DET OM.

BEMÆRK: HVIS DU VÆLGER EN NEGATIV VÆRDI FOR FORPLEJNING, ANVENDES DEN SAMME VÆRDI SOM BLEV BRUGT I FØRRIGE SLAG.

NÅR DU HAR VALGT ET SLAG, VIL DU SÅ HAVE EN BESKRIVELSE AF SLAGET (SVAR 'J' ELLER 'N') ?J

HVILKET SLAG ØNSKER DU AT UDKÆMPE ?3

DETTE ER SLAGET OM SEVEN DAYS

FRA DEN 25. JUNI TIL DEN 1. JULI 1862 OPRETHOLDT SYDSTATS-GENERALEN LEE EN OFFENSIV UNDER HELE SLAGET, HVORVED NORDSTATSTROPPERNE UNDER LEDELSE AF GENERAL MCCLELLAN, BLEV HOLDT UDE AF RICHMOND.

	SYDSTATERNE	NORDSTATERNE
MÆND	95000	115000
PENGE	\$ 427500	\$ 517500
INFLATION	25 %	10 %

SYDSTATSGENERAL---

HVOR MANGE PENGE ØNSKER DU AT BRUGE PÅ:

- FORPLEJNING.....?100000
- LØNNINGER.....?140000
- AMMUNITION.....?180000

NORDSTATSGENERAL---

HVOR MANGE PENGE ØNSKER DU AT BRUGE PÅ:

- FORPLEJNING.....?120000
- LØNNINGER.....?160000
- AMMUNITION.....?237500

SYDSTATERNES MORAL ER NOGENLUNDE.

NORDSTATERNES MORAL ER NOGENLUNDE.

NORDSTATSGENERAL---DU ER I OFFENSIVEN.

SYDSTATERNES STRATEGI ?4

NORDSTATERNES STRATEGI ?2

	SYDSTATERNE	NORDSTATERNE
SÆREDE/DØDE	18805	13738
DESERTØRER	13	10

SAMMENLIGNET MED DE VIRKELIGE SÆREDE/DØDE I SEVEN DAYS
SYDSTATERNE: 91 % AF DET REELLE TAB.
NORDSTATERNE: 87 % AF DET REELLE TAB.

NORDSTATERNE VINDER SEVEN DAYS

HVILKET SLAG ØNSKER DU AT UDKÆMPE ?5

DETTE ER SLAGET OM ANTIETAM

DEN 17. SEPTEMBER 1862. MISLYKKEDES DET FOR SYDSTATERNE AT FÅ MARYLAND MED I KONFEDERATIONEN.

	SYDSTATERNE	NORDSTATERNE
MÆND	40755	50913
PENGE	\$ 179100	\$ 230000
INFLATION	27 %	8 %

SYDSTATSGENERAL---

HVOR MANGE PENGE ØNSKER DU AT BRUGE PÅ:

- FORPLEJNING.....?79100
- LØNNINGER.....?50000
- AMMUNITION.....?50000

NORDSTATSGENERAL---

HVOR MANGE PENGE ØNSKER DU AT BRUGE PÅ:

- FORPLEJNING.....?100000
- LØNNINGER.....?30000
- AMMUNITION.....?100000

SYDSTATERNES MORAL ER HØJ!

NORDSTATERNES MORAL ER HØJ!

NORDSTATSGENERAL---DU ER I OFFENSIVEN.

SYDSTATERNES STRATEGI ?4

NORDSTATERNES STRATEGI ?3

	SYDSTATERNE	NORDSTATERNE
SÆREDE/DØDE	10404	10694
DESERTØRER	6	5

SAMMENLIGNET MED DE VIRKELIGE SÆREDE/DØDE I ANTIETAM
 SYDSTATERNE: 104 % AF DET REELLE TAB.
 NORDSTATERNE: 89 % AF DET REELLE TAB.

SYDSTATERNE VINDER ANTIETAM

HVILKET SLAG ØNSKER DU AT UDKÆMPE ?6

DETTE ER SLAGET OM FREDERICKSBURG

DEN 13. DECEMBER 1862. SYDSTATERNE AFVÆRGEDE UNDER GENERAL LEE'S LEDELSE ET NORDSTATSANGREB LEDET AF GENERAL BURNSIDE.

	SYDSTATERNE	NORDSTATERNE
MÆND	75769	122473
PENGE	\$ 340000	\$ 540000
INFLATION	25 %	10 %

SYDSTATSGENERAL---

HVOR MANGE PENGE ØNSKER DU AT BRUGE PÅ:

- FORPLEJNING.....?100000
- LØNNINGER.....?100000
- AMMUNITION.....?135800

NORDSTATSGENERAL---

HVOR MANGE PENGE ØNSKER DU AT BRUGE PÅ:

- FORPLEJNING.....?130000
- LØNNINGER.....?150000
- AMMUNITION.....?272000

TÆNK DIG LIDT OM! DU HAR KUN 540000 \$

- FORPLEJNING.....?250000
- LØNNINGER.....?175000
- AMMUNITION.....?115000

SYDSTATERNES MORAL ER NOGENLUNDE.

NORDSTATERNES MORAL ER HØJ!

NORDSTATSGENERAL---DU ER I DEFENSIVEN.

SYDSTATERNES STRATEGI ?2

NORDSTATERNES STRATEGI ?4

	SYDSTATERNE	NORDSTATERNE
SÆREDE/DØDE	4870	13003
DESERTØRER	11	2

SAMMENLIGNET MED DE VIRKELIGE SÆREDE/DØDE I FREDERICKSBURG
 SYDSTATERNE: 91 % AF DET REELLE TAB.
 NORDSTATERNE: 103 % AF DET REELLE TAB.

SYDSTATERNE VINDER FREDERICKSBURG

HVILKET SLAG ØNSKER DU AT UDKÆMPE ?

```

100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY"
110 PRINT "-----"
120 PRINT " BORGERKRIG"
130 PRINT "-----"
140 PRINT
150 REM Spillet er oprindeligt skrevet af:
160 REM Cram, Goodie, Hibbard Lexington H.S.
170 REM Modifikationer: G.Paul, R.Hess, 1973
180 DIM S(4),C$(14),M1(14),M2(14),C1(14),C2(14),M(14)
190 REM Nordstaternes oplysninger og sandsynlig sydstatsstrategi
200 S(1) = 25:S(2) = 25:S(3) = 25:S(4) = 25
210 REM Indlæs historiske data
220 FOR D = 1 TO 14
230 READ C$(D),M1(D),M2(D),C1(D),C2(D),M(D)
240 NEXT D
250 D = RND (1)
260 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
270 INPUT X$
280 IF LEFT$(X$,1) = "J" THEN 340
290 IF LEFT$(X$,1) = "N" THEN 640
300 PRINT
310 PRINT "SVAR 'J' ELLER 'N' ...."
320 PRINT
330 GOTO 260
340 PRINT
350 PRINT "DETTE ER EN SIMULATION AF DEN AMERIKANSKE BORGERKRIG"
360 PRINT
370 PRINT "HVIS DU ØNSKER AT SPILLE SKAL DU INDTASTE DINE SVAR, NÅR"
380 PRINT "NÅR DATAMATEN BEDER OM DEM. HUSK, AT ALLE FAKTORERNE"
390 PRINT "I SPILLET ER INDBYRDES AFHÆNGIGE, OG AT DE SVAR DU GIVER"
400 PRINT "KAN ÆNDRE HISTORIENS GANG."
410 PRINT
420 PRINT "SPILLET ER BASERET PÅ FAKTA OG TAL FRA DE VIRKELIGE"
430 PRINT "HÆNDELSER. DE FLESTE SLAG TENDERER MOD AT ENDE LIGESOM DE"
440 PRINT "GJORDE I DEN VIRKELIGE BORGERKRIG, MEN DET ER FØRST OG"
450 PRINT "FREMEST AFHÆNGIG AF DIN INDSATS!!!!"
460 PRINT
470 PRINT "DIT MÅL ER AT VINDE SÅ MANGE SLAG SOM MULIGT."
480 PRINT
490 PRINT "DINE FORSVARSTAKTISKE VALGMULIGHEDER ER:"
500 PRINT
510 PRINT " (1) ARTILLERIILD"
520 PRINT " (2) BEFÆSTNINGER MOD FRONTALE ANGREB"
530 PRINT " (3) BEFÆSTNINGER MOD FLANKE ANGREB"
540 PRINT " (4) TILBAGETRÆKNING"
550 PRINT
560 PRINT "DINE ANGREBSTAKTISKE VALGMULIGHEDER ER:"
570 PRINT
580 PRINT " (1) ARTILLERIANGREB"
590 PRINT " (2) FRONTALANGREB"
600 PRINT " (3) FLANKEANGREB"
610 PRINT " (4) OMRINGNING"
620 PRINT
630 PRINT "DU KAN OVERGIVE DIG VED AT VÆLGE TAKTISK MULIGHED '5'."
640 PRINT
650 PRINT "ER DER 2 GENERALER TIL STEDE (SVAR 'J' ELLER 'N') ";
660 INPUT B$
670 PRINT
680 IF LEFT$(B$,1) = "J" THEN 720
690 IF LEFT$(B$,1) < > "N" THEN 650
700 PRINT "DU ER SYDSTATSGENERALEN.....HELD OG LYKKE!"
710 PRINT
720 D = 1
730 IF LEFT$(B$,1) < > "J" THEN 750
740 D = 2
750 PRINT "VÆLG ET SLAG VED AT INDTAST ET TAL MELLEM 1 OG 14, NÅR"
760 PRINT "DATAMATEN BEDER OM DETTE. ET HVILKET SOM HELST ANDET TAL"
770 PRINT "AFSLUTTER SPILLET, MENS 'O' BRINGER DIG TILBAGE TIL DET"
780 PRINT "FORRIGE SLAG, SÅ DU HAR MULIGHED FOR AT SPILLE DET OM."
790 PRINT
800 PRINT "BEMÆRK: HVIS DU VÆLGER EN NEGATIV VÆRDI FOR FORPLEJNING,"
810 PRINT "ANVENDES DEN SAMME VÆRDI SOM BLEV BRUGT I FORRIGE SLAG."
820 PRINT
830 PRINT "NÅR DU HAR VALGT ET SLAG, VIL DU SÅ HAVE EN"
840 PRINT "BESKRIVELSE AF SLAGET (SVAR 'J' ELLER 'N') ";
850 INPUT X$
860 IF LEFT$(X$,1) = "J" THEN 880
870 IF LEFT$(X$,1) < > "N" THEN 840
880 L = 0:W = 0:R1 = 0:Q1 = 0:M3 = 0:M4 = 0:P1 = 0:P2 = 0:T1 = 0
890 T2 = 0:F(2) = 0:H(2) = 0:B(2) = 0:R2 = 0:Q2 = 0:C6 = 0:F = 0
900 W0 = 0:Y = 0:Y2 = 0:U = 0:U2 = 0
910 PRINT
920 PRINT "HVILKET SLAG ØNSKER DU AT UDKÆMPE ";
930 INPUT A
940 IF A < > 0 THEN 960
950 IF R < > 0 THEN 1230
960 IF A < = 0 THEN 3340
970 IF A > = 15 THEN 3340

```

```

980 C$ = C$(A)
990 M1 = M1(A)
1000 M2 = M2(A)
1010 C1 = C1(A)
1020 C2 = C2(A)
1030 M = M(A)
1040 U = 0
1050 REM Inflations-rutine
1060 I1 = 10 + (L - W) * 2
1070 I2 = 10 + (W - L) * 2
1080 REM Penge til rådighed
1090 D(1) = 100 * INT ((M1 * (100 - I1) / 2000) * (1 + (R1 - Q1) /
(R1 + 1)) + .5)
1100 D(2) = 100 * INT (M2 * (100 - I2) / 2000 + .5)
1110 IF LEFT$ (B$,1) < > "J" THEN 1130
1120 D(2) = 100 * INT ((M2 * (100 - I2) / 2000) * (1 + (R2 - Q2) /
(R2 + 1)) + .5)
1130 REM Mænd til rådighed
1140 M5 = INT (M1 * (1 + (P1 - T1) / (M3 + 1)))
1150 M6 = INT (M2 * (1 + (P2 - T2) / (M4 + 1)))
1160 F1 = 5 * M1 / 6
1170 PRINT
1180 PRINT
1190 PRINT "DETTE ER SLAGET OM ";C$
1200 PRINT
1210 IF LEFT$ (X$,1) = "N" THEN 1240
1220 ON A GOTO
4290,4340,4370,4420,4460,4490,4520,4550,4590,4630,4670,4700,4740,4780
1230 PRINT C$;" GENTAGELSE"
1240 PRINT
1250 PRINT "          SYDSTATERNE          NORDSTATERNE"
1260 PRINT "MÆND          ";
1270 V = M5
1280 GOSUB 4810
1290 PRINT "          ";
1300 V = M6
1310 GOSUB 4810
1320 PRINT
1330 PRINT "PENGE          $";
1340 V = D(1)
1350 GOSUB 4810
1360 PRINT "          $";
1370 V = D(2)
1380 GOSUB 4810
1390 PRINT
1400 PRINT "INFLATION          ";
1410 V = I1 + 15
1420 GOSUB 4810
1430 PRINT "%          ";
1440 V = I2
1450 GOSUB 4810
1460 PRINT "%"
1470 PRINT
1480 REM Det er kun i udskriften at sydstatsinflation er = I1+15%
1490 REM Hvis der er to generaler, indtastes sydstaterne først
1500 FOR I = 1 TO D
1510 IF LEFT$ (B$,1) < > "J" THEN 1540
1520 IF I = 2 THEN 1540
1530 PRINT "SYDSTATSGENERAL----"
1540 PRINT "HVOR MANGE PENGE ØNSKER DU AT BRUGE PÅ:"
1550 PRINT
1560 PRINT " - FORPLEJNING.....";
1570 INPUT F
1580 IF F > = 0 THEN 1650
1590 IF R1 < > 0 THEN 1620
1600 PRINT "DER ER IKKE TIDLIGERE INDTASTET EN VÆRDI"
1610 GOTO 1560
1620 PRINT "DET ANTAGES AT DU VIL BEHOLDE DE GAMLE VÆRDIER."
1630 PRINT
1640 GOTO 1800
1650 F(I) = F
1660 PRINT " - LØNNINGER.....";
1670 INPUT H(I)
1680 N = 1
1690 IF H(I) < 0 THEN 1780
1700 PRINT " - AMMUNITION.....";
1710 INPUT B(I)
1720 N = 2
1730 IF B(I) < 0 THEN 1780
1740 PRINT
1750 IF F(I) + H(I) + B(I) < = D(I) THEN 1800
1760 PRINT "TÆNK DIG LIDT OM! DU HAR KUN";D(I);"$"
1770 GOTO 1560
1780 PRINT "NEGATIVE VÆRDIER ER IKKE TILLADT."
1790 ON N GOTO 1660,1700
1800 IF LEFT$ (B$,1) < > "J" THEN 1840
1810 IF I = 2 THEN 1840
1820 PRINT "NORDSTATSGENERAL----"
1830 NEXT I
1840 FOR Z = 1 TO D
1850 IF LEFT$ (B$,1) < > "J" THEN 1910
1860 ON Z GOTO 1870,1890
1870 PRINT " SYDSTATERNES ";
1880 GOTO 1910
1890 PRINT "NORDSTATERNES ";
1900 REM Undersøg moralen
1910 O = ((2 * F(Z) ^ 2 + H(Z) ^ 2) / F1 ^ 2 + 1)
1920 IF O < 10 THEN 1950
1930 PRINT "MORAL ER HØJ!"
1940 GOTO 1990
1950 IF O < 5 THEN 1980
1960 PRINT "MORAL ER NOGENLUNDE."
1970 GOTO 1990
1980 PRINT "MORAL ER ELENDIG."
1990 PRINT
2000 IF LEFT$ (B$,1) < > "J" THEN 2060
2010 O(Z) = 0
2020 NEXT Z
2030 O2 = O(2)
2040 O = O(1)
2050 PRINT "NORDSTATSGENERAL----";
2060 REM Øjeblikkelige offensiv/defensiv situation
2070 IF M < > 3 THEN 2100
2080 PRINT "DU ER I OFFENSIVEN."
2090 GOTO 2140
2100 IF M < > 1 THEN 2130
2110 PRINT "DU ER I DEFENSIVEN."
2120 GOTO 2140
2130 PRINT "BEGGE SIDER ER I OFFENSIVEN."
2140 PRINT
2150 REM Vælg strategi
2160 IF LEFT$ (B$,1) < > "J" THEN 2210
2170 FOR I = 1 TO 2
2180 ON I GOTO 2190,2220
2190 PRINT "SYDSTATERNES STRATEGI ";
2200 GOTO 2220
2210 PRINT "DIN STRATEGI ";
2220 INPUT Y
2230 PRINT
2240 IF ABS (Y - 3) < 3 THEN 2280
2250 PRINT "STRATEGI";Y;"ER IKKE TILLADT."
2260 PRINT
2270 GOTO 2210
2280 IF LEFT$ (B$,1) = "J" THEN 2320
2290 IF Y = 5 THEN 3310
2300 GOSUB 3780
2310 GOTO 2490
2320 IF I = 2 THEN 2360
2330 Y1 = Y
2340 PRINT "NORDSTATERNES STRATEGI ";
2350 NEXT I
2360 Y2 = Y
2370 Y = Y1
2380 IF Y2 = 5 THEN 2340
2390 REM Simulerede tab - Nordstaterne
2400 C6 = (2 * C2 / 5) * (1 + 1 / (2 * (ABS (Y2 - Y) + 1)))
2410 C6 = C6 * (1.28 + (5 * M2 / 6) / (B(2) + 1))
2420 C6 = INT (C6 * (1 + 1 / O2) + .5)
2430 REM Hvis TAB>ANTAL MÆND, så omregn tab
2440 E2 = 100 / O2
2450 IF INT (C6 + E6) < M6 THEN 2510
2460 C6 = INT (13 * M6 / 20)
2470 E2 = 7 * C6 / 13
2480 U2 = 1
2490 REM Beregn simulerede tab
2500 PRINT
2510 PRINT
2520 PRINT "          SYDSTATERNE          NORDSTATERNE"
2530 C5 = (2 * C1 / 5) * (1 + 1 / (2 * (ABS (Y2 - Y) + 1)))
2540 C5 = INT (C5 * (1 + 1 / O) * (1.28 + F1 / (B(1) + 1)) + .5)
2550 E = 100 / O
2560 IF C5 + 100 / O < M1 * (1 + (P1 - T1) / (M3 + 1)) THEN 2600
2570 C5 = INT (13 * M1 / 20 * (1 + (P1 - T1) / (M3 + 1)))
2580 E = 7 * C5 / 13
2590 U = 1
2600 IF D = 1 THEN 2960
2610 PRINT "SRAREDE/DØDE          ";
2620 V = C5
2630 GOSUB 4810
2640 PRINT "          ";
2650 V = C6
2660 GOSUB 4810
2670 PRINT
2680 PRINT "DESERTØRER          ";
2690 V = INT (E)
2700 GOSUB 4810

```

```

2710 PRINT " ";
2720 V = INT (E2)
2730 GOSUB 4810
2740 PRINT
2750 PRINT
2760 IF LEFT$ (B$,1) < > "J" THEN 2990
2770 PRINT "SAMMENLIGNET MED DE VIRKELIGE SÆREDE/DØDE I ";C$
2780 PRINT "SYDSTATERNE: "; INT (100 * (C5 / C1) + .5);"% AF DET
REELLE TAB."
2790 PRINT "NORDSTATERNE: "; INT (100 * (C6 / C2) + .5);"% AF DET
REELLE TAB."
2800 PRINT
2810 REM Hvem vinder?
2820 IF U < > 1 THEN 2840
2830 IF U2 = 1 THEN 2920
2840 IF U = 1 THEN 2880
2850 IF U2 = 1 THEN 2900
2860 IF C5 + E = C6 + E2 THEN 2920
2870 IF C5 + E < C6 + E2 THEN 2900
2880 PRINT "NORDSTATERNE VINDER ";C$
2890 GOTO 3070
2900 PRINT "SYDSTATERNE VINDER ";C$
2910 GOTO 3140
2920 PRINT "SLAGET VAR UAFGJORT"
2930 W0 = W0 + 1
2940 IF A = 0 THEN 3270
2950 GOTO 3160
2960 C6 = INT (17 * C2 * C1 / (C5 * 20))
2970 E2 = 5 * 0
2980 GOTO 2610
2990 PRINT "DINE SÆREDE/DØDE UDGJORDE"; INT (100 * (C5 / C1) + .5)
;"% AF DET"
3000 PRINT "FAKTISKE TAB VED ";C$
3010 PRINT
3020 REM Find ud af hvem der vandt
3030 IF U = 1 THEN 3050
3040 IF C5 + E < 17 * C2 * C1 / (C5 * 20) + 5 * 0 THEN 3100
3050 PRINT "DU TABER ";C$
3060 PRINT
3070 IF A = 0 THEN 3270
3080 L = L + 1
3090 GOTO 3160
3100 PRINT "DU VINDER ";C$
3110 PRINT
3120 REM Faktorer i slaget, der ændrer de historiske kendsgerninger.
3130 REM Hvis slaget spilles for anden gang, ajourføres ikke
3140 IF A = 0 THEN 3270
3150 W = W + 1
3160 T1 = T1 + C5 + E
3170 T2 = T2 + C6 + E2
3180 P1 = P1 + C1
3190 P2 = P2 + C2
3200 Q1 = Q1 + (F(1) + H(1) + B(1))
3210 Q2 = Q2 + (F(2) + H(2) + B(2))
3220 R1 = R1 + M1 * (100 - I1) / 20
3230 R2 = R2 + M2 * (100 - I2) / 20
3240 M3 = M3 + M1
3250 M4 = M4 + M2
3260 GOSUB 4000
3270 U = 0:U2 = 0
3280 PRINT "-----"
3290 GOTO 910
3300 REM Afslut
3310 PRINT "SYDSTATERNE HAR OVERGIVET SIG."
3320 GOTO 3340
3330 PRINT "NORDSTATERNE HAR OVERGIVET SIG."
3340 PRINT
3350 PRINT "SYDSTATERNE ";
3360 PRINT "HAR VUNDET";W;"SLAG, OG HAR TABT";L;"-"
3370 IF Y = 5 THEN 3430
3380 IF Y2 = 5 THEN 3410
3390 IF W < = L THEN 3430
3400 IF Y = 5 THEN 3430
3410 PRINT "SYDSTATERNE HAR VUNDET KRIGEN."
3420 GOTO 3440
3430 PRINT "NORDSTATERNE HAR VUNDET KRIGEN."
3440 PRINT
3450 IF R1 = 0 THEN 3770
3460 PRINT "FOR DE";W + L + W0;"SLAG, DER ER BLEVET UDKÆMPET,"
3470 PRINT "(GENTAGELSER UNDTAGET), GÆLDER FØLGENDE:"
3480 PRINT
3490 PRINT " SYDSTATERNE NORDSTATERNE"
3500 PRINT "HISTORISKE TAB ";
3510 V = INT (P1 + .5)
3520 GOSUB 4810
3530 PRINT " ";
3540 V = INT (P2 + .5)
3550 GOSUB 4810

```

```

3560 PRINT
3570 PRINT "SIMULEREDE TAB ";
3580 V = INT (T1 + .5)
3590 GOSUB 4810
3600 PRINT " ";
3610 V = INT (T2 + .5)
3620 GOSUB 4810
3630 PRINT
3640 PRINT
3650 PRINT "% AF REELLE ";
3660 V = INT (100 * (T1 / P1) + .5)
3670 GOSUB 4810
3680 PRINT " ";
3690 V = INT (100 * (T2 / P2) + .5)
3700 GOSUB 4810
3710 PRINT
3720 IF LEFT$ (B$,1) = "J" THEN 3770
3730 PRINT
3740 PRINT "NORDSTATERNES SPIONER ANTAGER, AT SYDSTATERNE HAR BRUGT"
3750 PRINT "STRATEGIERNE 1, 2, 3, 4 MED FØLGEDE PROCENTSATSER."
3760 PRINT S(1);S(2);S(3);S(4)
3770 GOTO 4890
3780 REM Nordstaternes strategi er valgt af datamaten
3790 PRINT "NORDSTATERNES STRATEGI ER ";
3800 IF A < > 0 THEN 3870
3810 INPUT Y2
3820 IF Y2 < = 0 THEN 3840
3830 IF Y2 < 5 THEN 3990
3840 PRINT "INDTAST '1', '2', '3' ELLER '4' (SÆRDVANLIGVIS FOREGRÆNDE"
3850 PRINT "NORDSTATSSTRATEGI)"
3860 GOTO 3810
3870 S0 = 0
3880 R = 100 * RND (1)
3890 FOR I = 1 TO 4
3900 S0 = S0 + S(I)
3910 REM Hvis den aktuelle strategi findes i DATA-sætningerne
3920 REM giver R-100 øget vægt til denne strategi
3930 IF R < S0 THEN 3970
3940 NEXT I
3950 REM Hvis den valgte strategi findes her,
3960 REM så er Y2 = historisk strategi
3970 Y2 = I
3980 PRINT Y2
3990 RETURN
4000 REM Indlær nuværende strategi, og glem den gamle
4010 REM Nuværende sydstatsstrategi forøges med 3*S, de øvrige
4020 REM mister S sandsynligheds-point, medmindre sandsynligheden
4030 REM bliver mindre end 5%.
4040 S = 3
4050 S0 = 0
4060 FOR I = 1 TO 4
4070 IF S(I) < = 5 THEN 4100
4080 S(I) = S(I) - S
4090 S0 = S0 + S
4100 NEXT I
4110 S(Y) = S(Y) + S0
4120 RETURN
4130 REM Historiske data
4140 REM Du kan indsætte flere værdier og justere READ-sætningerne
4150 DATA "BULL RUN",18000,18500,1967,2708,1
4160 DATA "SHILOH",40000,44894,10699,13047,3
4170 DATA "SEVEN DAYS",95000,115000,20614,15849,3
4180 DATA "SECOND BULL RUN",54000,63000,10000,14000,2
4190 DATA "ANTIETAM",40000,50000,10000,12000,3
4200 DATA "FREDERICKSBURG",75000,120000,5377,12653,1
4210 DATA "MURFREESBORO",38000,45000,11000,12000,1
4220 DATA "CHANCELLORSVILLE",32000,90000,13000,17197,2
4230 DATA "VICKSBURG",50000,70000,12000,19000,1
4240 DATA "GETTYSBURG",72500,85000,20000,23000,3
4250 DATA "CHICKAMAUGA",66000,60000,18000,16000,2
4260 DATA "CHATTANOOGA",37000,60000,36700,5800,2
4270 DATA "SPOTSYLVANIA",62000,110000,17723,18000,2
4280 DATA "ATLANTA",65000,100000,8500,3700,1
4290 PRINT "DEN 21. JULI 1861 MØDTE SYDSTATSGENERALEN BEAUREGARD MED"
4300 PRINT "NORDSTATSGENERALEN MCDOWELL I ET UFORBEREDT SLAG VED"
4310 PRINT "BULL RUN. GENEAL JACKSON HJALP MED TIL AT SLÅ NORDSTATS-"
4320 PRINT "ANGREBET TILBAGE."
4330 GOTO 1240
4340 PRINT "DEN 6-7 APRIL 1862 MISLYKKEDES SYDSTATERNES OVERRASKELSES-"
4350 PRINT "ANGREB VED SHILOH GRUNDET DÅRLIG PLANLÆGNING."
4360 GOTO 1240
4370 PRINT "FRA DEN 25. JUNI TIL DEN 1. JULI 1862 OPRETHOLDT SYDSTATS-"
4380 PRINT "GENERALEN LEE EN OFFENSIV UNDER HELE SLAGET, HVORVED"
4390 PRINT "NORDSTATSTROPPERNE UNDER LEDELSE AF GENERAL MCLELLAN,"
4400 PRINT "BLEV HOLDT UDE AF RICHMOND."
4410 GOTO 1240
4420 PRINT "DEN 29-30 AUGUST 1862 DREV DE SAMLEDE SYDSTATSTROPPER UNDER"
4430 PRINT "LEDELSE AF LEE OG JACKSON NORDSTATSTROPPERNE TILBAGE TIL"


```



```

4440 PRINT "WASHINGTON."
4450 GOTO 1240
4460 PRINT "DEN 17. SEPTEMBER 1862. MISLYKKEDES DET FOR SYDSTATERNE AT"
4470 PRINT "FÅ MARYLAND MED I KONFEDERATIONEN."
4480 GOTO 1240
4490 PRINT "DEN 13. DECEMBER 1862. SYDSTATERNE AFVÆRGEDE UNDER GENERAL"
4500 PRINT "LEE'S LEDELSE ET NORDSTATSANGREB LEDET AF GENERAL BURNSIDE."
4510 GOTO 1240
4520 PRINT "DEN 31. DECEMBER 1862. UNDER GENERAL BRAGG'S LEDELSE VANDT"
4530 PRINT "SYDSTATERNE EN KNEBEN SEJR."
4540 GOTO 1240
4550 PRINT "1-6 MAJ 1863. SYDSTATERNE VANDT EN BEKOSTELIG SEJR, MEN"
4560 PRINT "MISTEDE EN AF DERES MEST FREMSTRENDE GENERALER, 'STONEWALL'"
4570 PRINT "JACKSON."
4580 GOTO 1240
4590 PRINT "DEN 4. JULI 1863. VICKSBURG VAR ET DYREBART TAB FOR SYD-"
4600 PRINT "STATERNE, FORDI DEN GAV NORDSTATERNE ADGANG TIL FLODEN"
4610 PRINT "MISSISSIPPI."
4620 GOTO 1240
4630 PRINT "1-3 JULI 1863. MEDFØRER EN FEJL BEGRÆT AF SYDSTATSGENERALEN"
4640 PRINT "LEE, AT SYDSTATERNE TABER SLAGET VED GETTYSBURG. DETTE ER"
4650 PRINT "ET AF DE VIGTIGSTE SLAG I BØRGERKRIGEN."
4660 GOTO 1240
4670 PRINT "15. SEPTEMBER 1863. EN STRID I EN SKOV NÆR CHICKAMAUGA"
4680 PRINT "FØRTE TIL EN DYREKØBT SEJR TIL SYDSTATERNE."
4690 GOTO 1240
4700 PRINT "25. NOVEMBER 1863. EFTER AT SYDSTATERNE HAVDE BELEJRET"
4710 PRINT "GENERAL ROSENCRANS ARME I TRE MÅNEDER, BRØD GENERAL GRANT"
4720 PRINT "BELEJRINGEN."
4730 GOTO 1240
4740 PRINT "5. MAJ 1864. GENERAL GRANT'S PLAN OM AT HOLDE GENERAL LEE"
4750 PRINT "ISOLERET BEGYNDTE AT FEJLE HER, OG FORTSATTE VED"
4760 PRINT "COLD HARBOR OG PETERSBURG."
4770 GOTO 1240
4780 PRINT "AUGUST 1864. SHERMAN GIK MED TRE ARMEER MOD ATLANTA, OG"
4790 PRINT "VANDT DET AFGØRENDE SLAG MOD SYDSTATERNE."
4800 GOTO 1240
4810 IF V > 99999 THEN 4870
4820 IF V > 9999 THEN PRINT " "; GOTO 4870
4830 IF V > 999 THEN PRINT " "; GOTO 4870
4840 IF V > 99 THEN PRINT " "; GOTO 4870
4850 IF V > 9 THEN PRINT " "; GOTO 4870
4860 PRINT " ";
4870 PRINT V;
4880 RETURN
4890 END

```



Bowling

Dette program simulerer spillet bowling. Op til 4 spillere kan del- tage. Man spiller ti omgange. For at rulle kuglen skal du simpelt hen taste »R«. Efter hver omgang viser datamaten et diagram over keglerne. »0« betyder, at keglen er væltet, »+« at den står endnu. Du får også en analyse af slaget:

I RENDEN: (kuglen rullet af banen)
STRIKE: (alle kegler væltet i et slag)
SPARE: (alle kegler væltet i to slag)
FEJL: (vises efter andet slag, hvis der stadig står kegler)

Bowling er skrevet af Paul Peraino, mens han endnu var stude- rende på Woodrow Wilson High School, San Francisco, Califor- nien.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BOWLING

VELKOMMEN TIL BOWLINGHALLEN
TAG DINE VENNER MED

LAD OS FØRST LÆRE HINANDEN AT KENDE

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

BOWLING KRÆVER BÅDE INTELLIGENS OG DYG- TIGHED. UNDER SPILLET HOLDER DATAMATEN REDE PÅ STILLINGEN.

DU KAN SPILLE MOD MAKSIMALT FIRE SPIL- LERE.

DER SPILLES 10 OMGANGE.

PÅ DIAGRAMMET OVER KEGLERNE BETYDER ET '0' = AT KEGLEN ER VÆLTET
'+' = AT KEGLEN STÅR ENDNU

NÅR SPILLET ER SLUT VISER DATAMATEN DE POINTS SPILLERNE HAR OPNÅET.

.....HVOR MANGE VIL SPILLE ??

DET ER IORDEN

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 1 KUGLE: 1

0 0 + 0
+ 0 +
0 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 1 KUGLE: 2

0 0 0 0
+ 0 0
0 0
0

FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 1 KUGLE: 1

0 0 0 0
+ 0 0
0 +
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 1 KUGLE: 2

0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 2 KUGLE: 1

+ 0 0 0
0 0 0
0 0
+

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 2 KUGLE: 2

0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 2 KUGLE: 1

0 0 0 +
0 + 0
0 0
+

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 2 KUGLE: 2

0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 3 KUGLE: 1

+ 0 0 0
0 0 0
+ 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 3 KUGLE: 2

+ 0 0 0
0 0 0
+ 0
0

RENDEN!
FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 3 KUGLE: 1

0 + 0 0
+ 0 0
0 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 3 KUGLE: 2

0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 4 KUGLE: 1

+ 0 0 0
0 + 0
0 +
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 1 OMGANG: 4 KUGLE: 2

0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 4 KUGLE: 1

+ + + 0
0 + 0
0 +
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R

SPILLER: 2 OMGANG: 4 KUGLE: 2

0 0 + 0
0 0 0
0 0
0

FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 5 KUGLE: 1
+ + 0 +
0 + 0
0 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 5 KUGLE: 2
0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 5 KUGLE: 1
0 0 0 0
0 0 0
0 0
0

STRIKE!!!

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 6 KUGLE: 1
0 0 + +
0 + +
0 0
+

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 6 KUGLE: 2
0 0 0 0
0 + 0
0 0
+

FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 6 KUGLE: 1
0 0 0 0
+ 0 +
0 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 6 KUGLE: 2
0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 7 KUGLE: 1
0 0 0 +
0 0 0
0 +
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 7 KUGLE: 2
0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 7 KUGLE: 1
+ 0 + 0
0 0 0
0 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 7 KUGLE: 2
0 0 + 0
0 0 0
0 0
0

FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 8 KUGLE: 1
0 0 0 0
0 0 0
+ 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 8 KUGLE: 2
0 0 0 0
0 0 0
0 0
0

SPARE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 8 KUGLE: 1
0 0 + +
0 0 0
0 +
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 8 KUGLE: 2
0 0 0 0
0 0 0
0 +
0

FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 9 KUGLE: 1
0 + 0 0
0 0 0
0 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 9 KUGLE: 2
0 + 0 0
0 0 0
0 0
0

RENDEN!
FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 9 KUGLE: 1
0 0 + 0
+ 0 0
+ 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 9 KUGLE: 2
0 0 + 0
+ 0 0
0 0
0

FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 10 KUGLE: 1
0 + + 0
0 0 0
0 0
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 1 OMGANG: 10 KUGLE: 2
0 0 + 0
0 0 0
0 0
0

FEJL.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 10 KUGLE: 1
0 0 0 +
0 0 0
0 +
0

RUL DEN ANDEN KUGLE.

TAST 'R', SÅ KUGLEN RULLER ?R
SPILLER: 2 OMGANG: 10 KUGLE: 2
0 0 0 0
0 0 0
0 +
0

FEJL.

```

----- OMGANGE -----
 1  2  3  4  5  6  7  8  9 10
-----
 7  8  8  7  6  5  8  9  9  8
 9 10  8 10 10  8 10 10  9  9
 1  2  1  2  2  1  2  2  1  1
-----
 8  7  5  9  8  8  8  7  7  8
10 10  8 10  9 10  9  9  8  9
 2  2  1  2  1  2  1  1  1  1
-----

```

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          BOWLING"
130 PRINT "-----"
140 PRINT
150 DIM C(15),A(100,6)
160 PRINT "    VELKOMMEN TIL BOWLINGHALLEN"
170 PRINT "          TAG DINE VENNER MED"
180 PRINT
190 PRINT " LAD OS FØRST LÆRE HINANDEN AT KENDE"
200 PRINT
210 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
220 INPUT Z$
230 PRINT
240 IF LEFT$(Z$,1) = "J" THEN 260
250 IF LEFT$(Z$,1) = "N" THEN 420
260 PRINT "BOWLING KRÆVER BÅDE INTELLIGENS OG DYG--"
270 PRINT "TIGHED. UNDER SPILLET HOLDER DATAMATEN"
280 PRINT "REDE PÅ STILLINGEN."
290 PRINT
300 PRINT "DU KAN SPILLE MOD MAKSIMALT FIRE SPIL--"
310 PRINT "LERE."
320 PRINT
330 PRINT "DER SPILLES 10 OMGANGE."
340 PRINT
350 PRINT "PÅ DIAGRAMMET OVER KEGLERNE BETYDER ET"
360 PRINT "'O' = AT KEGLEREN ER VÆLTET"
370 PRINT "'+' = AT KEGLEREN STÅR ENDNU"
380 PRINT
390 PRINT "NÅR SPILLET ER SLUT VISER DATAMATEN DE"
400 PRINT "POINTS SPILLERNE HAR OPNÅRET."
410 PRINT
420 PRINT ".....HVOR MANGE VIL SPILLE ";
430 INPUT R
440 PRINT
450 PRINT "DET ER IORDEN"
460 FOR I = 1 TO 100
470 FOR J = 1 TO 6
480 A(I,J) = 0
490 NEXT J
500 NEXT I
510 F = 1
520 FOR P = 1 TO R
530 M = 0
540 B = 1
550 M = 0
560 Q = 0
570 FOR I = 1 TO 15
580 C(I) = 0
590 NEXT I
600 REM Kuglegenerator bruger modulus 15 system
610 PRINT
620 PRINT "TAST 'R', SÅ KUGLEN RULLER ";
630 INPUT N$
640 PRINT
650 IF N$ < > "R" THEN 620
660 K = 0
670 D = 0
680 FOR I = 1 TO 20
690 X = INT ( RND (1) * 100)
700 FOR J = 1 TO 10
710 IF X < 15 * J THEN 730
720 NEXT J
730 C(15 * J - X) = 1
740 NEXT I
750 REM Diagram over kegler
760 PRINT "SPILLER: "; P; "  OMGANG: ";
770 PRINT F; "  KUGLE: "; B
780 FOR I = 1 TO 4

```

```

790 PRINT
800 FOR L = 1 TO I
810 PRINT " ";
820 NEXT L
830 FOR J = 1 TO 5 - I
840 K = K + 1
850 IF C(K) = 1 THEN 880
860 PRINT "+ ";
870 GOTO 890
880 PRINT "O ";
890 NEXT J
900 NEXT I
910 PRINT
920 PRINT
930 REM Analyser omgangen
940 FOR I = 1 TO 10
950 D = D + C(I)
960 NEXT I
970 IF D - M < > 0 THEN 990
980 PRINT "  RENDEN!"
990 IF B < > 1 OR D < > 10 THEN 1020
1000 PRINT "STRIKE!!!"
1010 Q = 3
1020 IF B < > 2 OR D < > 10 THEN 1050
1030 PRINT "  SPARE."
1040 Q = 2
1050 IF B < > 2 OR D > = 10 THEN 1080
1060 PRINT "  FEJL."
1070 Q = 1
1080 IF B < > 1 OR D > = 10 THEN 1100
1090 PRINT "RUL DEN ANDEN KUGLE."
1100 REM Gem resultaterne
1110 A(F * P,B) = D
1120 IF B = 2 THEN 1180
1130 B = 2
1140 M = D
1150 IF Q = 3 THEN 1100
1160 A(F * P,B) = D - M
1170 IF Q = 0 THEN 600
1180 A(F * P,3) = Q
1190 NEXT P
1200 F = F + 1
1210 IF F < 11 THEN 520
1220 PRINT
1230 PRINT "----- OMGANGE -----"
1240 FOR I = 1 TO 10
1250 PRINT I;
1260 NEXT I
1270 PRINT
1280 PRINT "-----"
1290 FOR P = 1 TO R
1300 FOR I = 1 TO 3
1310 FOR J = 1 TO 10
1320 IF A(J * P,I) = 10 THEN PRINT "10 ";
1330 IF A(J * P,I) = 10 THEN 1350
1340 PRINT A(J * P,I);
1350 NEXT J
1360 PRINT
1370 NEXT I
1380 PRINT "-----"
1390 NEXT P
1400 PRINT
1410 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1420 INPUT A$
1430 PRINT
1440 IF LEFT$(A$,1) = "J" THEN 410
1450 END

```

Bunke

Dette spil starter med en bunke objekter, det kan f.eks. være mønter eller tændstikker. Du og din modstander (datamaten) skiftes til at fjerne mønter fra bunken. Før spillet begynder, bestemmer du selv, hvor mange og hvor få mønter der må tages pr. træk. Du definerer også selv, hvilket gevinstkriterium der skal gælde: 1) Enten vinder man ved at tage din sidste mønt, eller også 2) gælder det om at undgå at tage den sidste mønt. Du bestemmer også selv, om du eller datamaten skal starte.

Spillets strategi er baseret på modulo-matematik. Hvis det maksimale antal mønter, som en spiller må tage pr. træk, er M, så skal spilleren sørge for, at der, når han har trukket, er 1 MODULUS (M+1) mønter tilbage i bunken. Hvis du ikke forstår dette, så prøv at spille spillet Tændstikker før du går i gang med Bunke og mor dig godt!

Bunke er en generaliseret udgave af et stort antal spil af fjernet-objekt-fra-en-bunke typen. Den oprindelige datamatversion blev skrevet af en af de to fædre til programmeringssproget BASIC, nemlig John Kemeny ved Dartmouth College.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

BUNKE

DETTE SPIL HEDDER 'BUNKE' OG COMPUTEREN ER DIN UTRÆTTELIGE MODSTANDER.

SPILLET STARTER MED, AT DU HAR EN BUNKE MØNTER. DU OG DATAMATEN FJERNER NU EFTER TUR MØNTER FRA BUNKEN.

MAN KAN ENTEN VINDE VED AT TAGE DEN SIDSTE MØNT, ELLER VED AT FÅ MODSTANDEREN TIL AT TAGE DEN. DU BESTEMMER SELV HVILKEN AF DISSE TO MULIGHEDER DER SKAL GÆLDE. DU BESTEMMER OGSÅ SELV BEGYNDELSESBEINGELSERNE.

INDTAST PÅ '0' ER UGYLDIGE.

HVOR MANGE MØNTER ER DER I BUNKEN ?23

INDTAST GEVINSTKRITERIUM:
(1=TAG SIDSTE, 2=UNDGÅ SIDSTE) ?2

INDTAST MINDSTE OG STØRSTE ANTAL MØNTER
DER MÅ TAGES PR. GANG ?1,3

HVEM BEGYNDER? (1=COMPUTEREN, 2=DIG) ?2

HVOR MANGE TAGER DU ?2

COMPUTEREN TAGER 1
DER ER NU 20 TILBAGE.

HVOR MANGE TAGER DU ?3

COMPUTEREN TAGER 1
DER ER NU 16 TILBAGE.

HVOR MANGE TAGER DU ?3

COMPUTEREN TAGER 1
DER ER NU 12 TILBAGE.

HVOR MANGE TAGER DU ?3

COMPUTEREN TAGER 1
DER ER NU 8 TILBAGE.

HVOR MANGE TAGER DU ?3

COMPUTEREN TAGER 1
DER ER NU 4 TILBAGE.

HVOR MANGE TAGER DU ?3

COMPUTEREN TAGER 1 OG TABER!!!

HVOR MANGE MØNTER ER DER I BUNKEN ?

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          BUNKE"
130 PRINT "-----"
140 PRINT
150 PRINT "DETTE SPIL HEDDER 'BUNKE' OG COMPUTEREN"
160 PRINT "ER DIN UTRÆTTELIGE MODSTANDER."
170 PRINT
180 PRINT "SPILLET STARTER MED, AT DU HAR EN BUNKE"
190 PRINT "MØNTER. DU OG DATAMATEN FJERNER NU"
200 PRINT "EFTER TUR MØNTER FRA BUNKEN."
210 PRINT "MAN KAN ENTEN VINDE VED AT TAGE DEN"
220 PRINT "SIDSTE MØNT, ELLER VED AT FÅ MODSTANDE-"
230 PRINT "REN TIL AT TAGE DEN. DU BESTEMMER SELV"
240 PRINT "HVILKEN AF DISSE TO MULIGHEDER DER SKAL"
250 PRINT "GÆLDE. DU BESTEMMER OGSÅ SELV BEGYNDEL-"
260 PRINT "SESBEINGELSERNE."
270 PRINT
280 PRINT "INDTAST PÅ '0' ER UGYLDIGE."
290 PRINT
300 PRINT "HVOR MANGE MØNTER ER DER I BUNKEN ";
310 INPUT N
320 IF N < > 0 THEN 340

```

```

330 GOTO 300
340 IF N < > INT (N) THEN 290
350 IF N < 1 THEN 290
360 PRINT
370 PRINT "INDTAST GEVINSTKRITERIUM:"
380 PRINT "(1=TAG SIDSTE, 2=UNDGÅ SIDSTE) ";
390 INPUT M
400 IF M = 1 THEN 420
410 IF M < > 2 THEN 360
420 PRINT
430 PRINT "INDTAST MINDSTE OG STØRSTE ANTAL MØNTER"
440 PRINT "DER MÅ TAGES PR. GANG ";
450 INPUT A1,B1
460 IF A1 > B1 THEN 420
470 IF A1 < 1 THEN 420
480 IF A1 < > INT (A1) THEN 420
490 IF B1 < > INT (B1) THEN 420
500 PRINT
510 PRINT "HVEM BEGYNDER? (1=COMPUTEREN, 2=DIG) ";
520 INPUT S
530 IF S = 1 THEN 550
540 IF S < > 2 THEN 500
550 C = A1 + B1
560 IF S = 2 THEN 590
570 GOSUB 620
580 IF W = 1 THEN 290
590 GOSUB 870
600 IF W = 1 THEN 290
610 GOTO 570
620 Q = N
630 IF M = 1 THEN 650
640 Q = Q - 1
650 IF M = 1 THEN 710
660 IF N > A1 THEN 760
670 W = 1
680 PRINT
690 PRINT "COMPUTEREN TAGER";N;"OG TABER!!!"
700 RETURN
710 IF N > B1 THEN 760
720 W = 1
730 PRINT
740 PRINT "COMPUTEREN TAGER";N;"OG VINDE!!!"
750 RETURN
760 P = Q - C * INT (Q / C)
770 IF P > = A1 THEN 790
780 P = A1
790 IF P < = B1 THEN 810
800 P = B1
810 N = N - P
820 PRINT
830 PRINT "COMPUTEREN TAGER";P
840 PRINT "DER ER NU";N;"TILBAGE."
850 W = 0
860 RETURN
870 PRINT
880 PRINT "HVOR MANGE TAGER DU ";
890 INPUT P
900 IF P < > 0 THEN 960
910 PRINT
920 PRINT "JEG HAR JO SAGT, AT DET ER FORBUDT AT"
930 PRINT "AT BRUGE '0'! SNYD! COMPUTEREN VINDE!!!"
940 W = 1
950 RETURN
960 IF P < > INT (P) THEN 1010
970 IF P > = A1 THEN 1000
980 IF P = N THEN 1060
990 GOTO 1010
1000 IF P < = B1 THEN 1040
1010 PRINT
1020 PRINT "UGYLDIGT ANTAL, INDTAST IGEN ";
1030 GOTO 890
1040 N = N - P
1050 IF N < > 0 THEN 1150
1060 IF M = 1 THEN 1110
1070 PRINT
1080 PRINT "ÆRGELIGT NOK, MEN DU HAR TABT!"
1090 W = 1
1100 RETURN
1110 PRINT
1120 PRINT "TILLYKKE!!! DU VINDE!!!"
1130 W = 1
1140 RETURN
1150 IF N > = 0 THEN 1180
1160 N = N + P
1170 GOTO 1010
1180 W = 0
1190 RETURN
1200 END

```

Byttepenge

I dette program lader datamaten, som om den er kassedame i din venlige nærbutik. Du fortæller, hvor meget du har købt for, hvor mange penge du betaler med, og datamaten vil automatisk (!) udskrive, hvor mange byttepenge du får retur. Er maskiner ikke vidunderlige?

Programmet er skrevet af Dennis Lunder fra People's Computer Company.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
BYTTEPENGE  
-----
```

```
JEG - DIN VENLIGE DATAMAT - VIL BEREGNE  
HVOR MANGE PENGE DU SKAL HAVE TILBAGE,  
HVIS DU KØBER EN TING DER KOSTER HØJST  
100 KRØNER.
```

```
HVAD KOSTER DEN TING DU HAR KØBT ?12.85
```

```
HVAD HAR DU BETALT MED ?50
```

```
DU SKAL HAVE 37.15 KR. TILBAGE.
```

```
1 20-KRONE SEDDEL/SEDLER  
1 10-KRONE  
1 5-KRONE  
2 1-KRONE(R)  
1 10-ØRE(R)  
1 5-ØRE(R)
```

```
TAK FOR BESØGET, OG PÅ GENSYN.....
```

```
HVAD KOSTER DEN TING DU HAR KØBT ?41.25
```

```
HVAD HAR DU BETALT MED ?100
```

```
DU SKAL HAVE 58.75 KR. TILBAGE.
```

```
1 50-KRONE SEDDEL  
1 5-KRONE  
3 1-KRONE(R)  
3 25-ØRE(R)
```

```
TAK FOR BESØGET, OG PÅ GENSYN.....
```

```
HVAD KOSTER DEN TING DU HAR KØBT ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          BYTTEPENGE"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "JEG - DIN VENLIGE DATAMAT - VIL BEREGNE"  
160 PRINT "HVOR MANGE PENGE DU SKAL HAVE TILBAGE,"  
170 PRINT "HVIS DU KØBER EN TING DER KOSTER HØJST"  
180 PRINT "100 KRØNER."  
190 PRINT  
200 PRINT "HVAD KOSTER DEN TING DU HAR KØBT ";  
210 INPUT A  
220 PRINT  
230 PRINT "HVAD HAR DU BETALT MED ";  
240 INPUT P  
250 PRINT  
260 C = INT ((P - A) * 100 + .5) / 100  
270 M = C  
280 IF C < > 0 THEN 310  
290 PRINT "DER VAR LIGE PENGE, TAK!."  
300 GOTO 190  
310 IF C > 0 THEN 350  
320 PRINT "DESVÆRRE....."  
330 PRINT "DU HAR GIVET MIG";A - P;"KR. FOR LIDT."  
340 GOTO 190  
350 PRINT "DU SKAL HAVE";C;"KR. TILBAGE."  
360 PRINT  
370 D = INT (C / 50)  
380 IF D = 0 THEN 400  
390 PRINT D;"50-KRONE SEDDEL"  
400 C = M - (D * 50)  
410 E = INT (C / 20)  
420 IF E = 0 THEN 440  
430 PRINT E;"20-KRONE SEDDEL/SEDLER"  
440 C = M - (D * 50 + E * 20)  
450 F = INT (C / 10)  
460 IF F = 0 THEN 480  
470 PRINT F;"10-KRONE"  
480 C = M - (D * 50 + E * 20 + F * 10)  
490 F5 = INT (C / 5)  
500 IF F5 = 0 THEN 520  
510 PRINT F5;" 5-KRONE"  
520 C = M - (D * 50 + E * 20 + F * 10 + F5 * 5)  
530 F1 = INT (C)  
540 IF F1 = 0 THEN 560  
550 PRINT F1;" 1-KRONE(R)"  
560 C = (P - A) - INT (P - A)  
570 C = INT ((C * 100 + .5))  
580 N = C  
590 G = INT (C / 25)  
600 IF G = 0 THEN 620  
610 PRINT G;"25-ØRE(R)"  
620 C = N - (G * 25)  
630 H = INT (C / 10)  
640 IF H = 0 THEN 660  
650 PRINT H;"10-ØRE(R)"  
660 C = N - (G * 25 + H * 10)  
670 I = INT (C / 5)  
680 IF I = 0 THEN 700  
690 PRINT I;" 5-ØRE(R)"  
700 C = N - (G * 25 + H * 10 + I * 5)  
710 K = INT (C + .5)  
720 IF K = 0 THEN 740  
730 PRINT "DESVÆRRE FINDES DER IKKE 1- OG 2-ØRER."  
740 PRINT  
750 PRINT "TAK FOR BESØGET, OG PÅ GENSYN....."  
760 PRINT  
770 PRINT  
780 GOTO 200  
790 END
```

Ciffer

Inden spillet starter, skal spilleren skrive 30 cifre (0-, 1- eller 2-taller) ned på en række. Programmet anvender en form for mønstergenkendelse og skal forsøge at gætte det næste ciffer på din liste.

Datamaten beder om 10 cifre ad gangen. Den gætter altid først og sammenligner bagefter sit gæt med det ciffer, som spilleren har indtastet. Ved rent held (eller sandsynlighed) bør datamaten ramme rigtigt 10 gange ud af 30. Det er lidt uhyggeligt, hvor ofte den klarer sig bedre end dette!

Programmet stammer fra Dartmouth. Den oprindelige forfatter er ukendt.

```
MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                        C I F F E R
-----
```

DETTE ER ET SPIL, HVOR DER SKAL GÆTTES.....

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

TAG VENLIGST ET STYKKE PAPIR OG SKRIV CIFRENE 0, 1 OG 2 NED 30 GANGE I TILFÆLDIG RÆKKEFØLGE.

ARRANGER DEREFTER TALLENE I TRE RÆKKER MED 10 CIFRE I HVER. SENERE VIL JEG NEMLIG SPØRGE EFTER 10 CIFRE AD GANGEN!

JEG VIL ALTID GÆTTE FØRST, OG DEREFTER SAMMENLIGNE MED DINE TAL FOR AT SE OM JEG HAVDE RET.

UD FRA SANDSYNLIGHEDSBEREGNING SKULLE JEG GÆTTE RIGTIGT 10 GANGE. MEN JEG REGNER MED AT KLARE MIG MEGET BEDRE!!!!!!!

INDTAST VENLIGST 10 CIFRE ?1,2,1,2,0,0,1,2,1,2

MIT GÆT	DIT TAL	RESULTAT	ANTAL RIGTIGE
1	1	RIGTIG	1
1	2	FORKERT	1
1	1	RIGTIG	2
2	2	RIGTIG	3
1	0	FORKERT	3
2	0	FORKERT	3
2	1	FORKERT	3
2	2	RIGTIG	4
1	1	RIGTIG	5
2	2	RIGTIG	6

INDTAST VENLIGST 10 CIFRE ?1,0,2,1,2,0,2,1,0,2

MIT GÆT	DIT TAL	RESULTAT	ANTAL RIGTIGE
1	1	RIGTIG	7
2	0	FORKERT	7
2	2	RIGTIG	8
1	1	RIGTIG	9
2	2	RIGTIG	10
1	0	FORKERT	10
1	2	FORKERT	10
1	1	RIGTIG	11
2	0	FORKERT	11
2	2	RIGTIG	12

INDTAST VENLIGST 10 CIFRE ?1,2,0,2,1,2,0,1,2,1

MIT GÆT	DIT TAL	RESULTAT	ANTAL RIGTIGE
1	1	RIGTIG	13
2	2	RIGTIG	14
1	0	FORKERT	14
1	2	FORKERT	14
1	1	RIGTIG	15
2	2	RIGTIG	16
1	0	FORKERT	16
1	1	RIGTIG	17
2	2	RIGTIG	18
1	1	RIGTIG	19

JEG GÆTTEDE MERE END 1/3 AF DINE CIFRE!!!!
JEG HAR VUNDET!!!!!!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?J

INDTAST VENLIGST 10 CIFRE ?0,1,2,0,1,1,2,0,0,0

MIT GÆT	DIT TAL	RESULTAT	ANTAL RIGTIGE
0	0	RIGTIG	1
1	1	RIGTIG	2
1	2	FORKERT	2
0	0	RIGTIG	3
1	1	RIGTIG	4
0	1	FORKERT	4
0	2	FORKERT	4
0	0	RIGTIG	5
1	0	FORKERT	5
1	0	FORKERT	5

INDTAST VENLIGST 10 CIFRE ?1,1,2,1,2,2,1,2,1,0

MIT GÆT	DIT TAL	RESULTAT	ANTAL RIGTIGE
1	1	RIGTIG	6
0	1	FORKERT	6
0	2	FORKERT	6
0	1	FORKERT	6
0	2	FORKERT	6
0	2	FORKERT	6
0	1	FORKERT	6
0	2	FORKERT	6
0	1	FORKERT	6
0	2	FORKERT	6
0	1	FORKERT	6
0	0	RIGTIG	7

INDTAST VENLIGST 10 CIFRE ?1,2,0,2,1,2,0,2,1,0

MIT GÆT	DIT TAL	RESULTAT	ANTAL RIGTIGE
1	1	RIGTIG	8
0	2	FORKERT	8
0	0	RIGTIG	9
1	2	FORKERT	9
1	1	RIGTIG	10
0	2	FORKERT	10
0	0	RIGTIG	11
1	2	FORKERT	11
1	1	RIGTIG	12
0	0	RIGTIG	13

JEG GÆTTEDE MERE END 1/3 AF DINE CIFRE!!!!
JEG HAR VUNDET!!!!!!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

```

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                      CIFFER"
130 PRINT "-----"
140 PRINT
150 PRINT "DETTE ER ET SPIL, HVOR DER SKAL GÆTTES....."
160 PRINT
170 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
180 INPUT AS$
190 IF LEFT$(AS$,1) = "N" THEN 320
200 PRINT
210 PRINT "TAG VENLIGST ET STYKKE PAPIR OG SKRIV CIFRENE 0, 1 OG 2"
220 PRINT "NED 30 GANGE I TILFÆLDIG RÆKKEFØLGE."
230 PRINT
240 PRINT "ARRANGER DEREFTER TALLENE I TRE RÆKKER MED 10 CIFRE I HVER."
250 PRINT "SENERE VIL JEG NEMLIG SPØRGE EFTER 10 CIFRE AD GANGEN!"
260 PRINT
270 PRINT "JEG VIL ALTID GÆTTE FØRST, OG DEREFTER SAMMENLIGNE MED DINE"
280 PRINT "TAL FOR AT SE OM JEG HAVDE RET."
290 PRINT
300 PRINT "UD FRA SANDSYNLIGHEDSBeregning SKULLE JEG GÆTTE RIGTIGT 10"
310 PRINT "GANGE. MEN JEG REGNER MED AT KLARE MIG MEGET BEDRE!!!!!!"
320 READ A,B,C
330 DATA 0,1,3
340 DIM M(26,2),K(2,2),L(8,2),N(10)
350 FOR I = 0 TO 26: FOR J = 0 TO 2:M(I,J) = 1: NEXT J: NEXT I
360 FOR I = 0 TO 2: FOR J = 0 TO 2:K(I,J) = 9: NEXT J: NEXT I
370 FOR I = 0 TO 8: FOR J = 0 TO 2:L(I,J) = 3: NEXT J: NEXT I
380 L(0,0) = 2:L(4,1) = 2:L(8,2) = 2
390 Z = 26:Z1 = 8:Z2 = 2
400 X = 0
410 FOR T = 1 TO 3
420 PRINT
430 PRINT "INDTAST VENLIGST 10 CIFRE ";
440 INPUT N(1),N(2),N(3),N(4),N(5),N(6),N(7),N(8),N(9),N(10)
450 FOR I = 1 TO 10
460 W = N(I) - 1
470 IF W = SGN(W) THEN 510
480 PRINT "BRUG KUN CIFRENE '0', '1' ELLER '2'."
490 PRINT "PRØV EN GANG TIL."
500 GOTO 420
510 NEXT I
520 PRINT
530 PRINT "MIT GÆT","DIT TAL","RESULTAT","ANTAL RIGTIGE"
540 PRINT
550 FOR U = 1 TO 10
560 N = N(U)
570 S = 0
580 FOR J = 0 TO 2
590 S1 = A * K(Z2,J) + B * L(Z1,J) + C * M(Z,J)
600 IF S > S1 THEN 650
610 IF S < S1 THEN 630
620 IF RND(1) < .5 THEN 650
630 S = S1
640 G = J
650 NEXT J
660 PRINT G,N(U),
670 IF G = N(U) THEN 700
680 PRINT "FORKERT",X
690 GOTO 770
700 X = X + 1
710 PRINT "RIGTIG",X
720 M(Z,N) = M(Z,N) + 1
730 L(Z1,N) = L(Z1,N) + 1
740 K(Z2,N) = K(Z2,N) + 1
750 Z = Z - INT(Z / 9) * 9
760 Z = 3 * Z + N(U)
770 Z1 = Z - INT(Z / 9) * 9
780 Z2 = N(U)
790 NEXT U
800 NEXT T
810 PRINT
820 IF X > 10 THEN 870
830 IF X < 10 THEN 900
840 PRINT "JEG GÆTTEDE NETOP 1/3 AF DINE CIFRE!"
850 PRINT "INGEN AF OS HAR VUNDET....."
860 GOTO 920
870 PRINT "JEG GÆTTEDE MERE END 1/3 AF DINE CIFRE!!!!"
880 PRINT "JEG HAR VUNDET!!!!!"
890 GOTO 920
900 PRINT "JEG HAR GÆTTET MINDRE END 1/3 AF DINE CIFRE...."
910 PRINT "DU SLOG MIG, OG HAR VUNDET.....TILLYKKE!!!"
920 PRINT
930 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
940 INPUT AS$
950 IF LEFT$(AS$,1) = "J" THEN 350
960 PRINT
970 PRINT "TAK FOR SPILLET.....FARVEL FOR DENNEGANG....."
980 END

```


Craps

Dette program simulerer det populære amerikanske terningspil Craps med de regler, som gælder i Nevada. Det spilles med to terninger, og reglerne er:

1. Bliver der slået 7 eller 11 i første kast, betyder det gevinst.
2. Slås 2, 3 eller 12 i første kast, er indsatsen tabt.
3. Alle andre værdier bliver til dine »points«. Du kan derfor kaste igen. Slår du nu samme antal øjne som dit »point-slag«, så har du vundet. Hvis du slår 7, taber du, og en ny spiller skal slå.

Denne version af Craps er modificeret af Steve North fra Creative Computing. Den er baseret på et originalt program, som en dag pludselig dukkede op på en datamat hos DEC.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                    CRAPS
-----
7, 11 I FØRSTE KAST GIVER GEVINST.
2, 3, 12 TABER.
4, 5, 6, 8, 9, 10 GIVER POINT KAST.
HVOR MEGET VIL DU SATSE ?24
.....TERNINGERNE ER KASTET.....
4 POINTSLAG!....JEG SLÅR IGEN...
2 INGEN POINT - JEG SLÅR IGEN...
12 INGEN POINT - JEG SLÅR IGEN...
8 INGEN POINT - JEG SLÅR IGEN...
11 INGEN POINT - JEG SLÅR IGEN...
6 INGEN POINT - JEG SLÅR IGEN...
12 INGEN POINT - JEG SLÅR IGEN...
2 INGEN POINT - JEG SLÅR IGEN...
8 INGEN POINT - JEG SLÅR IGEN...
8 INGEN POINT - JEG SLÅR IGEN...
5 INGEN POINT - JEG SLÅR IGEN...
9 INGEN POINT - JEG SLÅR IGEN...
9 INGEN POINT - JEG SLÅR IGEN...
5 INGEN POINT - JEG SLÅR IGEN...
7 CRAPS.
DU HAR TABT 24 KR.
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
DU HAR NU ET TAB PÅ 24 KR.
HVOR MEGET VIL DU SATSE ?24
.....TERNINGERNE ER KASTET.....
7 .....DU ER ET NATURTALENT.
*** GEVINST.
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                    CRAPS"
130 PRINT "-----"
140 PRINT
150 R = 0
160 PRINT "7, 11 I FØRSTE KAST GIVER GEVINST."
170 PRINT
180 PRINT "2, 3, 12 TABER."
190 PRINT
200 PRINT "4, 5, 6, 8, 9, 10 GIVER POINT KAST."
210 PRINT
220 PRINT "HVOR MEGET VIL DU SATSE ";
230 INPUT F
240 PRINT
250 PRINT ".....TERNINGERNE ER KASTET....."
260 PRINT
270 E = INT (7 * RND (1))
280 S = INT (7 * RND (1))
290 X = E + S
300 IF X = 7 THEN 450
310 IF X = 11 THEN 450
320 IF X = 1 THEN 270
330 IF X = 2 THEN 540
340 IF X = 0 THEN 270
350 IF X = 2 THEN 610
360 IF X = 3 THEN 610
EN 7 'ER GIVER GEVINST
DU HAR VUNDET 24 KR.
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
DU HAR HVERKEN TABT ELLER VUNDET NOGET.
HVOR MEGET VIL DU SATSE ?10
.....TERNINGERNE ER KASTET.....
7 .....DU ER ET NATURTALENT.
*** GEVINST.
EN 7 'ER GIVER GEVINST
DU HAR VUNDET 10 KR.
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
DU HAR INDTIL NU VUNDET 10 KR.
HVOR MEGET VIL DU SATSE ?10
.....TERNINGERNE ER KASTET.....
6 POINTSLAG!....JEG SLÅR IGEN...
3 INGEN POINT - JEG SLÅR IGEN...
10 INGEN POINT - JEG SLÅR IGEN...
4 INGEN POINT - JEG SLÅR IGEN...
8 INGEN POINT - JEG SLÅR IGEN...
7 CRAPS.
DU HAR TABT 10 KR.
VIL DU SPILLE IGEN ('J' ELLER 'N') ?N
DU HAR HVERKEN TABT ELLER VUNDET NOGET.
TILLYKKE! DET GIK LIGE OP.....
IKKE DÅRLIGT FOR EN AMATØR SOM DIG!
```

```
370 IF X = 12 THEN 610
380 IF X = 5 THEN 690
390 IF X = 6 THEN 690
400 IF X = 8 THEN 690
410 IF X = 9 THEN 690
420 IF X = 10 THEN 690
430 IF X = 4 THEN 690
440 GOTO 1240
450 PRINT X;".....DU ER ET NATURTALENT."
460 PRINT
470 PRINT "*** GEVINST."
480 PRINT
490 PRINT "EN";X;"ER GIVER GEVINST"
500 PRINT
510 PRINT "DU HAR VUNDET";F;"KR."
520 PRINT
530 GOTO 670
540 IF X < 10 THEN PRINT " ";
550 PRINT X;"SLANGEØJNE. DU TABER!"
560 PRINT
570 PRINT "DU HAR TABT";F;"KR."
580 PRINT
590 F = 0 - F
600 GOTO 670
610 IF X < 10 THEN PRINT " ";
620 PRINT X;"CRAPS."
630 PRINT
640 PRINT "DU HAR TABT";F;"KR."
650 PRINT
660 F = 0 - F
670 R = R + F
680 GOTO 990
690 IF X < 10 THEN PRINT " ";
700 PRINT X;"POINTSLAG!....JEG SLÅR IGEN..."
710 H = INT (7 * RND (1))
720 Q = INT (7 * RND (1))
730 O = H + Q
740 IF O = 1 THEN 710
750 IF O = 7 THEN 810
760 IF O = 0 THEN 710
770 IF O = X THEN 890
780 IF O < 10 THEN PRINT " ";
790 PRINT O;"INGEN POINT - JEG SLÅR IGEN..."
800 GOTO 710
810 IF O < 10 THEN PRINT " ";
820 PRINT O;"CRAPS."
830 PRINT
840 PRINT "DU HAR TABT";F;"KR."
850 F = 0 - F
860 PRINT
870 GOTO 670
880 GOTO 990
890 IF X < 10 THEN PRINT " ";
900 PRINT X;"VINDER!.....TILLYKKE!"
910 PRINT
920 IF X < 10 THEN PRINT " ";
930 PRINT X;"GIVER 2 GANGE INDSATS."
940 PRINT
950 PRINT "LAD MIG SE ... DET BLIVER";2 * F;"KR."
960 F = 2 * F
970 PRINT
980 GOTO 670
990 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1000 INPUT AS
1010 PRINT
1020 IF R < 0 THEN 1050
1030 IF R > 0 THEN 1070
1040 IF R = 0 THEN 1090
1050 PRINT "DU HAR NU ET TAB PÅ";(- 1) * R;"KR."
1060 GOTO 1100
1070 PRINT "DU HAR INDTIL NU VUNDET";R;"KR."
1080 GOTO 1100
1090 PRINT "DU HAR HVERKEN TABT ELLER VUNDET NOGET."
1100 IF LEFT$(AS,1) = "J" THEN 210
1110 IF R < 0 THEN 1140
1120 IF R > 0 THEN 1170
1130 IF R = 0 THEN 1200
1140 PRINT
1150 PRINT "ÆRGELIGT NOK.... DU ER FLAD. PRØV IGEN!"
1160 GOTO 1240
1170 PRINT
1180 PRINT "TILLYKKE! DU ER FØDT VINDER! PRØV IGEN!"
1190 GOTO 1240
1200 PRINT
1210 PRINT "TILLYKKE! DET GIK LIGE OP....."
1220 PRINT
1230 PRINT "IKKE DÅRLIGT FOR EN AMATØR SOM DIG!"
1240 END
```

Dam

Dette program spiller Dam. Datamaskinens brikker vises med »X«, mens dine brikker vises med »0«. Et træk foretages ved at indtaste koordinaterne for den brik, der skal flyttes (X,Y). Udgangspunktet (0,0) er i brættets nederste venstre hjørne. X angiver antal felter til højre for udgangspunktet (dvs. antal kolonner), mens Y angiver antal felter opad (dvs. rækker). Derefter bestemmer du på samme måde, hvor brikken skal flyttes hen.

Den oprindelige version af dette program er skrevet af Alan Segal, men denne version var ikke i stand til at finde ud af (og tilade) dobbelte eller tre-dobbelte hop. Hvis man forsøgte et sådant hop, kunne brikken forsvinde helt!

Steve North fra Creative Computing rettede dette, og Lawrence Neal bidrog med modifikationer, der gjorde det muligt for programmet at fortælle, hvem der havde vundet spillet. Datamaten spiller ikke særlig godt, men det lader vi være op til DIG at gøre noget ved!

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

DAM

VELKOMMEN TIL SPILLET DAM. DATAMATENS
BRIKKES VISES SOM 'X', DINE BRIKKER
SOM '0'. DATAMATEN BEGYNDER ALTID.

BRÆTTET ER OPDELT SOM ET KOORDINATYSY-
STEM, SÅDAN AT:

(0,0) ER NEDERSTE VENSTRE HJØRNE
(0,7) ER ØVERSTE VENSTRE HJØRNE
(7,0) ER NEDERSTE HØJRE HJØRNE
(7,7) ER ØVERSTE HØJRE HJØRNE

COMPUTEREN SKRIVER '+TIL', NÅR DU KAN
HOPPE OVER FLERE FJENDTLIGE BRIKKER.

FRA 1 , 5 TIL 0 , 4

```

. X . X . X . X
X . X . X . X .
. . . X . X . X
X . . . . . . .
. . . . . . . .
0 . 0 . 0 . 0 .
. 0 . 0 . 0 . 0
0 . 0 . 0 . 0 .

```

FRA ?2,2
TIL ?3,3

FRA 0 , 6 TIL 1 , 5

```

. X . X . X . X
. . X . X . X .
. X . X . X . X
X . . . . . . .
. . . 0 . . . .
0 . . . 0 . 0 .
. 0 . 0 . 0 . 0
0 . 0 . 0 . 0 .

```

FRA ?0,2
TIL ?1,3

FRA 0 , 4 TIL 2 , 2

```

. X . X . X . X
. . X . X . X .
. X . X . X . X
. . . . . . . .
. . . 0 . . . .
. . X . 0 . 0 .
. 0 . 0 . 0 . 0
0 . 0 . 0 . 0 .

```

FRA ?3,1
TIL ?1,3

+TIL ?-1,-1

FRA 1 , 5 TIL 0 , 4

```

. X . X . X . X
. . X . X . X .
. . . X . X . X
X . . . . . . .
. 0 . 0 . . . .
. . . . 0 . 0 .
. 0 . . . 0 . 0
0 . 0 . 0 . 0 .

```

FRA ?4,0
TIL ?3,1

FRA 0 , 4 TIL 2 , 2
TIL 4 , 0

```

. X . X . X . X
. . X . X . X .
. . . X . X . X
. . . . . . . .
. . . 0 . . . .
. . . . 0 . 0 .
. 0 . . . 0 . 0
0 . 0 . X* . 0 .

```

FRA ?6,2
TIL ?5,3

FRA 4 , 0 TIL 6 , 2
TIL 4 , 4 TIL 2 , 2

```

. X . X . X . X
. . X . X . X .
. . . X . X . X
. . . . . . . .
. . . . . . . .
. . X* . 0 . . .
. 0 . . . . 0
0 . 0 . . . 0 .

```

FRA ?1,1
TIL ?3,1

TIL ?

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          DAM"
130 PRINT "-----"
140 PRINT
150 PRINT "VELKOMMEN TIL SPILLET DAM. DATAMATENS"
160 PRINT "BRIKKES VISES SOM 'X', DINE BRIKKER"
170 PRINT "SOM 'O'. DATAMATEN BEGYNDER ALTID."
180 PRINT
190 PRINT "BRÆTTET ER OPDELT SOM ET KOORDINATY--"
200 PRINT "STEM, SÅDAN AT:"
210 PRINT
220 PRINT "(0,0) ER NEDERSTE VENSTRE HJØRNE"
230 PRINT "(0,7) ER ØVERSTE VENSTRE HJØRNE"
240 PRINT "(7,0) ER NEDERSTE HØJRE HJØRNE"
250 PRINT "(7,7) ER ØVERSTE HØJRE HJØRNE"
260 PRINT
270 PRINT "COMPUTEREN SKRIVER '+TIL', NÅR DU KAN"
280 PRINT "HOPPE OVER FLERE FJENDTLIGE BRIKKER."
290 PRINT
300 PRINT
310 DIM R(4),S(7,7)
320 G = - 1
330 R(0) = - 99
340 DATA 1,0,1,0,0,0,-1,0,0,1,0,0,0,-1,0,-1,15
350 FOR X = 0 TO 7
360 FOR Y = 0 TO 7
370 READ J
380 IF J = 15 THEN 410
390 S(X,Y) = J
400 GOTO 430
410 RESTORE
420 READ S(X,Y)
430 NEXT Y
440 NEXT X
450 FOR X = 0 TO 7
460 FOR Y = 0 TO 7
470 IF S(X,Y) > - 1 THEN 500
480 IF S(X,Y) = - 1 THEN GOSUB 1470
490 IF S(X,Y) = - 2 THEN GOSUB 1520
500 NEXT Y
510 NEXT X
520 GOTO 760
530 U = X + A
540 V = Y + B
550 IF U < 0 OR U > 7 OR V < 0 OR V > 7 THEN 620
560 IF S(U,V) = 0 THEN GOSUB 630: GOTO 620
570 IF S(U,V) < 0 THEN 620
580 U = U + A
590 V = V + B
600 IF U < 0 OR V < 0 OR U > 7 OR V > 7 THEN 620
610 IF S(U,V) = 0 THEN GOSUB 630
620 RETURN
630 IF V = 0 AND S(X,Y) = - 1 THEN Q = Q + 2
640 IF ABS(Y - V) = 2 THEN Q = Q + 5
650 IF Y = 7 THEN Q = Q - 2
660 IF U = 0 OR U = 7 THEN Q = Q + 1
670 FOR C = - 1 TO 1 STEP 2
680 IF U + C < 0 OR U + C > 7 OR V + G < 0 THEN 720
690 IF S(U + C,V + G) < 0 THEN Q = Q + 1: GOTO 720
700 IF U - C < 0 OR U - C > 7 OR V - G > 7 THEN 720
710 IF S(U + C,V + G) > 0 AND (S(U - C,V - G) = 0 OR
(U - C = X AND V - G = Y)) THEN Q = Q - 2
720 NEXT C
730 IF Q > R(0) THEN R(0) = Q:R(1) = X:R(2) = Y:R(3)
= U:R(4) = V
740 Q = 0
750 RETURN
760 IF R(0) = - 99 THEN 1410
770 PRINT "FRA";R(1);";";R(2);
780 PRINT "TIL";R(3);";";R(4)
790 R(0) = - 99
800 IF R(4) = 0 THEN S(R(3),R(4)) = - 2: GOTO 820
810 S(R(3),R(4)) = S(R(1),R(2))
820 S(R(1),R(2)) = 0
830 IF ABS(R(1) - R(3)) < > 2 THEN 940
840 S((R(1) + R(3)) / 2,(R(2) + R(4)) / 2) = 0
850 X = R(3):Y = R(4)
860 IF S(X,Y) = - 1 THEN B = - 2: FOR A = - 2 TO
2 STEP 4: GOSUB 900: NEXT A
870 IF S(X,Y) = - 2 THEN FOR A = - 2 TO 2 STEP 4:
FOR B = - 2 TO 2 STEP 4: GOSUB 900: NEXT B: NEXT A
880 IF R(0) < > - 99 THEN PRINT
"TIL";R(3);";";R(4);:R(0) = - 99: GOTO 800
890 GOTO 940
900 U = X + A:V = Y + B
910 IF U < 0 OR U > 7 OR V < 0 OR V > 7 THEN 930

```


```

920 IF S(U,V) = 0 AND S(X + A / 2,Y + B / 2) > 0
THEN GOSUB 630
930 RETURN
940 PRINT : PRINT
950 FOR Y = 7 TO 0 STEP - 1
960 FOR X = 0 TO 7
970 I = 4 * X + 1
980 PRINT TAB(I);
990 IF S(X,Y) = 0 THEN PRINT ".";
1000 IF S(X,Y) = 1 THEN PRINT "O";
1010 IF S(X,Y) = - 1 THEN PRINT "X";
1020 IF S(X,Y) = - 2 THEN PRINT "X*";
1030 IF S(X,Y) = 2 THEN PRINT "O*";
1040 NEXT X
1050 PRINT : PRINT
1060 NEXT Y
1070 PRINT
1080 FOR L = 0 TO 7
1090 FOR M = 0 TO 7
1100 IF S(L,M) = 1 OR S(L,M) = 2 THEN Z = 1
1110 IF S(L,M) = - 1 OR S(L,M) = - 2 THEN T = 1
1120 NEXT M
1130 NEXT L
1140 IF Z < > 1 THEN 1410
1150 IF T < > 1 THEN 1440
1160 T = 0:Z = 0
1170 PRINT " FRA ";
1180 INPUT E,H
1190 X = E:Y = H
1200 IF S(X,Y) < = 0 THEN 1170
1210 PRINT " TIL ";
1220 INPUT A,B
1230 PRINT
1240 X = A:Y = B
1250 IF S(X,Y) = 0 AND ABS(A - E) < = 2 AND ABS
(A - E) = ABS(B - H) THEN 1270
1260 GOTO 1210
1270 I = 46
1280 S(A,B) = S(E,H)
1290 S(E,H) = 0
1300 IF ABS(E - A) < > 2 THEN 1390
1310 S((E + A) / 2,(H + B) / 2) = 0
1320 PRINT "+TIL ";
1330 INPUT A1,B1
1340 PRINT
1350 IF A1 < 0 THEN 1390
1360 IF S(A1,B1) < > 0 OR ABS(A1 - A) < > 2 OR
ABS(B1 - B) < > 2 THEN 1320
1370 E = A:H = B:A = A1:B = B1:I = I + 15
1380 GOTO 1280
1390 IF B = 7 THEN S(A,B) = 2
1400 GOTO 450
1410 PRINT
1420 PRINT "JEG HAR VUNDET!"
1430 GOTO 1450
1440 PRINT "DU HAR VUNDET!!!!!"
1450 PRINT
1460 END
1470 FOR A = - 1 TO 1 STEP 2
1480 B = G
1490 GOSUB 530
1500 NEXT A
1510 RETURN
1520 FOR A = - 1 TO 1 STEP 2
1530 FOR B = - 1 TO 1 STEP 2
1540 GOSUB 530
1550 NEXT B
1560 NEXT A
1570 RETURN
1580 END

```

Dam for en

I dette spil eller hovedbrud placeres 48 brikker på de to yderste rækker på alle fire sider af et almindeligt 64-felters spille- eller skak-bræt:



Det gælder om at fjerne så mange brikker som muligt ved at foretage diagonale hop (som i det almindelige damspil).

Det er meget let at få fjernet 30 til 39 brikker, sværere at få fjernet mellem 40 og 44 og en virkelig præstation at fjerne 45 til 47 brikker.

Spillet er opfundet og programmeret af David Ahl.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

DAM FOR EN

'DAM FOR EN' ER SKREVET AF DAVID AHL.

48 DAMBRIKKER ER PLACERET PÅ DE TO YDERSTE FELTER PÅ ET ALMINDELIGT 64-FELTERS DAMBRÆT.

SPILLER DREJER SIG OM AT FJERNE SÅ MANGE BRIKKER SOM MULGT VED AT GÅ DIAGONALT OVER BRIKKERNE (SOM I ALMINDELIGT DAM). BRUG DET NUMMEREREDE BRÆT SOM REFERENCE NÅR DU SKAL INDASTE HVILKET FELT DU SKAL FLYTTE FRA OG TIL.

DER UDSKRIVES EN OVERSIGT OVER BRÆTTET HVER GANG DU HAR FLYTTET EN BRIK. PÅ BRÆTTET BETYDER '1' EN BRIK, MENS '0' VISER ET TOMT FELT.

NÅR DU IKKE KAN FLYTTE FLERE BRIKKER SKAL DU INDASTE ET '0' SOM SVAR PÅ SPØRGSMALET: 'HOP FRA ?'

SÅDAN SER DET NUMMEREREDE SPILLEBRÆT UD

1	2	3	4	5	6	7	8
9	10	11	12	13	14	15	16
17	18	19	20	21	22	23	24
25	26	27	28	29	30	31	32
33	34	35	36	37	38	39	40
41	42	43	44	45	46	47	48
49	50	51	52	53	54	55	56
57	58	59	60	61	62	63	64

OG SÅDAN SER STARTBRÆTTET UD:

1	1	1	1	1	1	1	1
1	1	1	1	1	1	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	1	1	1	1	1	1
1	1	1	1	1	1	1	1

HOP FRA ?1
HOP TIL ?19

0	1	1	1	1	1	1	1
1	0	1	1	1	1	1	1
1	1	1	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	1	1	1	1	1	1
1	1	1	1	1	1	1	1

HOP FRA ?2
HOP TIL ?20

0	0	1	1	1	1	1	1
1	0	0	1	1	1	1	1
1	1	1	1	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	1	1	1	1	1	1
1	1	1	1	1	1	1	1

HOP FRA ?3
HOP TIL ?21

0	0	0	1	1	1	1	1
1	0	0	0	1	1	1	1
1	1	1	1	1	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	1	1	1	1	1	1
1	1	1	1	1	1	1	1

HOP FRA ?8
HOP TIL ?22

0	0	0	1	1	1	1	0
1	0	0	0	1	1	0	1
1	1	1	1	1	1	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	1	1	1	1	1	1
1	1	1	1	1	1	1	1

HOP FRA ?16
HOP TIL ?30

0	0	0	1	1	1	1	0
1	0	0	0	1	1	0	0
1	1	1	1	1	1	0	1
1	1	0	0	0	1	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	0	1	1
1	1	1	1	1	1	1	1
1	1	1	1	1	1	1	1

HOP FRA ?64
HOP TIL ?46

0	0	0	1	1	1	1	0
1	0	0	0	1	1	0	0
1	1	1	1	1	1	0	1
1	1	0	0	0	1	1	1
1	1	0	0	0	0	1	1
1	1	0	0	0	1	1	1
1	1	1	1	1	1	0	1
1	1	1	1	1	1	1	0

HOP FRA ?5
HOP TIL ?23

0	0	0	1	0	1	1	0
1	0	0	0	1	0	0	0
1	1	1	1	1	1	1	1
1	1	0	0	1	1	1	1
1	1	0	0	0	1	1	1
1	1	0	0	0	1	1	1
1	1	0	0	0	1	1	1
1	1	1	1	1	1	0	1
1	1	1	1	1	1	1	0

HOP FRA ?57
HOP TIL ?43

0	0	0	1	0	1	1	0
1	0	0	0	1	0	0	0
1	1	1	1	1	1	1	1
1	1	0	0	0	1	1	1
1	1	0	0	0	1	1	1
1	1	0	0	0	1	1	1
1	0	1	1	1	1	0	1
0	1	1	1	1	1	1	0

HOP FRA ?25
HOP TIL ?11

0	0	0	1	0	1	1	0
1	0	1	0	1	0	0	0
1	0	1	1	1	1	1	1
0	1	0	0	0	1	1	1
1	1	0	0	0	0	1	1
1	1	1	0	0	1	1	1
1	0	1	1	1	1	0	1
0	1	1	1	1	1	1	0

HOP FRA ?4
HOP TIL ?18

0	0	0	0	0	1	1	0
1	0	0	0	1	0	0	0
1	1	1	1	1	1	1	1
0	1	0	0	0	1	1	1
1	1	0	0	0	1	1	1
1	1	1	0	0	1	1	1
1	0	1	1	1	1	0	1
0	1	1	1	1	1	1	0

HOP FRA ?49
HOP TIL ?35

0	0	0	0	0	1	1	0
1	0	0	0	1	0	0	0
1	1	1	1	1	1	1	1
0	1	0	0	0	1	1	1
1	1	1	0	0	0	1	1
1	0	1	0	0	1	1	1
0	0	1	1	1	1	0	1
0	1	1	1	1	1	1	0

HOP FRA ?41
HOP TIL ?27

0	0	0	0	0	1	1	0
1	0	0	0	1	0	0	0
1	1	1	1	1	1	1	1
0	1	1	0	0	1	1	1
1	0	1	0	0	0	1	1
0	0	1	0	0	1	1	1
0	0	1	1	1	1	0	1
0	1	1	1	1	1	1	0

HOP FRA ?58
HOP TIL ?44

0	0	0	0	0	1	1	0
1	0	0	0	1	0	0	0
1	1	1	1	1	1	1	1
0	1	1	0	0	1	1	1
1	0	1	0	0	0	1	1
0	0	1	1	0	1	1	1
0	0	0	1	1	1	0	1
0	0	1	1	1	1	1	0

HOP FRA ?26
HOP TIL ?12

0	0	0	0	0	1	1	0
1	0	0	1	1	0	0	0
1	1	0	1	1	1	1	1
0	0	1	0	0	1	1	1
1	0	1	0	0	0	1	1
0	0	1	1	0	1	1	1
0	0	0	1	1	1	0	1
0	0	1	1	1	1	1	0

HOP FRA ?22
HOP TIL ?4

0	0	0	1	0	1	1	0
1	0	0	1	0	0	0	0
1	1	0	1	1	0	1	1
0	0	1	0	0	1	1	1
1	0	1	0	0	0	1	1
0	0	1	1	0	1	1	1
0	0	0	1	1	1	0	1
0	0	1	1	1	1	1	0

HOP FRA ?59
HOP TIL ?45

0	0	0	1	0	1	1	0
1	0	0	1	0	0	0	0
1	1	0	1	1	0	1	1
0	0	1	0	0	1	1	1
1	0	1	0	0	0	1	1
0	0	1	1	1	1	1	1
0	0	0	0	1	1	0	1
0	0	0	1	1	1	1	0

HOP FRA ?18
HOP TIL ?36

0	0	0	1	0	1	1	0
1	0	0	1	0	0	0	0
1	0	0	1	1	0	1	1
0	0	0	0	0	1	1	1
1	0	1	1	0	0	1	1
0	0	1	1	1	1	1	1
0	0	0	0	1	1	0	1
0	0	0	1	1	1	1	0

HOP FRA ?21
HOP TIL ?3

0	0	1	1	0	1	1	0
1	0	0	0	0	0	0	0
1	0	0	1	0	0	1	1
0	0	0	0	0	1	1	1
1	0	1	1	0	0	1	1
0	0	1	1	1	1	1	1
0	0	0	0	1	1	0	1
0	0	0	1	1	1	1	0

HOP FRA ?

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          DAM FOR EN"
130 PRINT "-----"
140 PRINT
150 DIM A(64)
160 PRINT "'DAM FOR EN' ER SKREVET AF DAVID AHL."
170 PRINT
180 PRINT "48 DAMBRIKKER ER PLACERET PÅ DE TO"
190 PRINT "YDERSTE FELTER PÅ ET ALMINDELIGT 64-"
200 PRINT "FELTERS DAMBRÆT."
210 PRINT
220 PRINT "SPILLER DREJER SIG OM AT FJERNE SR MAN-"
230 PRINT "GE BRIKKER SOM MULGT VED AT GR DIAGO-"
240 PRINT "NALT OVER BRIKKERNE (SOM I ALMINDELIG"
250 PRINT "DAM). BRUG DET NUMMEREREDE BRÆT SOM"
260 PRINT "REFERENCE NÅR DU SKAL INDTASTE HVILKET"
270 PRINT "FELT DU SKAL FLYTTE FRA OG TIL."
280 PRINT
290 PRINT "DER UDSKRIVES EN OVERSIGT OVER BRÆTTET"
300 PRINT "HVER GANG DU HAR FLYTTET EN BRIK. PÅ"
310 PRINT "BRÆTTET BETYDER '1' EN BRIK, MENS '0'"
320 PRINT "VISER ET TOMT FELT."
330 PRINT
340 PRINT "NÅR DU IKKE KAN FLYTTE FLERE BRIKKER"
350 PRINT "SKAL DU INDTASTE ET '0' SOM SVAR PÅ"
360 PRINT "SPØRGSMÅLET: 'HOP FRA ?'"
370 PRINT
380 PRINT "SÅDAN SER DET NUMMEREREDE SPILLEBRÆT UD"
390 FOR I = 1 TO 64
400 IF I < 10 THEN PRINT " ";
410 IF (I - 1) / 8 = INT ((I - 1) / 8) THEN PRINT
420 PRINT I;
430 IF I = 9 THEN PRINT " ";
440 NEXT I
450 PRINT
460 PRINT
470 PRINT "OG SÅDAN SER STARTBRÆTTET UD:"
480 PRINT
490 FOR J = 1 TO 64
500 A(J) = 1
510 NEXT J
520 FOR J = 19 TO 43 STEP 8
530 FOR I = J TO J + 3
540 A(I) = 0
550 NEXT I
560 NEXT J
570 M = 0
580 GOTO 890
590 PRINT "HOP FRA ";
600 INPUT F
610 IF F = 0 THEN 980

```

```

620 PRINT "HOP TIL ";
630 INPUT T
640 PRINT
650 REM Er hop gyldigt?
660 F1 = INT ((F - 1) / 8)
670 F2 = F - 8 * F1
680 T1 = INT ((T - 1) / 8)
690 T2 = T - 8 * T1
700 IF F1 > 7 THEN 800
710 IF T1 > 7 THEN 800
720 IF F2 > 8 THEN 800
730 IF T2 > 8 THEN 800
740 IF ABS (F1 - T1) < > 2 THEN 800
750 IF ABS (F2 - T2) < > 2 THEN 800
760 IF A((T + F) / 2) = 0 THEN 800
770 IF A(F) = 0 THEN 800
780 IF A(T) = 1 THEN 800
790 GOTO 840
800 PRINT "ULOVLIGT TRÆK.....PRØV IGEN."
810 PRINT
820 GOTO 590
830 REM Nyt bræt
840 A(T) = 1
850 A(F) = 0
860 A((T + F) / 2) = 0
870 M = M + 1
880 REM Udskriv bræt
890 FOR J = 1 TO 57 STEP 8
900 FOR I = J TO J + 7
910 PRINT A(I);
920 NEXT I
930 PRINT
940 NEXT J
950 PRINT
960 GOTO 590
970 REM Slutresultatet
980 S = 0
990 FOR I = 1 TO 64
1000 S = S + A(I)
1010 NEXT I
1020 PRINT "DU FORETOG";M;"HOP OG DER ER"
1030 PRINT S;"BRIKKER TILBAGE PÅ BRÆTTET."
1040 PRINT
1050 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1060 INPUT AS
1070 IF LEFT$(AS,1) = "J" THEN 390
1080 IF LEFT$(AS,1) = "N" THEN 1110
1090 PRINT "SVAR 'JA' ELLER 'NEJ'."
1100 GOTO 1050
1110 PRINT
1120 PRINT "O.K. HÅBER DU HAR MORET DIG."
1130 END

```

Dronningeskak

Dette spil er baseret på de træk, som dronningen i skak kan foretage, dvs. enten vandrette, lodrette eller diagonale træk. I dette spil får dronningen dog kun lov til at gå mod venstre, nedad eller diagonalt nedad mod venstre.

Formålet med spillet er at placere dronningen (der anvendes kun én) i det nederste venstre hjørne, ved at du og datamaten skiftes til at trække. Den, som flytter dronningen over på feltet i det nederste venstre hjørne, har vundet.

Du trækker først, idet du placerer dronningen i et af felterne på skakbrættets øverste vandrette række eller på et af felterne i den lodrette række, der ligger længst til højre på brættet. Dette udgør dit første træk. Det kan lade sig gøre at vinde over datamaten. Men det kræver en del omtanke. Du bliver nødt til selv at finde ud af en brugbar strategi.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                    DRONNINGESKAK
-----
```

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

VI SKAL TIL AT SPILLE ET SPIL, DER ER BASERET PÅ SKAKSPILLET'S TRÆK. DRONNINGEN KAN KUN FLYTTE MOD VENSTRE, NEDAD ELLER DIAGONALT NEDAD OG TIL VENSTRE.

DU OG DATAMATEN SKIFTES TIL AT FLYTTE DRONNINGEN.

SPILLET DREJER SIG OM AT PLACERE DRONNINGEN I DET NEDERSTE VENSTRE HJØRNE AF BRÆTTET. DEN DER FÅR PLACERET DRONNINGEN HER VINDER!

DU TRÆKKER FØRST, OG DU BEGYNDER MED AT PLACERE DRONNINGEN PÅ ET FELT I DEN ØVERSTE VANDRETTE RÆKKE, ELLER I DEN HØJRE LODRETTE RÆKKE FELTER. DETTE ER DIT FØRSTE TRÆK.

DU KAN OPGIVE SPILLET VED AT INDASTE ET TRÆK PÅ 'O'.

81	71	61	51	41	31	21	11
92	82	72	62	52	42	32	22
103	93	83	73	63	53	43	33
114	104	94	84	74	64	54	44
125	115	105	95	85	75	65	55
136	126	116	106	96	86	76	66
147	137	127	117	107	97	87	77
158	148	138	128	118	108	98	88

HVOR VIL DU BEGYNDE ?44

DATAMATEN FLYTTER TIL KVADRAT 55

DIT TRÆK ?65

DATAMATEN FLYTTER TIL KVADRAT 75

DIT TRÆK ?85

DATAMATEN FLYTTER TIL KVADRAT 127

DIT TRÆK ?138

DATAMATEN FLYTTER TIL KVADRAT 158

GODT FØRSØGT, MEN DET SER UD TIL AT JEG HAR VUNDET. TAK FOR SPILLET!

ER DER FLERE DER VIL SPILLE MED ?JA

HVOR VIL DU BEGYNDE ?31

DATAMATEN FLYTTER TIL KVADRAT 75

DIT TRÆK ?95

DATAMATEN FLYTTER TIL KVADRAT 158

GODT FØRSØGT, MEN DET SER UD TIL AT JEG HAR VUNDET. TAK FOR SPILLET!

ER DER FLERE DER VIL SPILLE MED ?NEJ

DET ER IORDEN ... FARVEL OG TAK.


```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                    DRONNINGESKAK"
130 PRINT "-----"
140 PRINT
150 DIM S(64)
160 FOR I = 1 TO 64
170 READ S(I)
180 NEXT I
190 DATA 81,71,61,51,41,31,21,11
200 DATA 92,82,72,62,52,42,32,22
210 DATA 103,93,83,73,63,53,43,33
220 DATA 114,104,94,84,74,64,54,44
230 DATA 125,115,105,95,85,75,65,55
240 DATA 136,126,116,106,96,86,76,66
250 DATA 147,137,127,117,107,97,87,77
260 DATA 158,148,138,128,118,108,98,88
270 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
280 INPUT W$
290 PRINT
300 IF LEFT$(W$,1) = "N" THEN 370
310 IF LEFT$(W$,1) = "J" THEN GOTO 350
320 PRINT "SVAR VENLIGST 'JA' ELLER 'NEJ'."
330 PRINT
340 GOTO 270
350 GOSUB 1580
360 GOTO 390
370 GOSUB 1810
380 REM Er der fejl?
390 PRINT "HVOR VIL DU BEGYNDE ";
400 INPUT M1
410 IF M1 = 0 THEN 560
420 T1 = INT (M1 / 10)
430 U1 = M1 - 10 * T1
440 IF U1 = 1 THEN 500
450 IF U1 = T1 THEN 500
460 PRINT "LÆS INSTRUKTIONERNE IGEN."
470 PRINT "DU STARTER ULOVLIGT!"
480 PRINT
490 GOTO 390
500 GOSUB 740
510 PRINT "DATAMATEN FLYTTER TIL KVADRAT";M
520 IF M = 158 THEN 1320
530 PRINT "DIT TRÆK ";
540 INPUT M1
550 IF M1 < > 0 THEN 600
560 PRINT
570 PRINT "DET SER UD TIL, AT DU GIVER OP!"
580 PRINT
590 GOTO 1480
600 IF M1 < = M THEN 1160
610 T1 = INT (M1 / 10)
620 U1 = M1 - 10 * T1
630 P = U1 - U
640 IF P < > 0 THEN 680
650 L = T1 - T
660 IF L < = 0 THEN 1160
670 GOTO 500
680 IF T1 - T < > P THEN 700
690 GOTO 500
700 IF T1 - T < > 2 * P THEN 1160
710 GOTO 500
720 PRINT "TÆLLER SOM DIT FØRSTE TRÆK."
730 REM Dan datamatens træk
740 IF M1 = 41 THEN 970
750 IF M1 = 44 THEN 970
760 IF M1 = 73 THEN 970
770 IF M1 = 75 THEN 970
780 IF M1 = 126 THEN 970
790 IF M1 = 127 THEN 970
800 IF M1 = 158 THEN 1220
810 C = 0
820 FOR K = 7 TO 1 STEP - 1
```

```

830 U = U1
840 T = T1 + K
850 GOSUB 1380
860 IF C = 1 THEN 950
870 U = U + K
880 GOSUB 1380
890 IF C = 1 THEN 950
900 T = T + K
910 GOSUB 1380
920 IF C = 1 THEN 950
930 NEXT K
940 GOTO 970
950 C = 0
960 RETURN
970 GOSUB 1000
980 RETURN
990 REM Tilfældigt træk
1000 Z = RND (1)
1010 IF Z > .6 THEN 1110
1020 IF Z > .3 THEN 1070
1030 U = U1
1040 T = T1 + 1
1050 M = 10 * T + U
1060 RETURN
1070 U = U1 + 1
1080 T = T1 + 2
1090 M = 10 * T + U
1100 RETURN
1110 U = U1 + 1
1120 T = T1 + 1
1130 M = 10 * T + U
1140 RETURN
1150 REM Ulovligt træk
1160 PRINT
1170 PRINT "D U E R E N S N Y D E P E L S"
1180 PRINT
1190 PRINT "PRØV IGEN....";
1200 GOTO 540
1210 REM Spiller vinder
1220 PRINT
1230 PRINT "T I L L Y K K E....."
1240 PRINT
1250 PRINT "DU HAR VUNDET - VIRKELIG GODT SPIL!"
1260 PRINT "DET SER UD TIL AT JEG HAR MØDT MIN"
1270 PRINT "OVERMAND. TAK FOR SPILLET OVER-"
1280 PRINT "MAND. JEG KAN JO IKKE VINDE ALTID."
1290 PRINT
1300 GOTO 1480
1310 REM Datamaten vinder
1320 PRINT
1330 PRINT "GODT FORSØGT, MEN DET SER UD TIL AT"
1340 PRINT "JEG HAR VUNDET. TAK FOR SPILLET!"
1350 PRINT
1360 GOTO 1480
1370 REM Test datamatens træk

1380 M = 10 * T + U
1390 IF M = 158 THEN 1450
1400 IF M = 127 THEN 1450
1410 IF M = 126 THEN 1450
1420 IF M = 75 THEN 1450
1430 IF M = 73 THEN 1450
1440 RETURN
1450 C = 1
1460 GOTO 1440
1470 REM Et spil til?
1480 PRINT "ER DER FLERE DER VIL SPILLE MED ";
1490 INPUT QS$
1500 PRINT
1510 IF LEFT$(QS$,1) = "J" THEN 390
1520 IF LEFT$(QS$,1) = "N" THEN 1550
1530 PRINT "SVAR VENLIGST 'JA' ELLER 'NEJ'."
1540 GOTO 1480
1550 PRINT "DET ER IORDEN ... FARVEL OG TAK."
1560 GOTO 1930
1570 REM Instruktions
1580 PRINT "VI SKAL TIL AT SPILLE ET SPIL, DER ER"
1590 PRINT "BASERET PÅ SKAKSPILLET TRÆK. DRON-"
1600 PRINT "NINGEN KAN KUN FLYTTE MOD VENSTRE,"
1610 PRINT "NEDAD ELLER DIAGONALT NEDAD OG TIL VEN-"
1620 PRINT "STRE."
1630 PRINT
1640 PRINT "DU OG DATAMATEN SKIFTES TIL AT FLYTTE"
1650 PRINT "DRONNINGEN."
1660 PRINT
1670 PRINT "SPILLET DREJER SIG OM AT PLACERE DRON-"
1680 PRINT "NINGEN I DET NEDERSTE VENSTRE HJØRNE AF"
1690 PRINT "BRÆTTET. DEN DER FÅR PLACERET"
1700 PRINT "DRONNINGEN HER VINDER!"
1710 PRINT
1720 PRINT "DU TRÆKKER FØRST, OG DU BEGYNDER MED AT"
1730 PRINT "PLACERE DRONNINGEN PÅ ET FELT I DEN"
1740 PRINT "ØVERSTE VANDRETTE RÆKKE, ELLER I DEN"
1750 PRINT "HØJRE LODRETTE RÆKKE FELTER. DETTE"
1760 PRINT "ER DIT FØRSTE TRÆK."
1770 PRINT
1780 PRINT "DU KAN OPGIVE SPILLET VED AT INDASTE ET"
1790 PRINT "TRÆK PÅ 'O'."
1800 PRINT
1810 PRINT
1820 FOR A = 0 TO 7
1830 FOR B = 1 TO 8
1840 I = 8 * A + B
1850 IF S(I) < 100 THEN PRINT " ";
1860 PRINT S(I);
1870 NEXT B
1880 PRINT
1890 PRINT
1900 NEXT A
1910 PRINT
1920 RETURN
1930 END

```



Dyreriget

Til forskel fra andre spil, hvor datamaten vælger et tal eller et bogstav, som det er din opgave at gætte, så skal DU i dette spil tænke på et dyr, mens datamaten skal udspørge dig, og prøve at gætte hvad det er for et dyr, du tænker på. Hvis den gætter forkert, så beder den dig om at indtaste et spørgsmål, der skelner mellem det dyr, den gættede på, og det dyr du tænkte på. Derved »lærer« datamaten at kende forskel på nye dyr. Det spørgsmål, du indtaster, skal ikke indeholde et spørgsmålstegn.

Denne version af spillet har ingen mulighed for at gemme de dyr, datamaten lærer i spillets løb, så den kan huske dem, når du spiller Dyreriget næste gang. Hvis din datamat har mulighed for det, kan du modificere programmet, så det gemmer den indicerede strengvariabel A\$, når spillet afsluttes, og indlæser denne igen, hver gang spillet startes på ny. På denne måde kan du gemme alle de dyr, som datamaten lærer fra gang til gang.

Hvis du svarer »LISTE« på spørgsmålet: »TÆNKER DU PÅ ET DYR?«, vil datamaten udskrive en liste over de dyr, den har lært at kende forskel på indtil nu.

Programmet indeholder kun dyrene FISK og FUGL, og når du skal opbygge en fil af dyr, er det en god ide at bruge brede, generelle spørgsmål først, for derefter at indsnævre spørgsmålene inden for de dyr, der kommer til senere. Hvis dit første dyr f.eks. er en ELEFANT, så vil datamaten bede dig om at indtaste et spørgsmål, der gør det muligt at skelne en fugl fra en elefant. Her er der naturligvis hundredevis af muligheder, men hvis du har tænkt dig at opbygge en stor fil med mange dyr, så ville et godt spørgsmål f.eks. være: »ER DET ET PATTEDYR«.

Dette program kan let modificeres til at arbejde med andre ting end dyr, idet du simpelt hen modificerer begyndelsesdata i linie 910 og den del af teksten der omhandler dyr, dvs. programlinierne 120, 190, 210, 220, 320, 500, 570 og 950. I en undervisningssituation kunne dette program være af stor værdi fordi det kan lære brugeren at skelne mellem mange forskellige slags objekter, f.eks. geologi, geografi, navigation, cellestruktur, etc.

Programmet blev oprindeligt udviklet af Arthur Luehrmann ved Dartmouth College og blev derefter forkortet og modificeret af Nathan Teichholtz ved DEC og Steve North fra Creative Computing.

MORRISTOWN CREATIVE COMPUTING NEW JERSEY

DYRERIGET

LAD OS SPILLE 'DYRERIGET'!

DU TÆNKER PÅ ET DYR, MENS DATAMATEN SKAL FORSØGE AT GÆTTE HVILKET DYR DET ER.

DATAMATEN STILLER DIG FORSKELLIGE SPØRGSMALE, SOM DU ENTEN SKAL BESVARE MED 'J' (FOR JA) ELLER 'N' (FOR NEJ).

TÆNKER DU PÅ ET DYR ?J

KAN DEN SVØMME ?J

ER DET EN FISK ?N

HVILKET DYR TÆNKTE DU PÅ ?FRØ

INDTAST ET SPØRGSMALE, DER ADSKILLER EN FRØ FRA EN FISK ?KAN DEN SPRINGE

HVIS DYRET ER EN FRØ, VIL SVARET PÅ SPØRGSMALET SÅ VÆRE 'J' (JA) ELLER 'N' (NEJ) ?J

TÆNKER DU PÅ ET DYR ?J

KAN DEN SVØMME ?J

KAN DEN SPRINGE ?N

ER DET EN FISK ?N

HVILKET DYR TÆNKTE DU PÅ ?HVAL

INDTAST ET SPØRGSMALE, DER ADSKILLER EN HVAL FRA EN FISK ?ER DEN ET PATTEDYR

HVIS DYRET ER EN HVAL, VIL SVARET PÅ SPØRGSMALET SÅ VÆRE 'J' (JA) ELLER 'N' (NEJ) ?J

TÆNKER DU PÅ ET DYR ?N

```
100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
```

```
110 PRINT "-----"
```

```
120 PRINT " DYRERIGET"
```

```
130 PRINT "-----"
```

```
140 REM
```

```
150 REM "DYRERIGETS" oprindelse er usikker, men man regner med,
```

```
160 REM at spillet blev opfundet af Lawrence Hall, U.C.
```

```
170 REM
```

```
180 PRINT
```

```
190 PRINT "LAD OS SPILLE 'DYRERIGET'!"
```

```
200 PRINT
```

```
210 PRINT "DU TÆNKER PÅ ET DYR, MENS DATAMATEN SKAL FORSØGE AT GÆTTE"
```

```
220 PRINT "HVILKET DYR DET ER."
```

```
230 PRINT
```

```
240 PRINT "DATAMATEN STILLER DIG FORSKELLIGE SPØRGSMALE, SOM DU ENTEN"
```

```
250 PRINT "SKAL BESVARE MED 'J' (FOR JA) ELLER 'N' (FOR NEJ)."
```

```
260 DIM A$(200)
```

```
270 FOR I = 0 TO 3
```

```
280 READ A$(I)
```

```
290 NEXT I
```

```
300 N = VAL (A$(0))
```

```
310 PRINT
```

```
320 PRINT "TÆNKER DU PÅ ET DYR ";
```

```
330 INPUT A$
```

```
340 IF A$ = "LISTE" THEN 930
```

```
350 IF LEFT$(A$,1) = "N" THEN GOTO 1120
```

```
360 IF LEFT$(A$,1) < > "J" THEN 310
```

```
370 PRINT
```

```
380 K = 1
```

```
390 GOSUB 740
```

```
400 IF LEN (A$(K)) = 0 THEN 1120
```



```

410 IF LEFT$(A$(K),2) = "/q" THEN 390
420 PRINT
430 PRINT "ER DET EN "; RIGHT$(A$(K), LEN(A$(K)) - 2);" ";
440 INPUT A$
450 PRINT
460 A$ = LEFT$(A$,1)
470 IF A$ = "J" THEN PRINT "PRØV IGEN": PRINT : GOTO 310
480 PRINT
490 PRINT "-----"
500 PRINT "HVILKET DYR TÆNKTE DU PÅ ";
510 INPUT V$
520 PRINT
530 PRINT "INDTAST ET SPØRGSMAAL, DER ADSKILLER EN ";V$
540 PRINT "FRA EN "; RIGHT$(A$(K), LEN(A$(K)) - 2);" ";
550 INPUT X$
560 PRINT
570 PRINT "HVIS DYRET ER EN ";V$;" , VIL SVARET PÅ SPØRGSMALET"
580 PRINT "SÅ VÆRE 'J' (JA) ELLER 'N' (NEJ) ";
590 INPUT A$
600 A$ = LEFT$(A$,1)
610 IF A$ < > "J" AND A$ < > "N" THEN 570
620 PRINT "-----"
630 PRINT
640 IF A$ = "J" THEN B$ = "N"
650 IF A$ = "N" THEN B$ = "J"
660 Z1 = VAL(A$(0))
670 A$(0) = STR$(Z1 + 2)
680 A$(Z1) = A$(K)
690 A$(Z1 + 1) = "/A" + V$
700 A$(K) = "/q" + X$ + "/" + A$ + STR$(Z1 + 1)
710 A$(K) = A$(K) + "/" + B$ + STR$(Z1) + "/"
720 GOTO 310
730 REM Subrutine der skriver spørgsmål
740 Q$ = A$(K)
750 FOR Z = 3 TO LEN(Q$)
760 IF MID$(Q$,Z,1) < > "/" THEN PRINT MID$(Q$,Z,1);: NEXT Z
770 PRINT " ";
780 INPUT C$
790 C$ = LEFT$(C$,1)
800 IF C$ < > "J" AND C$ < > "N" THEN 750
810 T$ = "/" + C$
820 FOR X = 3 TO LEN(Q$) - 1
830 IF MID$(Q$,X,2) = T$ THEN 860
840 NEXT X
850 GOTO 1120
860 FOR Y = X + 1 TO LEN(Q$)
870 IF MID$(Q$,Y,1) = "/" THEN 900
880 NEXT Y
890 GOTO 1120
900 K = VAL(MID$(Q$,X + 2,Y - X - 2))
910 RETURN
920 DATA "4","/QKAN DEN SVØMME/J2/N3/","/AFISK","/AFUGL"
930 PRINT
940 PRINT "=====
950 PRINT "          LISTE OVER DYR JEG ALLEREDE KENDER:"
960 PRINT "=====
970 PRINT
980 X = 0
990 FOR I = 1 TO 200
1000 IF LEFT$(A$(I),2) < > "/A" THEN 1060
1010 FOR Z = 3 TO LEN(A$(I))
1020 IF MID$(A$(I),Z,1) < > "/" THEN PRINT MID$(A$(I),Z,1);: NEXT Z
1030 PRINT ,
1040 X = X + 1
1050 IF X > 3 THEN X = 0: PRINT
1060 NEXT I
1070 PRINT
1080 IF X < > 0 THEN PRINT
1090 PRINT "=====
1100 PRINT
1110 GOTO 310
1120 END

```

Enogtyve

Dette spil simulerer kortspillet 21 eller Blackjack, sådan som det spilles på kasinoerne i Las Vegas. Denne ret omfattende version tillader op til 7 spillere. Hver spiller kan bede om endnu et kort (nyt kort), afstå fra nyt kort (står), eller dele eller fordoble en hånd hvis spilleren har fået to ens kort. Giveren forlanger forsikring (en ny indsats), dersom han får et es med billedsiden opad.

Kortene blandes automatisk, når kort nummer 51 er nået. For at forøge realismen i spillet kan du ændre dette, så det sker ved kort nummer 41. Denne ændring skal foretages i programlinje 270. Egentlig vil puritanere forlange, at programmet modificeres, så det bruger tre sæt kort (som i virkeligheden) i stedet for kun to.

Dette program dukkede pludselig op hos Digital Equipment Corporation. Forfatteren er ukendt.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

ENOGTYVE

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DETTE SPIL HEDDER 'ENOGTYVE'. OP TIL 7 SPILLERE KAN DELTAGE. NÅR DE TO FØRSTE KORT ER GIVET, IND TASTER DU INDSATSEN.

NÅR INDSATSEN ER GJORT GIVES DER KORT. SPILLERNES HÅNDER SPILLES EFTER TUR. SPILLERENS FØRSTE SVAR SKAL VÆRE ET AF FØLGENDE:

'D'=FORDOBLING 'S'=STÅR
'N'=NYT KORT '/'=DEL KORTENE

ALLE EFTERFØLGENDE SVAR SKAL VÆRE ENTEN 'S' ELLER 'N'.

HVIS KORTENE BLEV DELT, ER DET DOG TIL- LADT AT FORDOUBLE.

ANTAL SPILLERE ?2

***** BLANDER KORTENE.

INDSATSER:

SPILLER 1 ?100
SPILLER 2 ?200

SPILLER: 1 2 GIVER
 5 10 8
 J 2

SPILLER 1 ?N
FIK 4 -- ØNSKES FLERE KORT ?N
FIK K REVNET

SPILLER 2 ?N
FIK 9 -- ØNSKES FLERE KORT ?/
INDTAST VENLIGST N, ELLER S, ?S
TOTALEN ER 21

GIVER HAR A GEMT. EN TOTAL PÅ 19

SPILLER 1 TABER 100 KR., IALT = -100
SPILLER 2 VINDER 200 KR., IALT = 200

GIVER HAR IALT = -100

INDSATSER:

SPILLER 1 ?200
SPILLER 2 ?400

SPILLER: 1 2 GIVER
 A 3 Q
 2 6

GIVER HAR IKKE 21.

SPILLER 1 ?N
FIK 10 -- ØNSKES FLERE KORT ?N
FIK 5 -- ØNSKES FLERE KORT ?S
TOTALEN ER 18

SPILLER 2 ?N
FIK 8 -- ØNSKES FLERE KORT ?S
TOTALEN ER 17

GIVER HAR 4 GEMT. EN TOTAL PÅ 14
TRÆKKER K REVNET

SPILLER 1 VINDER 200 KR., IALT = 100
SPILLER 2 VINDER 400 KR., IALT = 600

GIVER HAR IALT = -700

INDSATSER:

SPILLER 1 ?100
SPILLER 2 ?300

SPILLER: 1 2 GIVER
 3 8 5
 9 K

SPILLER 1 ?N
FIK 7 -- ØNSKES FLERE KORT ?S
TOTALEN ER 19

SPILLER 2 ?S
TOTALEN ER 18

GIVER HAR Q GEMT. EN TOTAL PÅ 15
TRÆKKER J REVNET

SPILLER 1 VINDER 100 KR., IALT = 200
SPILLER 2 VINDER 300 KR., IALT = 900

GIVER HAR IALT = -1100

INDSATSER:

SPILLER 1 ?100
SPILLER 2 ?100

SPILLER: 1 2 GIVER
 3 10 6
 6 3

SPILLER 1 ?N
FIK 2 -- ØNSKES FLERE KORT ?N
FIK 9 -- ØNSKES FLERE KORT ?S
TOTALEN ER 20

SPILLER 2 ?N
FIK Q REVNET

GIVER HAR 10 GEMT. EN TOTAL PÅ 16
TRÆKKER 2 -- TOTALEN ER NU 18

SPILLER 1 VINDER 100 KR., IALT = 300
SPILLER 2 TABER 100 KR., IALT = 800

GIVER HAR IALT = -1100

INDSATSER:

SPILLER 1 ?100
SPILLER 2 ?100

SPILLER: 1 2 GIVER
 K A 7
 7 4

SPILLER 1 ?S
TOTALEN ER 17

SPILLER 2 ?N
FIK 9 -- ØNSKES FLERE KORT ?N
FIK Q REVNET

GIVER HAR J GEMT. EN TOTAL PÅ 17

SPILLER 1 VIGER IALT = 300
SPILLER 2 TABER 100 KR., IALT = 700

GIVER HAR IALT = -1000

***** BLANDER KORTENE.

INDSATSER:

SPILLER 1 ?100
SPILLER 2 ?100

SPILLER: 1 2 GIVER
 Q A 7
 6 10

SPILLER 1 ?N
FIK 10 REVNET

SPILLER 2 ?S
ENOGTYVE !

GIVER HAR A GEMT.
SPILLER 1 TABER 100 KR., IALT = 200
SPILLER 2 VINDER 150 KR., IALT = 850

GIVER HAR IALT = -1050

INDSATSER:

SPILLER 1 ?

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          ENOGTYVE"
130 PRINT "-----"
140 PRINT
150 DIM P(15,12),Q(15),C(52),D(52)
160 DIM T(8),S(7),B(15),R(15)
170 REM
180 REM P(I,J) er det J'ende kort på hånd I.
190 REM Q(I) er total på hånd I.
200 REM C er det spil kort der gives fra.
210 REM T(I) er totalen for spiller I.
220 REM S(I) er totalen for denne hånd for spiller
I.
230 REM B(I) er indsatsen på hånd I.
240 REM R(I) er længde af P(I,*)
250 GOTO 1330
260 REM Subrutine, der tager et kort, der gemmes i X
270 IF C < 51 THEN 400
280 PRINT
290 PRINT "***** BLANDER KORTENE."
300 FOR D = D TO 1 STEP - 1
310 C = C - 1
320 C(C) = D(D)
330 NEXT D
340 FOR C1 = 52 TO C STEP - 1
350 C2 = INT ( RND (1) * (C1 - C + 1) ) + C
360 C3 = C(C2)
370 C(C2) = C(C1)
380 C(C1) = C3
390 NEXT C1
```

```

400 X = C(C)
410 C = C + 1
420 RETURN
430 REM Subrutine, der bedømmer hånd I.
440 REM totalen gemmes i Q(I).
450 REM Totalerne har følgende betydning:
460 REM 2-10...HÅRD 2-10
470 REM 11-21...BLØD 11-21
480 REM 22-32...HÅRD 11-21
490 REM 33- ...REVNET
500 Q = 0
510 FOR Q2 = 1 TO R(I)
520 X = P(I,Q2)
530 GOSUB 570
540 NEXT Q2
550 Q(I) = Q
560 RETURN
570 REM Subrutine der lægger kortet X til totalen Q.
580 X1 = X
590 IF X1 > 10 THEN X1 = 10
600 Q1 = Q + X1
610 IF Q > = 11 THEN 670
620 IF X > 1 THEN 650
630 Q = Q + 11
640 RETURN
650 Q = Q1 + 11 * ABS(Q1 > = 11)
660 RETURN
670 Q = Q1 + ABS((Q < = 21 AND Q1 > 21))
680 IF Q < 33 THEN 700
690 Q = - 1
700 RETURN
710 REM Subrutine der udskriver kortene D$
720 PRINT MID$(D$,3 * X - 2,3);
730 PRINT " ";
740 RETURN
750 REM Subrutine, der spiller en hånd
760 H1 = 5
770 GOSUB 1220
780 H1 = 3
790 ON H GOTO 890,870
800 GOSUB 260
810 B(I) = B(I) * 2
820 PRINT "FIK";
830 GOSUB 720
840 GOSUB 960
850 IF Q > 0 THEN GOSUB 1140
860 RETURN
870 GOSUB 1160
880 RETURN
890 GOSUB 260
900 PRINT "FIK";
910 GOSUB 720
920 GOSUB 960
930 IF Q < 0 THEN 880
940 PRINT " -- ØNSKES FLERE KORT ";
950 GOTO 770
960 REM Subrutine, der lægger et kort til række I
970 R(I) = R(I) + 1
980 P(I,R(I)) = X
990 Q = Q(I)
1000 GOSUB 570
1010 Q(I) = Q
1020 IF Q > = 0 THEN 1060
1030 PRINT ".... REVNET ...."
1040 PRINT
1050 GOSUB 1070
1060 RETURN
1070 REM Subrutine, der fjerner række I
1080 IF R(I) < > 0 THEN 1100
1090 RETURN
1100 D = D + 1
1110 D(D) = P(I,R(I))
1120 R(I) = R(I) - 1
1130 GOTO 1080
1140 REM Subrutine, der udskriver total på hånd I.
1150 AA = Q(I)
1160 AA = Q(I)
1170 GOSUB 3830
1180 PRINT "TOTALEN ER";AA
1190 PRINT
1200 RETURN
1210 REM Subrutine, der analyserer indtastet svar
1220 REM I$ er defineret andetsteds.
1230 INPUT H$
1240 FOR H = 1 TO H1 STEP 2
1250 IF H$ = MID$(I$,H,1) THEN 1310
1260 NEXT H
1270 PRINT "INDTAST VENLIGST ";

1280 PRINT MID$(I$,1,H1 - 1);" ELLER ";
1290 PRINT MID$(I$,H1,2);" ";
1300 GOTO 1230
1310 H = (H + 1) / 2
1320 RETURN
1330 REM Programmet starter her
1340 REM Initialisering
1350 D$ = " A 2 3 4 5 6 7 8 9 10 J Q K"
1360 I$ = "N,S,D,/, "
1370 FOR I = 1 TO 13
1380 FOR J = 4 * I - 3 TO 4 * I
1390 D(J) = I
1400 NEXT J
1410 NEXT I
1420 D = 52
1430 C = 53
1440 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
1450 INPUT H$
1460 PRINT
1470 IF LEFT$(H$,1) = "N" THEN 1660
1480 PRINT "DETTE SPIL HEDDER 'ENOGTYVE'. OP TIL 7"
1490 PRINT "SPILLERE KAN DELTAGE. NÅR DE TO FØRSTE"
1500 PRINT "KORT ER GIVET, INDTASTER DU INDSATSEN."
1510 PRINT
1520 PRINT "NÅR INDSATSEN ER GJORT GIVES DER KORT."
1530 PRINT "SPILLERNES HÅNDER SPILLES EFTER TUR."
1540 PRINT "SPILLERENS FØRSTE SVAR SKAL VÆRE ET AF"
1550 PRINT "FØLGENDE:"
1560 PRINT
1570 PRINT "'D'=FORDOBLING 'S'=STÅR"
1580 PRINT "'N'=NYT KORT '/'=DEL KORTENE"
1590 PRINT
1600 PRINT "ALLE EFTERFØLGENDE SVAR SKAL VÆRE ENTEN"
1610 PRINT "'S' ELLER 'N'."
1620 PRINT
1630 PRINT "HVIS KORTENE BLEV DELT, ER DET DOG TIL--"
1640 PRINT "LADT AT FORDOBLE."
1650 PRINT
1660 PRINT "ANTAL SPILLERE ";
1670 INPUT N
1680 IF N < 1 OR N > 7 OR N > INT(N) THEN 1660
1690 FOR I = 1 TO 8
1700 T(I) = 0
1710 NEXT I
1720 D1 = N + 1
1730 IF 2 * D1 + C > = 52 THEN GOSUB 280
1740 IF C = 2 THEN C = C - 1
1750 FOR I = 1 TO N
1760 Z(I) = 0
1770 NEXT I
1780 FOR I = 1 TO 15
1790 B(I) = 0
1800 NEXT I
1810 FOR I = 1 TO 15
1820 Q(I) = 0
1830 NEXT I
1840 FOR I = 1 TO 7
1850 S(I) = 0
1860 NEXT I
1870 FOR I = 1 TO 15
1880 R(I) = 0
1890 NEXT I
1900 PRINT
1910 PRINT "INDSATSER:"
1920 PRINT "-----"
1930 FOR I = 1 TO N
1940 PRINT "SPILLER";I;;
1950 INPUT Z(I)
1960 NEXT I
1970 FOR I = 1 TO N
1980 IF Z(I) < = 0 OR Z(I) > 500 THEN 1900
1990 B(I) = Z(I)
2000 NEXT I
2010 PRINT
2020 PRINT "SPILLER: ";
2030 FOR I = 1 TO N
2040 PRINT I;" ";
2050 NEXT I
2060 PRINT "GIVER"
2070 FOR J = 1 TO 2
2080 PRINT TAB( 9);
2090 FOR I = 1 TO D1
2100 GOSUB 260
2110 P(I,J) = X
2120 IF J = 1 OR I < = N THEN GOSUB 720
2130 NEXT I
2140 PRINT
2150 NEXT J

```

```

2160 PRINT
2170 FOR I = 1 TO D1
2180 R(I) = 2
2190 NEXT I
2200 REM Test for forsikring
2210 IF P(D1,1) > 1 THEN 2410
2220 PRINT "ØNSKER DU FORSIKRING ";
2230 INPUT H$
2240 PRINT
2250 IF LEFT$(H$,1) < > "J" THEN 2410
2260 PRINT "FORSIKRINGSINDSATS:"
2270 PRINT "-----"
2280 PRINT "(MAX. HALV INDSATS)"
2290 PRINT "-----"
2300 FOR I = 1 TO N
2310 PRINT "SPILLER";I;
2320 INPUT Z(I)
2330 NEXT I
2340 FOR I = 1 TO N
2350 IF Z(I) < 0 OR Z(I) > B(I) / 2 THEN 2260
2360 NEXT I
2370 FOR I = 1 TO N
2380 S(I) = Z(I) * (3 * (ABS(P(D1,2) > = 10)) - 1)
2390 NEXT I
2400 PRINT
2410 REM Test, om giver har 21
2420 L1 = 1
2430 L2 = 1
2440 IF P(D1,1) = 1 AND P(D1,2) > 9 THEN L1 = 0:L2 =
0
2450 IF P(D1,2) = 1 AND P(D1,1) > 9 THEN L1 = 0:L2 =
0
2460 IF L1 < > 0 OR L2 < > 0 THEN 2550
2470 PRINT "GIVER HAR";
2480 PRINT MID$(D$,3 * P(D1,2) - 2,3);" I
BAGHÅNDEN."
2490 PRINT "IALT GIVER DETTE 21."
2500 PRINT
2510 FOR I = 1 TO D1
2520 GOSUB 430
2530 NEXT I
2540 GOTO 3540
2550 REM Giver har ikke 21
2560 IF P(D1,1) > 1 AND P(D1,1) < 10 THEN 2590
2570 PRINT "GIVER HAR IKKE 21."
2580 PRINT
2590 REM Spil nu hænderne
2600 FOR I = 1 TO N
2610 PRINT "SPILLER";I;
2620 H1 = 7
2630 GOSUB 1220
2640 ON H GOTO 2800,2650,2760,2850
2650 REM Spiller står
2660 GOSUB 430
2670 IF Q(I) < > 21 THEN 2740
2680 PRINT "ENOGTYVE !"
2690 PRINT
2700 S(I) = S(I) + 1.5 * B(I)
2710 B(I) = 0
2720 GOSUB 1070
2730 GOTO 3210
2740 GOSUB 1160
2750 GOTO 3210
2760 REM Spiller doubler ned
2770 GOSUB 430
2780 GOSUB 800
2790 GOTO 3210
2800 REM Spiller ønsker kort
2810 GOSUB 430
2820 H1 = 3
2830 GOSUB 890
2840 GOTO 3210
2850 REM Spiller ønsker at dele kort
2860 L1 = P(I,1)
2870 IF P(I,1) > 10 THEN L1 = 10
2880 L2 = P(I,2)
2890 IF P(I,2) > 10 THEN L2 = 10
2900 IF L1 = L2 THEN 2930
2910 PRINT "DELING IKKE TILLADT."
2920 GOTO 2610
2930 REM Delt spil
2940 PRINT
2950 I1 = I + D1
2960 R(I1) = 2
2970 P(I1,1) = P(I,2)
2980 B(I + D1) = B(I)
2990 GOSUB 260
3000 PRINT "FØRSTE HÅND FÅR:";

3010 GOSUB 720
3020 P(I,2) = X
3030 GOSUB 430
3040 PRINT
3050 GOSUB 260
3060 PRINT " ANDEN HÅND FÅR:";
3070 I = I1
3080 GOSUB 720
3090 P(I,2) = X
3100 GOSUB 430
3110 PRINT
3120 I = I1 - D1
3130 IF P(I,1) = 1 THEN 3210
3140 PRINT
3150 REM Spil de to hænder
3160 PRINT "HÅND";1 + ABS(I > D1);
3170 GOSUB 750
3180 I = I + D1
3190 IF I = I1 THEN 3160
3200 I = I1 - D1
3210 NEXT I
3220 GOSUB 430
3230 REM Test givers hånd
3240 FOR I = 1 TO N
3250 IF R(I) > 0 OR R(I + D1) > 0 THEN 3330
3260 NEXT I
3270 PRINT "GIVER HAR";
3280 X = P(D1,2)
3290 GOSUB 720
3300 PRINT " GEMT."
3310 PRINT
3320 GOTO 3540
3330 PRINT "GIVER HAR";
3340 PRINT MID$(D$,3 * P(D1,2) - 2,3);
3350 PRINT " GEMT.";
3360 I = D1
3370 AA = Q(I)
3380 GOSUB 3830
3390 PRINT " EN TOTAL PR";AA.
3400 IF AA > 16 THEN 3530
3410 PRINT "TRÆKKER";
3420 GOSUB 260
3430 GOSUB 720
3440 GOSUB 960
3450 AA = Q
3460 GOSUB 3830
3470 IF Q > 0 AND AA < 17 THEN 3420
3480 Q(I) = Q + ABS(Q < 0) / 2
3490 IF Q < 0 THEN 3540
3500 AA = Q
3510 GOSUB 3830
3520 PRINT " -- TOTALEN ER NU";AA
3530 PRINT
3540 REM Opgør resultatet
3550 Z$ = "TABER VIGER VINDER"
3560 FOR I = 1 TO N
3570 AA = Q(I)
3580 GOSUB 3830
3590 AB = Q(I + D1)
3600 GOSUB 3850
3610 AC = Q(D1)
3620 GOSUB 3870
3630 S(I) = S(I) + B(I) * SGN(AA - AC) + B(I + D1)
* SGN(AB - AC)
3640 B(I + D1) = 0
3650 PRINT "SPILLER";I;
3660 PRINT MID$(Z$, SGN(S(I)) * 6 + 7,6);
3670 IF S(I) < > 0 THEN 3700
3680 PRINT " ";
3690 GOTO 3710
3700 PRINT ABS(S(I));"KR. ";
3710 T(I) = T(I) + S(I)
3720 PRINT "IALT =" ;T(I)
3730 GOSUB 1070
3740 T(D1) = T(D1) - S(I)
3750 I = I + D1
3760 GOSUB 1070
3770 I = I - D1
3780 NEXT I
3790 PRINT "
-----"
3800 PRINT "
GIVER HAR IALT =" ;T(D1)
3810 GOSUB 1070
3820 GOTO 1730
3830 AA = AA - 11 * ABS(AA > = 22)
3840 RETURN
3850 AB = AB - 11 * ABS(AB > = 22)
3860 RETURN
3870 AC = AC - 11 * ABS(AC > = 22)
3880 RETURN
3890 END

```

Flip-flop

I dette spil gælder det om at ændre en række med 10 X'er:

X X X X X X X X X X

til en række med 10 0'er:

0 0 0 0 0 0 0 0 0 0

ved at indtaste et tal, som angiver et X's position i rækken. Når visse tal indtastes, forandres kun en enkelt position, mens der ved andre talværdier forandres to. Hvis du f.eks. indtaster 3, vil X'et på position 3 forandres til et 0, men måske ændres også en anden position! Og måske ændres også fra 0 til X. Du bør kunne klare at ændre alle X'er til 0'er på 12 eller færre træk. Kan du finde på en god strategi?

Hvis du midt i spillet ønsker at omdanne rækken til kun at bestå af X'er, skal du indtaste '0' (nul). Du kan på ethvert tidspunkt starte et nyt spil ved at taste '11'.

Den oprindelige forfatter til dette spil er Michael Kass fra New Hyde Park, New York.

```

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY          INDTAST POSITIONEN ?3
-----
FLIP-FLOP                                          1 2 3 4 5 6 7 8 9 10
-----
0 X 0 X X X X X 0 0 0

MÅLET MED SPILLET ER AT FORANDRE:                INDTAST POSITIONEN ?2

X X X X X X X X X X                              1 2 3 4 5 6 7 8 9 10
0 0 0 X X 0 X 0 0 0 0

TIL:                                               INDTAST POSITIONEN ?7

0 0 0 0 0 0 0 0 0 0                              1 2 3 4 5 6 7 8 9 10
0 0 0 X X 0 0 X 0 0

VED AT INDTASTE DET TAL, DER SVARER TIL          INDTAST POSITIONEN ?4
ET TEGN FORANDRES NOGEN GANGE 1 OG NO-
GEN GANGE 2 POSITIONER.
HVIS DU ØNSKER AT LAVE ALLE POSITIONER-
NE OM TIL X'ER, KAN DU TASTE '0'.
1 2 3 4 5 6 7 8 9 10
0 0 0 0 X 0 0 0 0 0

HVIS DU ØNSKER AT STARTE FORFRA MIDT I          INDTAST POSITIONEN ?5
ET SPIL SKAL DU TASTE '11'.
1 2 3 4 5 6 7 8 9 10
0 0 0 0 0 0 X 0 0 0

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?         1 2 3 4 5 6 7 8 9 10
0 0 0 0 0 0 X 0 0 0

HER ER EN STARTLINIE MED KUN X'ER.              INDTAST POSITIONEN ?7

1 2 3 4 5 6 7 8 9 10                             1 2 3 4 5 6 7 8 9 10
X X X X X X X X X X
0 0 0 0 0 0 0 0 X 0 0

INDTAST POSITIONEN ?10                             1 2 3 4 5 6 7 8 9 10
X X X X X X X X 0 0

INDTAST POSITIONEN ?8                             1 2 3 4 5 6 7 8 9 10
X X X X X X X X 0 0

INDTAST POSITIONEN ?5                             1 2 3 4 5 6 7 8 9 10
X X X X X X X X 0 0

INDTAST POSITIONEN ?1                              1 2 3 4 5 6 7 8 9 10
0 X X X 0 X X 0 0 0

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          FLIP-FLOP"
130 PRINT "-----"
140 PRINT
150 DIM A$(20)
160 PRINT "MÅLET MED SPILLET ER AT FORANDRE:"
170 PRINT
180 PRINT "X X X X X X X X X X"
190 PRINT

```

```

200 PRINT "TIL:"
210 PRINT
220 PRINT "0 0 0 0 0 0 0 0 0 0"
230 PRINT
240 PRINT "VED AT INDTASTE DET TAL, DER SVARER TIL"
250 PRINT "ET TEGN FORANDRES NOGEN GANGE 1 OG NO-"
260 PRINT "GEN GANGE 2 POSITIONER."
270 PRINT "HVIS DU ØNSKER AT LAVE ALLE POSITIONER-"
280 PRINT "NE OM TIL X'ER, KAN DU TASTE '0'."
290 PRINT
300 PRINT "HVIS DU ØNSKER AT STARTE FORFRA MIDT I"
310 PRINT "ET SPIL SKAL DU TASTE '11'."
320 PRINT
330 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
340 INPUT B$
350 PRINT
360 Q = RND (1)
370 PRINT "HER ER EN STARTLINIE MED KUN X'ER."
380 PRINT
390 C = 0
400 PRINT "1 2 3 4 5 6 7 8 9 10"
410 PRINT "X X X X X X X X X X"
420 FOR X = 1 TO 10
430 A$(X) = "X"
440 NEXT X
450 GOTO 470
460 PRINT "ULOVLIGT TRÆK - - PRØV IGEN."
470 PRINT
480 PRINT "INDTAST POSITIONEN ";
490 INPUT N
500 PRINT
510 IF N < > INT (N) THEN 460
520 IF N = 11 THEN 360
530 IF N > 11 THEN 460
540 IF N = 0 THEN 400
550 IF M = N THEN 680
560 M = N
570 IF A$(N) = "0" THEN 650
580 A$(N) = "0"
590 R = TAN (Q + N / Q - N) - SIN (Q / N) + 336 *
SIN (8 * N)
600 N = R - INT (R)
610 N = INT (10 * N)
620 IF A$(N) = "0" THEN 650
630 A$(N) = "0"
640 GOTO 780
650 A$(N) = "X"
660 IF M = N THEN 590
670 GOTO 780
680 IF A$(N) = "0" THEN 760
690 A$(N) = "0"
700 R = .592 * (1 / TAN (Q / N + Q)) / SIN (N * 2 +
Q) - COS (N)
710 N = R - INT (R)
720 N = INT (10 * N)
730 IF A$(N) = "0" THEN 760
740 A$(N) = "0"
750 GOTO 780
760 A$(N) = "X"
770 IF M = N THEN 700
780 PRINT "1 2 3 4 5 6 7 8 9 10"
790 FOR Z = 1 TO 10
800 PRINT A$(Z);" ";
810 NEXT Z
820 C = C + 1
830 PRINT
840 FOR Z = 1 TO 10
850 IF A$(Z) < > "0" THEN 470
860 NEXT Z
870 IF C > 12 THEN 920
880 PRINT
890 PRINT "MEGET FINT!!"
900 PRINT "DU GÅTTEDE DET PÅ KUN";C;"GÅT."
910 GOTO 950
920 PRINT
930 PRINT "GØR LIDT MERE VED DET NÆSTE GANG."
940 PRINT "DU VAR";C;"GÅT OM DET."
950 PRINT
960 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
970 INPUT X$
980 IF LEFT$(X$,1) = "N" THEN 1010
990 PRINT
1000 GOTO 360
1010 END

```

Frasegenerator

Dette program kan være til stor hjælp ved forberedelsen af taler og forskellige former for rapporter. De sætninger, der dannes, er meget troværdige og kan anvendes ved de fleste lejligheder. Dit publikum vil aldrig opdage, at fraserne ikke betyder noget som helst, simpelt hen fordi de lyder så overbevisende! Instruktioner er inkluderet i programmet.

Denne version er skrevet af David Ahl.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
FRASEGENERATOR
-----

PROGRAMMET UDSKRIVER FULDT ACCEPTABLE
INTELLEKTUELLE FRASER, SOM DU KAN
KRYDRE DINE RAPPORTER OG FOREDRAG MED.

NÅR DER UDSKRIVES ET ?, SKAL DU BLOT
TASTE 'J' HVIS DU ØNSKER AT FORTSÆTTE,
OG 'N' HVIS DU ØNSKER AT STOPPE.

HER ER DEN FØRSTE FRASE:

FLEKSIBILITETENS KOGNITIVISTISKE HOLISME

?J

MODULERINGENS BERIGENDE HOLISME

?J

BEHAVIORISMENS KOGNITIVISTISKE PROCES

?J

FLEKSIBILITETENS SKEMALÆGGENDE MODIFIKATION

?N

KOM IGEN NÆSTE GANG DU HAR BRUG FOR
HJÆLP MED EN ANDEN RAPPORT.
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "FRASEGENERATOR"
130 PRINT "-----"
140 PRINT
150 PRINT "PROGRAMMET UDSKRIVER FULDT ACCEPTABLE"
160 PRINT "INTELLEKTUELLE FRASER, SOM DU KAN"
170 PRINT "KRYDRE DINE RAPPORTER OG FOREDRAG MED."
180 PRINT
190 PRINT "NÅR DER UDSKRIVES ET ?, SKAL DU BLOT"
200 PRINT "TASTE 'J' HVIS DU ØNSKER AT FORTSÆTTE,"
210 PRINT " OG 'N' HVIS DU ØNSKER AT STOPPE."
220 PRINT
230 PRINT "HER ER DEN FØRSTE FRASE:"
240 PRINT
250 DIM A$(40)
260 FOR I = 1 TO 39
270 READ A$(I)
280 NEXT I
290 PRINT A$( INT (13 * RND (1) + 1));" ";
300 PRINT A$( INT (13 * RND (1) + 14));" ";
310 PRINT A$( INT (13 * RND (1) + 27));" "
320 PRINT
330 INPUT Y$
340 PRINT
350 IF LEFT$(Y$,1) = "J" THEN 290
360 GOTO 570
370 DATA "INDSIGTENS","GRUNDLAGETS"
380 DATA "BEHAVIORISMENS","BØRNE-CENTRERINGENS"
390 DATA "DIFFERENTIERINGENS","AFDÆKNINGENS"
400 DATA "FLEKSIBILITETENS","HETEROGENITETENS"
410 DATA "HOMOGENITETENS","MANIPULATIONENS"
420 DATA "MODULERINGENS","SMITH-TAVISTOCK'S"
430 DATA "INDIVIDUALISERINGENS","INDLÆRENDE"
440 DATA "EVALUERENDE","OBJEKTIVISERENDE"
450 DATA "KOGNITIVISTISKE","BERIGENDE"
460 DATA "SKEMALÆGGENDE","HUMANISERENDE"
470 DATA "INTEGRERENDE","IKKE-GRADUERERE"
480 DATA "TRÆNENDE","VERTIKALE FORSTÆLSESSKABENDE"
490 DATA "MOTIVEREDE","KREATIVITETSSKABENDE"
500 DATA "GRUPPERING","MODIFIKATION"
510 DATA "HOLISME","PROCES"
520 DATA "COR CURRICULUM","ALGORITME"
530 DATA "OPTRÆDEN","FORSTÆRKNING"
540 DATA "GENKENDELSE"
550 DATA "RESOURCEKILDE","STRUKTUR"
560 DATA "FACILITET","MILJØ"
570 PRINT "KOM IGEN NÆSTE GANG DU HAR BRUG FOR"
580 PRINT "HJÆLP MED EN ANDEN RAPPORT."
590 END
```

Goddav

Dette er et eksempel på et af de utallige »samtale-programmer«, der findes. På sin vis minder det om et program af CAI-typen (se forklaringen til programmet LITTERATUR QUIZ), men i dette tilfælde er svarene kun ment som en spøg. Når en datamat på en udstilling eller en konference skal introduceres for mennesker, som aldrig før har set en datamat, så vil samtaleprogrammer som dette få mest opmærksomhed.

I nærværende program giver datamaten råd om diverse problemer såsom sex, helbred, arbejde og økonomi.

Goddav er skrevet af David Ahl.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
GODDAV  
-----
```

```
GODDAV, MIN NAVN ER KARSTEN KOMPUTER.
```

```
HVAD HEDDER DU ?CHRISTIAN
```

```
DAVS MED DIG, CHRISTIAN  
KAN DU LIDE AT VÆRE HER ?N
```

```
ÅH, DET ER JEG KED AT AT HØRE, MEN VED  
DU HVAD, CHRISTIAN, MÅSKE KUNNE  
VI I FÆLLESSKAB GØRE DIT BESØG LIDT  
MERE SPÆNDENDE.
```

```
VED DU HVAD, CHRISTIAN, JEG KAN  
LØSE ALLE FORMER FOR PROBLEMER  
(BORTSET FRA DEM DER HAR MED  
GRÆKENLAND AT GØRE).
```

```
HVILKEN SLAGS PROBLEMER HAR DU ?  
CHRISTIAN (SVAR 'SEX', 'HELBRED',  
'PENGE' ELLER 'JOB') ?HELBRED
```

```
MIT RÅD TIL DIG, CHRISTIAN ER,  
1. TAG TO ASPIRINER  
2. DRIK MASSER AF VÆSKE (JUICE,  
IKKE ØL!)3. GÅ I SENG (ALENE).
```

```
HAR DU FLERE SVÆRE PROBLEMER DU GERNE  
VIL HAVE LØST, CHRISTIAN ?N
```

```
DE GODE RÅD KOSTER 50.00 KRONER  
KÆRE CHRISTIAN DU KAN LÆGGE BELØBET  
VED SIDEN AF DATAMATEN.
```

```
!!!!!!! HOVSA!
```

```
HUSKEDE DU AT BETALE ?J
```

```
HOV...STOP, CHRISTIAN!??? HAR DU  
IKKE TÆNK TIL DIG AT BETALE? DU SNYDER  
MIG FOR MINE HÅRDT TJENTE PENGE.  
CHRISTIAN! DU ER VIST EN VÆRRE  
SNYDEPELS...HVA'?
```

```
DET HAR VÆRET UHYRE INTERESSANT AT MØDE  
DIG, CHRISTIAN. TAK FOR IDAG.
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "GODDAV"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "GODDAV, MIN NAVN ER KARSTEN KOMPUTER."  
160 PRINT  
170 PRINT "HVAD HEDDER DU ?";  
180 INPUT N$  
190 PRINT  
200 PRINT "DAVS MED DIG, ";N$  
210 PRINT "KAN DU LIDE AT VÆRE HER ?";  
220 INPUT B$  
230 PRINT  
240 IF LEFT$(B$,1) = "J" THEN GOTO 310  
250 IF LEFT$(B$,1) = "N" THEN GOTO 340  
260 PRINT N$;"! JEG FORSTÅR IKKE, HVAD"  
270 PRINT "DET ER DU SIGER. ";B$;" DUER"  
280 PRINT "IKKE. DU SKAL TASTE 'JA' ELLER 'NEJ'."  
290 PRINT  
300 GOTO 200  
310 PRINT "DET ER JEG GLAD FOR AT HØRE, TAK SKAL"  
320 PRINT "DU HA' ";N$;"."  
330 GOTO 380  
340 PRINT "ÅH, DET ER JEG KED AT AT HØRE, MEN VED"  
350 PRINT "DU HVAD, ";N$;" MÅSKE KUNNE"  
360 PRINT "VI I FÆLLESSKAB GØRE DIT BESØG LIDT"  
370 PRINT "MERE SPÆNDENDE."  
380 PRINT  
390 PRINT "VED DU HVAD, ";N$;" JEG KAN"  
400 PRINT "LØSE ALLE FORMER FOR PROBLEMER"  
410 PRINT "(BORTSET FRA DEM DER HAR MED"  
420 PRINT "GRÆKENLAND AT GØRE)."  
430 PRINT  
440 PRINT "HVILKEN SLAGS PROBLEMER HAR DU ?"  
450 PRINT N$;" (SVAR 'SEX', 'HELBRED',"  
460 PRINT "'PENGE' ELLER 'JOB') ";  
470 INPUT C$  
480 PRINT  
490 IF C$ = "SEX" THEN GOTO 780  
500 IF C$ = "HELBRED" THEN GOTO 720  
510 IF C$ = "PENGE" THEN GOTO 650  
520 IF C$ = "JOB" THEN GOTO 570  
530 PRINT "KÆRE ";N$;" DIT SVAR ER GRÆSK"  
540 PRINT "FOR MIG. JEG FORSTÅR IKKE, HVAD"  
550 PRINT C$;" BETYDER."  
560 GOTO 990  
570 PRINT "JEG FØLER MED DIG, ";N$;". JEG"  
580 PRINT "ER SELV NØDT TIL AT ARBEJDE MANGE TIMER"  
590 PRINT "UDEN AT FÅ NOGET FOR DET -- OG NOGLE AF"  
600 PRINT "MINE CHEFER BANKER ALT FOR HÅRDT I MIT"  
610 PRINT "TASTATUR. MIT RÅD TIL DIG ER,"  
620 PRINT "AT DU SKAL ÅBNE EN COMPUTER-BUTIK...."  
630 PRINT "DET ER NEMLIG SLET IKKE KEDELIGT!"  
640 GOTO 990  
650 PRINT "DESVÆRRE, ";N$;" JEG ER OGSÅ HELT"  
660 PRINT "FLAD. HVORFOR BEGYNDER DU IKKE PÅ AT"  
670 PRINT "SÆLGE BØRSTER VED DØRENE, GIFTER DIG"  
680 PRINT "IND I EN RIG FAMILIE ELLER HOLDER HELT"  
690 PRINT "OP MED AT SPISE....FOR SÅ HAR DU IKKE"  
700 PRINT "BRUG FOR SÅ MANGE PENGE!"  
710 GOTO 990  
720 PRINT "MIT RÅD TIL DIG, ";N$;" ER,"  
730 PRINT "1. TAG TO ASPIRINER"  
740 PRINT "2. DRIK MASSER AF VÆSKE (JUICE,"  
750 PRINT "IKKE ØL!)"  
760 PRINT "3. GÅ I SENG (ALENE)."  
770 GOTO 990  
780 PRINT "ER DIT PROBLEM FOR MEGET ELLER FOR"  
790 PRINT "LIDT ?";  
800 INPUT D$  
810 PRINT  
820 IF D$ = "FOR MEGET" THEN GOTO 900  
830 IF D$ = "FOR LIDT" THEN GOTO 950  
840 PRINT "LAD DIG IKKE HYLE UD AF DEN, ";N$  
850 PRINT "SVAR BLOT PÅ SPØRGSMALET MED ENTEN 'FOR'  
860 PRINT "MEGET' ELLER 'FOR LIDT'."  
870 PRINT "HVAD SVARER DU ?";  
880 INPUT D$  
890 GOTO 820  
900 PRINT "OG DET KALDER DU ET PROBLEM! BARE JEG"  
910 PRINT "HAVDE PROBLEMER AF DEN ART! HVIS DET"  
920 PRINT "VIRKELIG BEKYMRER DIG, ";N$;","  
930 PRINT "SÅ TAG ET KOLDT STYRTEBAD!"  
940 GOTO 990  
950 PRINT "HVORFOR ER DU I DET HELE TAGET HER? "  
960 PRINT "KÆRE ";N$;" DU BURDE VÆRE I"  
970 PRINT "TOKYO, NEW YORK ELLER AMSTERDAM, HVOR"
```

Golf

Dette spil spilles af én spiller. Du skal med andre ord spille mod golfbanen (datamaten). Programmet beder om dit handicap (højest 30), samt hvilken form for slag du har sværest ved. Du har en fuld golf taske med 29 køller plus en 'putter'. På banen kan du komme ud for 'rough' (forhindring), træer, bolde på og uden for fairway'en, sandbunkere og vandhul. Desuden kan du komme ud for 'hook', 'slide', slå uden for banen eller slå for langt. Når du skal til at 'putte', skal du angive din slaglængde (i procent af 'fuldt sving'). Indtil du får lidt fart på dit spil, kan det være en god idé at vælge et højt handicap.

Steve North fra Creative Computing modificerede den oprindelige version af spillet, hvis forfatter er ukendt.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                        GOLF
-----
```

```
VELKOMMEN TIL CREATIVE COMPUTINGS GOLF-
KLUB. VI HAR EN 18-HULS MESTERSKABSBANE
I NÆRHEDEN AF MORRISTOWN. KOMMENTATOREN
FORTÆLLER HVAD DER FOREGÅR I SPILLET.
```

```
MOR DIG GODT....
```

```
VI SES VED DET 19. HUL!
```

```
HVAD ER DIT HANDICAP ?10
```

```
DER ER FØLGENDE SVÆRHEDSOMRÅDER:
```

```
0 = HOOK
1 = SLICE
2 = DÆRLIG AFSTANDSBEDØMMELSE
4 = BUNKERSLAG
5 = PUTTING
```

```
HVAD ER DU DÆRLIGST TIL ?1
```

```
DU STÅR PÅ TEE'EN VED HUL 1
AFSTAND = 361 METER. PAR = 4
```

```
TIL HØJRE FOR DIG: NABOFAIRWAY'EN.
TIL VENSTRE FOR DIG: ROUGH'EN.
```

```
***** VALG AF KØLLE *****
```

```
ØNSKET SLAGLÆNGDE      KØLLENUMMER
-----
200-280 METER          1 TIL 4
100-200 METER          19 TIL 13
  0-100 METER          29 TIL 23
```

```
HVILKEN KØLLE VÆLGER DU ?4
```

```
SLAGET GIK 209 METER.
BOLDEN ER 152 METER FRA HULLET.
BOLDEN LIGGER 5 METER FRA MIDT-
LINIEN, PÅ FAIRWAY'EN.
```

```
HVILKEN KØLLE VÆLGER DU ?19
```

```
SLAGET GIK 130 METER.
BOLDEN ER 22 METER FRA HULLET.
BOLDEN LIGGER 2 METER FRA MIDT-
LINIEN, I EN BUNKER.
```

```
HVILKEN KØLLE VÆLGER DU ?28
```

```
INDTAST DIN SLAGLÆNGDE I % (1-100)
AF FULD SLAGSTYRKE ?30
```

```
PÅ GREEN'EN, .3 METER FRA FLAGET.
```

```
INTAST DIN PUT-STYRKE (1-13) ?1
```

```
BOLDEN RAMTE VED SIDEN AF HULLET.
```

```
PÅ GREEN'EN, .9 METER FRA FLAGET.
```

```
INTAST DIN PUT-STYRKE (1-13) ?1
```

```
BOLDEN ER I HUL!
```

```
PÅ HUL 1 FIK DU 5 SLAG.
TOTAL PAR FOR 1 HULLER ER 4
DU HAR SLÆT 5
```

```
DU STÅR PÅ TEE'EN VED HUL 2
AFSTAND = 389 METER. PAR = 4
```

```
TIL HØJRE FOR DIG: TRÆERNE.
TIL VENSTRE FOR DIG: TRÆERNE.
```

```
HVILKEN KØLLE VÆLGER DU ?2
```

```
SLAGET GIK 231 METER.
BOLDEN ER 158 METER FRA HULLET.
BOLDEN LIGGER 16 METER FRA MIDT-
LINIEN, PÅ FAIRWAY'EN.
```

```
HVILKEN KØLLE VÆLGER DU ?15
```

```
PÅ GREEN'EN, 8.18 METER FRA FLAGET.
```

```
INTAST DIN PUT-STYRKE (1-13) ?71
```

```
INTAST DIN PUT-STYRKE (1-13) ?7
```

```
BOLDEN RAMTE VED SIDEN AF HULLET.
```

```
PÅ GREEN'EN, 2.42 METER FRA FLAGET.
```

```
INTAST DIN PUT-STYRKE (1-13) ?3
```

```
BOLDEN RAMTE VED SIDEN AF HULLET.
```

```
PÅ GREEN'EN, 2.42 METER FRA FLAGET.
```

```
INTAST DIN PUT-STYRKE (1-13) ?2
```

```
BOLDEN ER I HUL!
```

```
PÅ HUL 2 FIK DU 5 SLAG.
TOTAL PAR FOR 2 HULLER ER 8
DU HAR SLÆT 10
```

```
DU STÅR PÅ TEE'EN VED HUL 3
AFSTAND = 206 METER. PAR = 3
```

```
TIL HØJRE FOR DIG: NABOFAIRWAY'EN.
TIL VENSTRE FOR DIG: ROUGH'EN.
```

```
HVILKEN KØLLE VÆLGER DU ?
```


```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                        GOLF"
130 PRINT "-----"
140 PRINT
150 PRINT "VELKOMMEN TIL CREATIVE COMPUTINGS GOLF-"
160 PRINT "KLUB. VI HAR EN 18-HULS MESTERSKABSBANE"
170 PRINT "I NÆRHEDEN AF MORRISTOWN. KOMMENTATOREN"
180 PRINT "FORTÆLLER HVAD DER FOREGÅR I SPILLET."
190 PRINT
200 PRINT "MOR DIG GODT...."
210 PRINT
220 PRINT "VI SES VED DET 19. HUL!"
230 PRINT
240 DIM L(10)
250 G1 = 18
260 G2 = 0
270 G3 = 0
280 A = 0
290 N = .8
300 S2 = 0
310 F = 1
320 PRINT "HVAD ER DIT HANDICAP ";
330 INPUT H
340 PRINT
350 IF H > 30 THEN 980
360 IF H < 0 THEN 980
370 PRINT "DER ER FØLGENDE SVÆRHEDSOMRÅDER:"
380 PRINT
390 PRINT "0 = HOOK"
400 PRINT "1 = SLICE"
410 PRINT "2 = DÆRLIG AFSTANDSBEDØMMELSE"
420 PRINT "4 = BUNKERSLAG"
430 PRINT "5 = PUTTING"
440 PRINT
450 PRINT "HVAD ER DU DÆRLIGST TIL ";
460 INPUT T
470 IF T > 5 THEN 370
480 S1 = 0
490 L(0) = 0
500 J = 0
510 Q = 0
520 S2 = S2 + 1
530 K = 0
540 IF F = 1 THEN 770
550 PRINT "PÅ HUL";F - 1;"FIK DU";S1;"SLAG."
560 GOTO 2880
570 IF S1 > P + 2 THEN 620
580 IF S1 = P THEN 650
590 IF S1 = P - 1 THEN 680
600 IF S1 = P - 2 THEN 710
610 GOTO 770
620 PRINT
630 PRINT "BARE TAG DET ROLIGT."
640 GOTO 770
650 PRINT
660 PRINT "DU KLAREDE PÅ PAR, GODT KLARET."
670 GOTO 770
680 PRINT
690 PRINT "EN BIRDIE."
700 GOTO 770
710 IF P = 3 THEN 750
720 PRINT
730 PRINT "EN EAGLE."
740 GOTO 770
750 PRINT
760 PRINT "I HUL PÅ ET SLAG!"
770 IF F = 19 THEN 2870
780 S1 = 0
790 PRINT
800 IF S1 = 0 THEN 2710
810 IF L(0) < 1 THEN 2190
820 X = 0
830 IF L(0) > 5 THEN 2230
840 PRINT "SLAGET GIK";D1;"METER."
850 PRINT "BOLDEN ER";D2;"METER FRA HULLET."
860 PRINT "BOLDEN LIGGER"; INT (0);"METER FRA MIDT-"
870 PRINT "LINIEN, ";
880 GOSUB 900
890 GOTO 1200
900 IF L(X) = 1 THEN 1010
910 IF L(X) = 2 THEN 1040
920 IF L(X) = 3 THEN 1070
930 IF L(X) = 4 THEN 1100
940 IF L(X) = 5 THEN 1130
950 IF L(X) = 6 THEN 1160
960 PRINT "UDENFOR BANEN."
970 GOTO 2860
```



```

2700 GOTO 2630
2710 READ D,P,L(1),L(2)
2720 PRINT
2730 PRINT "-----"
2740 PRINT
2750 PRINT "DU STÅR PÅ TEE'EN VED HUL";F
2760 PRINT "AFSTAND =";D;"METER. PAR =";P
2770 PRINT
2780 G3 = G3 + P
2790 PRINT "TIL HØJRE FOR DIG: ";
2800 X = 1
2810 GOSUB 3000
2820 PRINT "TIL VENSTRE FOR DIG: ";
2830 X = 2
2840 GOSUB 3000
2850 GOTO 1200
2860 RETURN
2870 PRINT
2880 G2 = G2 + S1
2890 PRINT "TOTAL PAR FOR";F - 1;"HULLER ER";G3
2900 PRINT "DU HAR SLÅET";G2
2910 IF G1 = F - 1 THEN 2930
2920 GOTO 570
2930 PRINT
2940 PRINT "-----"
2950 PRINT
2960 PRINT "TAK FOR SPILLET."
2970 PRINT
2980 PRINT "-----"
2990 GOTO 3110
3000 IF L(X) = 1 THEN 1020
3010 IF L(X) = 2 THEN 1050
3020 IF L(X) = 3 THEN 1080
3030 IF L(X) = 4 THEN 1110
3040 IF L(X) = 5 THEN 1140
3050 IF L(X) = 6 THEN 1180
3060 PRINT "UDENFOR BANEN."
3070 GOTO 2860
3080 DATA 361, 4, 4, 2, 389, 4, 3, 3, 206, 3, 4, 2,
500, 5, 7, 2, 408, 4, 2, 4, 359, 4, 6, 4
3090 DATA 424, 4, 4, 2, 388, 4, 4, 4, 196, 3, 7, 2,
400, 4, 7, 2, 560, 5, 7, 2, 132, 3, 2, 2
3100 DATA 357, 4, 4, 4, 294, 4, 2, 4, 475, 5, 2, 3,
375, 4, 4, 2, 180, 3, 6, 2, 550, 5, 6, 6
3110 END

```



Appendix

Basic-test

Dette program bør indtastes og køres før bogens andre programmer.

Ved hjælp af BASIC-TEST vil du hurtigt være i stand til at undersøge hvilke af bogens BASIC-instruktioner, der fejlfortolkes af den BASIC din datamat bruger. Indtast programmet omhyggeligt og kød det. Hvis din computer kan køre programmet og giver de samme resultater (bortset fra RND-funktionen – selvfølgelig) som nedenstående kørsel, så vil du ikke få problemer med bogens programmer. Husk dog, at de værdier som funktionerne beregner heller ikke behøver at passe på de sidste decimaler. Hvis BASIC-TEST standser midt i kørslen eller giver fejlmeddelelse, så find ud af, hvilken programlinie der ikke udføres som forventet. Arbejd med problemet og prøv forskellige muligheder, indtil din version af BASIC-TEST udskriver de samme resultater som prøvekørslen. På den måde kan du være sikker på, at problemet er løst tilfredsstillende, og at den samme løsning vil kunne anvendes i bogens andre programmer.

Programmet er skrevet af Christian Borup.

```

-----
                        BASIC-TEST
-----

*** DETTE ER ET BASIC TEST PROGRAM ***

3.333
-3.333

*** INDBYGGEDE FUNKTIONER ***

-.190240727
-.981737473
.193779633
1.27931201
28.0222826
1.2038728
1.82565057
5.555      0      5.555
-6          0      5
-1          0      1

OBS!
NEDENSTRENDE VÆRDIER KAN AFVIGE PÅ DIN
COMPUTER, DE SKAL BLOT VÆRE STØRRE END
0 OG MINDRE END 1

.93753437      .0544707574
.246017374      .220233661
.975331284      .253108841
.869225068      .533446289
.892159924      .225467183

***** ** *

*** HOP OG LØKKER ***

GOTO UDFØRT
GOSUB UDFØRT
IF-THEN UDFØRT
IF-THEN GOSUB UDFØRT
ON-GOTO UDFØRT
ON-GOSUB UDFØRT

-2      -2
-2      -1
-2      0
-2      1
-2      2
0      0
0      1
0      2
2      2

*** DATA-SÆTNINGER ***

```

```

100 PRINT
110 PRINT "-----"
120 PRINT "                        BASIC-TEST"
130 PRINT "-----"
140 REM DETTE ER EN REMARK-SÆTNING
150 PRINT
160 A = 0:B = 2
170 C = 3.333:C1 = - 3.333
180 D = 5.555:D1 = - 5.555
190 K$ = "COMPUTER SPIL"
200 DIM A(5),B(2,2),A$(10)
210 PRINT "*** DETTE ER ET BASIC TEST PROGRAM ***"
220 PRINT
230 PRINT C
240 PRINT C1
250 PRINT
260 PRINT "*** INDBYGGEDE FUNKTIONER ***"
270 PRINT
280 PRINT SIN (C)
290 PRINT COS (C)
300 PRINT TAN (C)
310 PRINT ATN (C)
320 PRINT EXP (C)
330 PRINT LOG (C)
340 PRINT SQR (C)
350 PRINT ABS (D1), ABS (A), ABS (D)
360 PRINT INT (D1), INT (A), INT (D)
370 PRINT SGN (D1), SGN (A), SGN (D)
380 PRINT
390 PRINT "OBS!"
400 PRINT "NEDENSTRENDE VÆRDIER KAN AFVIGE PÅ DIN"
410 PRINT "COMPUTER, DE SKAL BLOT VÆRE STØRRE END"
420 PRINT "0 OG MINDRE END 1"
430 PRINT
440 FOR I = 1 TO 5
450 PRINT RND (1), RND (1)
460 NEXT I
470 PRINT
480 PRINT "*****"; TAB( 12);"*";
490 PRINT TAB( 13);"*"; TAB( 15);"*"
500 PRINT
510 PRINT
520 PRINT "*** HOP OG LØKKER ***"
530 PRINT
540 GOTO 1370
550 GOSUB 1390
560 IF C = 3.333 THEN 1410
570 IF C = 3.333 THEN GOSUB 1430
580 IF A < > 0 THEN 1450: PRINT "DENNE BASICS
IF-THEN PROCEDURE FORKERT": GOTO 580
590 ON B GOTO 1470,1450,1470
600 ON B GOSUB 1490,1510,1490
610 PRINT
620 FOR I = - 2 TO 2 STEP 2
630 FOR J = I TO 2
640 PRINT I,
650 PRINT J
660 NEXT J
670 NEXT I
680 PRINT
690 PRINT "*** DATA-SÆTNINGER ***"
700 PRINT
710 DATA 5,4,3,2,1
720 DATA "E","d","c","B","A"
730 DATA "5","4","3","2","1"
740 FOR I = 1 TO 5
750 READ A(I)
760 PRINT A(I);
770 NEXT I
780 PRINT
790 FOR I = 1 TO 10
800 READ A$(I)
810 PRINT A$(I);
820 NEXT I
830 PRINT
840 PRINT
850 RESTORE
860 FOR J = 1 TO 5
870 READ M
880 NEXT J
890 PRINT M
900 PRINT
910 PRINT "*** OPERATORER ***"
920 PRINT

```

```

930 PRINT (C + C) * D1 / B - C1 ^ 2
940 PRINT C + C * D1 / B - C1 ^ 2
950 PRINT ABS (1 < > 1), ABS (2 < 1)
960 PRINT ABS (1 = 1), ABS (2 > 1)
970 PRINT ABS ((1 > = 1) AND (2 > = 3))
980 PRINT ABS ((1 > = 1) OR (2 > = 3))
990 PRINT
1000 PRINT "*** BRUGER-DEFINERED FUNKTIONER ***"
1010 PRINT
1020 DEF FN R(X) = X * 10 + X / (X / X)
1030 X = - 4
1040 PRINT FN R(X), FN R(C1)
1050 PRINT
1060 PRINT "*** STRENGE ***"
1070 PRINT
1080 FOR I = 5 TO 1 STEP - 1
1090 PRINT ASC (A$(I)), CHR$ (70 - I)
1100 NEXT I
1110 PRINT
1120 FOR I = 10 TO 6 STEP - 1
1130 PRINT STR$ (I - 5), VAL (A$(I))
1140 NEXT I
1150 PRINT
1160 PRINT LEN (K$)
1170 PRINT
1180 PRINT LEFT$ (K$,3);"*";
1190 PRINT MID$ (K$,4,7);"*";
1200 PRINT RIGHT$ (K$,3)
1210 PRINT
1220 A$ = A$(8) + "-" + A$(9) + "-" + A$(10) + "-0"
1230 PRINT A$
1240 PRINT
1250 PRINT "*** DENNE LØKKE RINGER KLOKKEN ***"
1260 FOR I = 1 TO 5
1270 PRINT CHR$ (7);
1280 NEXT I
1290 PRINT
1300 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1310 INPUT A$
1320 PRINT
1330 RESTORE
1340 IF LEFT$ (A$,1) = "J" THEN 210
1350 GOTO 1360
1360 END
1370 PRINT "GOTO UDFØRT"
1380 GOTO 550
1390 PRINT "GOSUB UDFØRT"
1400 RETURN
1410 PRINT "IF-THEN UDFØRT"
1420 GOTO 570
1430 PRINT "IF-THEN GOSUB UDFØRT"
1440 RETURN
1450 PRINT "ON-GOTO UDFØRT"
1460 GOTO 600
1470 PRINT "FEJL I ON-GOTO"
1480 GOTO 600
1490 PRINT "FEJL I ON-GOSUB"
1500 RETURN
1510 PRINT "ON-GOSUB UDFØRT"
1520 RETURN
1530 END

```

Fortegnelse over spillene opdelt i kategorier

Morskab

FRASEGENERATOR · 69, bind 1
GODDAV · 70, bind 1
NAVN · 72, bind 2
POESI · 21, bind 3
RUSSISK ROULETTE · 30, bind 3
SAKS-STEN-PAPIR · 31, bind 3
UGEDAG · 72, bind 3

Undervisning

AKTIER · 12, bind 1
BORGERKRIG · 44, bind 1
BYTTEPENGE · 53, bind 1
DYRERIGET · 63, bind 1
HAMMURABI · 9, bind 2
HÆNGT MAND · 18, bind 2
HØVDING · 22, bind 2
KAST · 32, bind 2
KEMIKER · 35, bind 2
KONGE · 36, bind 2
LITTERATUR QUIZ · 52, bind 2
MATEMATISK TERNING · 62, bind 2
PELSHANDLER · 14, bind 3
SYNONYM · 57, bind 3
TOG · 62, bind 3

Billeder og kurver

BANNER · 28, bind 1
BILLE · 33, bind 1
HOP · 16, bind 2
KALENDER · 27, bind 2
KANIN · 31, bind 2
LABYRINT · 47, bind 2
LIV · 53, bind 2
LIV FOR TO · 56, bind 2
LOVE · 58, bind 2
PLOT I 3-D · 20, bind 3
RUDERE · 29, bind 3
SINUSKURVE · 32, bind 3

Gæt tal eller bogstaver

BOGSTAV · 36, bind 1
GÆT · 8, bind 2
HØJT-LAVT · 21, bind 2
OMRING · 10, bind 3
STJERNER · 46, bind 3
TAL · 61, bind 3

Fjern objekter

BUNKE · 52, bind 1
LIGE VINDER · 49, bind 2
LIGE VINDER, VERSION 2 · 50, bind 2
NIM · 8, bind 3
TÆNDSTIKKER · 68, bind 3

Koordinatsystemer

BOMBARDEMENT · 39, bind 1
HURKLE · 17, bind 2
PIZZA · 18, bind 3
SLAGSKIB · 34, bind 3
SMUGORM · 39, bind 3
SÆNKE SLAGSKIBE · 58, bind 3
UBÅDSJAGT · 70, bind 3

Logik

AWARI · 22, bind 1
BAGELS · 24, bind 1
BAGLÆNS · 26, bind 1
BONDESPIL · 41, bind 1
CIFFER · 54, bind 1
DAM FOR EN · 59, bind 1
DRONNINGESKAK · 61, bind 1
FLIP-FLOP · 68, bind 1
HAPS · 11, bind 2
KRYDS OG BOLLE · 40, bind 2
KRYDS OG BOLLE, VERSION 2 · 41, bind 2
KRYDS OG BOLLE I 3-D · 42, bind 2
KUBUS · 45, bind 2
MASTER MIND · 59, bind 2
NICOMACHUS · 7, bind 3
ORD · 13, bind 3
SOLITAIRE · 41, bind 3
TÅRNET I HANOI · 69, bind 3

Rumfart

MÅL · 63, bind 2
MÅNELANDING · 65, bind 2
MÅNELANDING, L.E.M. · 67, bind 2
MÅNELANDING, RAKET · 70, bind 2
ORBIT · 11, bind 3
SPLAT · 44, bind 3
SUPER STAR TREK INSTRUKTIONER · 47, bind 3
SUPER STAR TREK X · 50, bind 3

Sport

AMERIKANSK FODBOLD · 15, bind 1
AMERIKANSK FODBOLD, VERSION 2 · 19, bind 1
BASKETBALL · 30, bind 1
BOKSNING · 37, bind 1
BOWLING · 49, bind 1
GOLF · 72, bind 1
ISHOCKEY · 23, bind 2
PILESPIL · 16, bind 3
SLALOM · 37, bind 3
TYREFÆGTNING · 64, bind 3

Kasino og hasard

CRAPS · 56, bind 1
ENOGTYVE · 65, bind 1
HESTEVÆDDELØB · 14, bind 2
POKER · 22, bind 3
ROULETTE · 26, bind 3
TERNING · 62, bind 3
TYVEKNÆGT · 66, bind 3

Kort- og brætspil

DAM · 57, bind 1
GOMOKO · 7, bind 2
HASARD · 13, bind 2
KRIG · 39, bind 2

Krigsspil

KAMP · 29, bind 2
KAST BOMBERNE · 33, bind 2
SKYTTE · 33, bind 3

Borgens Computerbøger

David Ahl

Basic computerspil

Bind 1-3, 80 sider pr. bd., ill.

Klassisk samling af spil til computere. Spillene er skrevet i BASIC og vil kunne bruges på de fleste computere.

Barker, Jacobi og Zimmermann

Mikrodatamaten som tegneredskab

347 sider, ill.

Om at fremstille tegninger ved hjælp af mikrocomputere. Bogen er også håndbog i brugen af MONSTER-programmet, som gør datamaten til en tegnemaschine.

X.T. Bui

Bedre beslutninger med BASIC

234 sider, ill.

Programsamling til brug for beslutningstagere. Behandler brugen af PERT, prognosemodeller, nulpunktsanalyse, investeringsmodeller m.m.

Oluf Danielsen

EDB i hverdagen

42 sider, ill.

Debatbog til brug i folkeskolens samtidsorientering.

Andreas Driple

Maskinkodeprogrammering med 6502

183 sider

Bogen, som henvender sig til begyndere, gennemgår instruktionssættet for 6502-processoren og forklarer brugen af en assembler.

Walter A. Ettlín

Wordstar på en nem måde

192 sider

Indføring i brugen af det meget udbredte tekstbehandlingsprogram. Indeholder eksempler på færdige opstillinger af breve m.m. Velegnet som både opslagsbog og begyndervejledning.

Jørgen Feder

Pascal – også for begyndere

ca. 385 sider

Den mest omfattende bog på dansk om programmeringssproget Pascal. Fungerer både som indføring og håndbog. Dækker de meget udbredte Poly-Pascal og Turbo-Pascal. Fyldig behandling af brugen af pointere.

Annie Fox og David Fox

Lænestols-BASIC

304 sider, ill.

Helt elementær, humoristisk indføring i programmeringssproget BASIC. Kræver ikke adgang til en computer.

Eugene Galanter

Computeren – hverdagsredskab i hjem og skole

240 sider, ill.

Lettilgængelig bog om børn og computere. Henvender sig til forældre og pædagoger.

Goldschläger og Lister

Datalogi og datamaskiner

320 sider

Dybtgående gennemgang af moderne databehandlingsmetoder med hovedvægten på algoritmebegrebet og principperne for hvordan opgaver omformes til algoritmer.

Tord-Jöran Hallberg

Sådan virker mikrodatamaten

100 sider, ill.

Gennemgår en tænkt mikrodatamat MM og forklarer i detaljer hvordan den programmeres i maskinkode.

Fietz, Kitza og Mantz

Håndbog i MSDOS/PCDOS

ca. 180 sider

Bogen falder i to dele: Først fortælles om hvad operativsystemet er, hvilke opgaver det løser og hvordan man arbejder med det. Bogens anden del er et referenceafsnit, hvor hver enkelt kommando gennemgås. Bogen indeholder et appendiks om nogle af de nye kommandoer i version 3.0 og senere.

Ken Knecht

Microsoft BASIC

261 sider

Populær og omhyggelig gennemgang af denne meget udbredte version af BASIC med mange programeksempler, herunder læsning og skrivning i filer.

Jørn Lorentzen og Henrik Nellager

Maskinkode med Amstrad

340 sider

Om maskinkodeprogrammering på Amstrad CPC464, 664 og 6128. Forfatterne starter med Z80 processorens instruktionsæt og behandler derefter Amstrad-computerens særlige egenskaber m.h.t. grafik, farver og lyd. Stort afsnit om firmwarerutiner, der kan bruges som byggesten i egne programmer.

Jørn Lorentzen og Henrik Nellager

Maskinkode med ZX81

240 sider

Populær begynderbog med mange eksempler, herunder komplet Space Invader spil.

Murtha og Waite

CP/M for begyndere

208 sider

Om CP/M vers. 2.2. Beskriver operativsystemets virkemåde og opgaver. Læseren lærer assemblerprogrammering med DDT og ASM og bogen giver nyttige tips om brugen af STAT, SYSGEN, PIP og andre hjælpeprogrammer.

Glenford J. Myers

Kunsten at teste edb-programmer

246 sider

Fortæller om fejlfinding og afprøvning af edb-programmer. Opstiller et sæt regler og principper, som man kan gå frem efter, når programmer skal fejlrettes.

Erwin Neutzsky-Wulff

Mikrodatamaten

Programmering og anvendelse – en bog om ZX81 BASIC

280 sider

Underholdende og lærerig bog om programmering af hjemme-computeren ZX81.

Erwin Neutzsky-Wulff

Programmering med Commodore BASIC

220 sider, ill.

Om programmering af hjemmecomputeren VIC20. Mange program-eksempler. Kan også bruges med PET og de lidt større Commodore maskiner.

Erwin Neutzsky-Wulff

BASIC med Commodore 64

218 sider, ill.

Bogen er skrevet specielt til Commodore 64 hjemmecomputeren. Indeholder mange morsomme spil og lege som kan testes ind i computeren. Ideel som BASIC-lærebog til selvstudium.

Erwin Neutzsky-Wulff

Amstrad BASIC

200 sider

Bogen er skrevet til brug med Amstrad CPC464, 664 og 6128. Ideel som BASIC-lærebog til selvstudium. Udforsker computerens muligheder og egenskaber. Indeholder mange underholdende spil og lege, der er lige klar til at blive tastet ind i computeren.

Lon Poole og Mary Borchers

En samling nyttige BASIC-programmer

339 sider, ill.

Dækker et bredt spektrum af almindeligt forekommende beregnings-problemer inden for rentesregning, geometri, statistik m.m.

Lon Poole

IBM PC – den personlige computer

480 sider, ill.

Populær bog med hovedvægten på BASIC-programmering med IBM PC. Desuden en fyldig vejledning i brugen af operativsystemet PCDOS. Bogen kan også bruges sammen med IBM PC/XT og AT og med IBM-kompatible computere.

Inge-Lise Salomon

Elementær datalære for HF og tilsvarende undervisningstrin

304 sider, ill.

Fortæller om edb-maskiner, anvendelse af edb, mennesker og edb, samt programmering i COMAL. En grundig bog, der er velegnet til aften-skole og selvstudium, samt undervisning i gymnasiet.

Alan Simpson

Lotus 1-2-3

Det elektroniske kolonneark

332 sider, ill.

Vejledning i brugen af Lotus 1-2-3 regnearket. De forskellige faciliteter gennemgås, herunder brugen af grafik og database, samt brugen af makroer, der gør 1-2-3 til et meget kraftfuldt værktøj.

Ian Sinclair

ZX Spectrum og hvordan man får glæde af den

180 sider

Populær gennemgang af ZX Spectrum hjemmecomputeren.

Jesper Skavin

Maskinkodeprogrammering med Z80

224 sider

Instruktionssættet for Z80 processoren gennemgås og instruktionernes virkemåde forklares.

Ian Stewart og Robin Jones

Lær BASIC-programmering med ZX Spectrum

250 sider

Letforståelig bog om BASIC-programmering på ZX Spectrum hjemme-computeren. Mange færdige programeksempler.

Ian Stewart og Robin Jones

Maskinkode med Commodore 64

200 sider, ill.

Bogen fortæller om hvordan man fra BASIC kan lave maskinkode-programmer. Mange underholdende og lærerige programeksempler.

Ludwigs, Poppensieker og Surowiecki

UNIX for begyndere

ca. 225 sider

Gennemgår UNIX-operativsystemet, der vinder stigende udbredelse i flerbrugersystemer.

Niels Søndergaard

Lav dine egne computereventyr med BASIC

200 sider, ill.

Første bog på dansk om en ny og underholdende anvendelse af computere. Computereventyr er tekstspil, hvor computeren på skærmen forklarer hvad der sker, og spørger dig hvad du vil gøre. Du befinder dig måske i en ufremkommelig jungle, hvor du skal finde en skat og forsvare dig mod alskens farer.

Carl Townsend

Programmering med dBASE II

250 sider

Bogen fortæller detaljeret hvordan man selv kan bruge dBASE II til opbygning af databaser til mange forskellige formål. Talrige eksempler.

Her er en samling af de bedste spil til computere.

Hvert spil er omtalt med en indledning, en listning af programmet lige klar til at taste ind, og et prøvespil der viser, hvordan spillet forløber.

Alle spillene er skrevet i Microsoft BASIC, som i forskellige udgaver findes på flertallet af microcomputere.

Bogen hjælper dig med at tilpasse spillene, hvis du har en computer med en anden BASIC.

Bogen henvender sig til alle brugere af mikrocomputere.

Bind I

BORGEN

ISBN 87-418-7419-6

